

75 PROGRAMMI IN BASIC PER IL VOSTRO COMPUTER

Lon Poole
Mary Borchers
Carrol Donahue



EDIZIONE
ITALIANA

GRUPPO
EDITORIALE
JACKSON

75 PROGRAMMI IN BASIC PER IL VOSTRO COMPUTER

di
**Lon Poole
Mary Barchers
Carroll Donahue**



**GRUPPO
EDITORIALE
JACKSON**
Via Rosellini, 12
20124 Milano

© Copyright per l'edizione originale McGraw-Hill, Inc. 1980

© Copyright per l'edizione italiana Gruppo Editoriale Jackson 1982

Il Gruppo Editoriale Jackson ringrazia per il prezioso lavoro svolto nella stesura dell'edizione italiana la signora Francesca Di Fiore e l'ing. Roberto Pancaldi.

Tutti i diritti sono riservati. Stampato in Italia. Nessuna parte di questo libro può essere riprodotta, memorizzata in sistemi di archivio, o trasmessa in qualsiasi forma o mezzo, elettronico, meccanico, fotocopia, registrazione o altri senza la preventiva autorizzazione scritta dell'editore.

Prima edizione novembre 1982

Stampato in Italia da:
S.p.A. Alberto Matarelli - Milano - Stabilimento Grafico

SOMMARIO

Introduzione	V
Valore futuro di un investimento	1
Valore futuro di un deposito regolare (attualità)	3
Deposito regolare	5
Prelievi regolari da un investimento	7
Investimento iniziale	9
Minimo investimento per prelievi	11
Tasso di interesse nominale su un investimento	13
Tasso di interesse effettivo su un investimento	15
Tabella dell'interesse guadagnato	17
Quota di deprezzamento	21
Ammontare del deprezzamento	23
Valore risparmiato	25
Sconto di una tratta commerciale	27
Capitale su un prestito	29
Pagamento regolare di un prestito	31
Ultimo pagamento di un prestito	33
Bilancio rimanente di un prestito	36
Termine di un prestito	39
Quota di interesse annuale su un prestito	41
Tabella di ammortamento di una ipoteca	44
Massimo comun denominatore	49
Fattori primi di numeri interni	51
Area di un poligono	52
Parti di un triangolo	54
Analisi di due vettori	58
Operazioni su due vettori	60
Conversione angolare: radianti in gradi	61
Conversione angolare: gradi in radianti	63
Conversione di coordinate	65
Disegno di coordinate	67
Grafico di un'equazione polare	71
Rappresentazione di funzioni	75
Interpolazione lineare	79

Interpolazione curvilinea	81
Integrazione: regola di Simpson	83
Integrazione: regola trapezoidale	85
Integrazione: quadratura Gaussiana	87
Derivata	89
Radici di equazioni quadratiche	90
Radice reali di un polinomio: Newton	91
Radici di polinomi: ricerca a mezzo intervallo	93
Polinomio trigonometrico	95
Sistemi di equazione	97
Programmazione lineare	99
Addizione, sottrazione e moltiplicazione per uno scalare di matrici	104
Moltiplicazione tra matrici	107
Inversione di matrici	110
Permutazioni e combinazioni	113
Il test U di Mann - Whitney	115
Media, varianza, deviazione standard	118
Media e deviazione geometrica	121
Distribuzione binomiale	122
Distribuzione di Poisson	124
Distribuzione normale	125
Distribuzione del chi-quadro	127
Test del chi-quadro	130
Distribuzione t di student	132
Test della distribuzione t di student	134
Distribuzione F	137
Coefficiente di correlazione lineare	140
Regressione lineare	142
Regressione lineare multipla	144
Regressione di ordine ennesimo	148
Regressione geometrica	151
Regressione esponenziale	153
Affidabilità di un sistema	155
Tasso medio di crescita, proiezioni	157
Tabella di ammortamento di un onere	159
Stampa di disegni	163
Costo di una ricetta	167
Controllo o rilevamento di mappa	171
Giorno della settimana	178
Numero di giorni tra due date	180
Conversioni di unità di misura	183
Ordinamento alfabetico	186

PREFAZIONE

In questo libro sono raccolti alcuni programmi scritti in linguaggio BASIC (Basic è un marchio registrato dalla Trustees del collegio di Dartmouth). Questi programmi riguardano una vasta casistica di problemi pratici. I programmi sono scritti in Basic standard, questo tipo di Basic è compatibile con molte versioni del Basic oggi utilizzato sui microcomputer. Questo libro può essere utilizzato sia da chi sa scrivere un programma, sia da chi non lo sa scrivere. In esso non viene insegnato come si programma in Basic, vi sono centinaia di libri che forniscono queste indicazioni, ma vengono sviluppati completamente alcuni programmi includendo degli esempi ed il listing di ogni programma trattato.

Nel caso non si abbia familiarità col Basic, non si deve far altro che compilare il listing del programma sul computer; gli esempi illustrano l'esecuzione del programma.

Nel listing sono incluse alcune osservazioni che servono di aiuto al programmatore, in quanto, chiariscono il lavoro che viene svolto dal programma. Esse aiutano anche ad identificare le varie parti del programma, in questo modo le singole parti potranno facilmente essere riutilizzate in altri programmi. Le osservazioni precedono le linee contenenti le operazioni commentate. Gli stati REM possono essere omessi al momento di introdurre il programma, poichè, essi vengono ignorati dal computer.

Alcuni programmi contengono delle opzioni. Una opzione è una possibile alternativa in cui vengono variati i formati delle entrate, o delle uscite, rispetto al programma originale. le opzioni possono suggerire variazioni da apportare ai programmi, esse includono una breve descrizione, un esempio di campione eseguito, ed il listing parziale di ogni opzione. Tale listing parziale include gli stati del programma che si debbono variare per passare dal programma originale a quello opzionale.

Tutti i programmi possono essere eseguiti usando un teletype (Teletype è un marchio registrato della Teletype Corporation) o sistemi ingresso/uscita similari contenenti 72 caratteri.

Nel caso in cui la lunghezza della vostra linea di uscita sia minore di 72 caratteri, dovreste variare gli stati PRINT del programma in modo da renderlo adatto alla vostra uscita.

VALORE FUTURO DI UN INVESTIMENTO

Questo programma calcola il valore futuro di un investimento quando l'interesse è un fattore conosciuto. Bisogna fornire il valore dell'investimento iniziale, il tasso dell'interesse nominale, il numero di periodi composti per anno ed il numero di anni dell'investimento.

Considerando che non vi siano depositi aggiuntivi e che non vi siano prelievi, il valore è basato sulla seguente formula:

$$T = P (1 + i/N)^{N \cdot Y}$$

dove: T = valore totale dopo Y anni (Valore Futuro)
 P = investimento iniziale
 i = tasso d'interesse nominale
 N = numero di periodi composti per anno
 Y = numero di anni

Esempi:

Carl fa un investimento di \$ 6800 al 9,5%. Sapendo che si hanno 4 periodi composti per anno, quale sarà il valore di questo investimento dopo 10 anni?

Il sig. Rossi acquista una proprietà per \$ 16050. Il valore della proprietà aumenta con una media annuale del 7%. Quanto varrà la proprietà del sig. Rossi dopo 5 anni e 1/2?

```
VALORE FUTURO DI UN INVESTIMENTO

INVESTIMENTO INIZIALE? 6800
TASSO DI INTERESSE NOMINALE? 9.5
NUMERO DI PERIODI COMPOSTI PER ANNO? 4
NUMERO DI ANNI? 10
VALORE FUTURO  = $ 17388.64

ALTRI DATI ? (1=SI,0=NO)? 1

INVESTIMENTO INIZIALE? 16050
TASSO DI INTERESSE NOMINALE? 7
NUMERO DI PERIODI COMPOSTI PER ANNO? 1
NUMERO DI ANNI? 5.5
VALORE FUTURO  = $ 23285.51

ALTRI DATI ? (1=SI,0=NO)? 0

READY.
```

Listing del programma

```
10 PRINT "VALORE FUTURO DI UN INVESTIMENTO"
20 PRINT
29 REM - LE RIGHE DA 30 A 100 RICHIEDONO L'INTERVENTO DELL'UTENTE
```

```

30 PRINT "INVESTIMENTO INIZIALE";
40 INPUT P
50 PRINT "TASSO DI INTERESSE NOMINALE";
60 INPUT I
70 PRINT "NUMERO DI PERIODI COMPOSTI PER ANNO";...
80 INPUT N
90 PRINT "NUMERO DI ANNI";
100 INPUT Y
108 REM - CALCOLA IL TASSO DI INTERESSE PER PERIODO
109 REM - CONVERTI DA PERCENTO A DECIMALI
110 I=I/N/100
119 REM - CALCOLA IL VALORE FUTURO CON LA FORMULA
120 T=P*(1+I)T(N*Y)
129 REM - ARROTONDA ALLE CENTINAIA, STAMPA
130 PRINT "VALORE FUTURO = $";INT(T*100+.5)/100
139 REM - STAMPA UNA LINEA BIANCA PER SEPARARE I DATI DALLE DOMANDE
140 PRINT
149 REM - RIPARTI O TERMINI IL PROGRAMMA? SI RICHIEDE L'INTERVENTO DELL'UTENTE
150 PRINT "ALTRI DATI ? (1=SI,0=NO)";
160 INPUT X
170 IF X=1 THEN GOTO20
180 END

```

OPZIONE

Questo programma permette l'introduzione del termine dell'investimento in anni sia come un numero intero che come decimale. In alcuni casi, dovendo introdurre il termine dell'investimento in anni e mesi invece che in soli anni, il programma deve essere cambiato come nell'esempio successivo.

Esempio:

Alberto investe \$ 12000 all'8% di interesse. L'interesse è composto trimestralmente. Quale è il valore di questo investimento dopo 10 anni e 7 mesi?

```

VALORE FUTURO DI UN INVESTIMENTO

INVESTIMENTO INIZIALE? 12000
QUOTA DI INTERESSE NOMINALE? 8
NUMERO DI PERIODI COMPOSTI PER ANNI? 4
NUMERO DI ANNI, MESI ? 10 , 7
VALORE FUTURO = $ 27749.5

ALTRI DATI ? (1=SI,0=NO)? 0

READY.

```

Opzione

```

1 REM - OPZIONE 90-105
10 PRINT "VALORE FUTURO DI UN INVESTIMENTO"
:
:
80 INPUT N
90 PRINT "NUMERO DI ANNI, MESI ";
100 INPUT Y0,M
104 REM - CALCOLA GLI ANNI DAGLI ANNI E MESI
105 Y=(12*Y0+M)/12
108 REM - CALCOLA IL TASSO DI INTERESSE PER PERIODO
:
:
180 END

```

VALORE FUTURO DI UN DEPOSITO REGOLARE (ANNUALITA')

Questo programma calcola il valore futuro di un deposito regolare. Tutti i depositi sono uguali. Bisogna fornire l'ammontare di ogni deposito, il numero di depositi per anno, il numero di anni, e il tasso di interesse nominale.

Considerando che l'interesse è composto ad ogni deposito, il calcolo si basa sulla seguente formula:

$$T = R \cdot \left(\frac{(1 + i/N)^{N \cdot Y} - 1}{i/N} \right)$$

dove: T = valore totale dopo Y anni (valore futuro)
 R = ammontare del deposito regolare
 N = numero di depositi per anno
 Y = numero di anni
 i = tasso di interesse nominale.

Esempi:

Matteo deposita ogni mese \$ 50 su un conto di risparmio del Christmas Club con un interesse del 5%. Quanto riceverà Matteo alla fine dell'anno?

Tim effettua un pagamento annuale di \$ 175. L'interesse è del 5,5 %. Quanto avrà accumulato Tim in 15 anni?

VALORE FUTURO DI UN DEPOSITO REGOLARE (ANNUALITA')

AMMONTARE DEL DEPOSITO REGOLARE? 50
TASSO DI INTERESSE NOMINALE? 5
NUMERO DI DEPOSITI PER ANNO? 12
NUMERO DI ANNI? 1
VALORE FUTURO = \$ 613.94

ALTRI DATI? (1=SI,0=NO)? 1

AMMONTARE DEL DEPOSITO REGOLARE? 175
TASSO DI INTERESSE NOMINALE? 5.5
NUMERO DI DEPOSITI PER ANNO? 1
NUMERO DI ANNI? 15
VALORE FUTURO = \$ 3921.52

ALTRI DATI? (1=SI,0=NO)? 0

Listing del programma

```
10 PRINT "VALORE FUTURO DI UN DEPOSITO REGOLARE (ANNUALITA')"  
20 PRINT  
29 REM - LE RIGHE DA 30 A 100 RICHIEDONO L'INTERVENTO DELL'UTENTE
```

```

30 PRINT "AMMONTARE DEL DEPOSITO REGOLARE";
40 INPUT R
50 PRINT "TASSO DI INTERESSE NOMINALE";
60 INPUT I
70 PRINT "NUMERO DI DEPOSITI PER ANNO";
80 INPUT N
90 PRINT "NUMERO DI ANNI",
100 INPUT Y
108 REM - CALCOLA IL TASSO DI INTERESSE PER IL DEPOSITO,
109 REM - CONVERTI DA PERCENTO IN DECIMALI
110 I=I/N/100
119 REM - CALCOLA IL VALORE FUTURO DALLA FORMULA
120 T=R*((1+I)^(N*Y)-1)/I
129 REM - ARROTONDA ALLE CENTINAIA, STAMPA
130 PRINT "VALORE FUTURO = $";INT(T*100+.5)/100
139 REM - STAMPA UNA LINEA BIANCA PER SEPARARE I DATI DALLA DOMANDA
140 PRINT
149 REM - RIPARTI O TERMINI IL PROGRAMMA? SI RICHIEDE L'INTERVENTO DELL'UTENTE
150 PRINT "ALTRI DATI? (1=SI,0=NO)";
160 INPUT X
170 IF X=1 THEN GOTO 30
180 END

```

OPZIONE

È possibile introdurre il termine dell'investimento in anni e mesi anziché in anni. Il programma va variato come nell'esempio successivo.

Esempio:

Quanto riceverà Ron dopo 10 anni e 5 mesi se egli deposita ogni mese \$ 50 al 5% di interesse?

```

DEPOSITO REGOLARE
/VALORE FUTURO DI UN DEPOSITO REGOLARE (ANNUALITA')

AMMONTARE DEL DEPOSITO REGOLARE? 50
TASSO DI INTERESSE NOMINALE? 5
NUMERO DI DEPOSITI PER ANNO? 12
NUMERO DI ANNI, MESI? 10 , 5
/VALORE FUTURO = $ 8179.31

ALTRI DATI? (1=SI,0=NO)? 0

```

Opzione

```

1 REM - OPZIONE 90-105
10 PRINT "VALORE FUTURO DI UN DEPOSITO REGOLARE (ANNUALITA')
:
:
30 INPUT N
90 PRINT "NUMERO DI ANNI, MESI";
100 INPUT Y0,M
104 REM - CALCOLA GLI ANNI DAGLI ANNI E MESI
105 Y=(12*Y0+M)/12
108 REM - CALCOLA IL TASSO DI INTERESSE PER IL DEPOSITO,
:
:
180 END
READY.

```


DEPOSITO REGOLARE

Questo programma calcola l'ammontare di un deposito regolare necessario al fine di fornire un determinato valore futuro in un periodo di tempo specifico. Tutti i depositi sono uguali. È necessario fornire il valore futuro, il tasso di interesse nominale, il numero di depositi per anno e il numero di anni.

Il calcolo del deposito regolare è basato sulla seguente formula:

$$R = T \left(\frac{i/N}{(1 + i/N)^{N \cdot Y} - 1} \right)$$

dove: R = ammontare del deposito regolare
T = valore futuro
i = tasso di interesse nominale
N = numero di depositi per anno
Y = numero di anni.

Esempio:

Maria vuole avere \$ 1000 sul suo conto alla fine di un anno. Quanto dovrà depositare ogni mese all'8% di interesse perchè ciò si verifichi?

```
DEPOSITO REGOLARE
VALORE TOTALE DOPO Y ANNI? 1000
TASSO NOMINALE DI INTERESSE ? 8
NUMERO DI DEPOSITI PER ANNO? 12
NUMERO DI ANNI? 1
DEPOSITO REGOLARE = $ 80.32
ALTRI DATI ? (1=SI,0=NO)? 0
```

Listing del programma

```
10 PRINT "DEPOSITO REGOLARE";
20 PRINT
25 REM - LE ISTRUZIONI DA 30 A 100 RICHIEDONO L'INTERVENTO DELL'UTENTE
30 PRINT "VALORE TOTALE DOPO Y ANNI";
40 INPUT T
50 PRINT "TASSO NOMINALE DI INTERESSE";
60 INPUT I
70 PRINT "NUMERO DI DEPOSITI PER ANNO";
80 INPUT N
90 PRINT "NUMERO DI ANNI";
100 INPUT Y
105 REM - CALCOLA IL TASSO DI INTERESSE PER IL DEPOSITO
109 REM - CONVERTI DA PERCENTO IN DECIMALI
110 I=I/100
115 REM - CALCOLA IL TOTALE DEL DEPOSITO REGOLARE CON LA FORMULA
120 R=T*I/((I+1)^(N*Y)-1)
125 REM - ARROTONDA ALLE CENTINAIA, STAMPA
```

```

130 PRINT "DEPOSITO REGOLARE = $";INT(R*100+.5)/100
139 REM - STAMPA UNA LINEA BIANCA PER SEPARARE I DATI DALLA DOMANDA
140 PRINT
149 REM - RIPARTI O FINISCI IL PROGRAMMA ? SI RICHIEDE L'INTERVENTO DELL'UTENTE
150 PRINT "ALTRI DATI ? (1=SI,0=NO)";
160 INPUT X
170 IF X=1 THEN GOTO 20
180 END

```

OPZIONE

È possibile introdurre il termine dell'investimento in anni e mesi anziché in anni. Il programma va variato come nell'esempio successivo.

Esempio:

Ed vuole risparmiare \$ 2000 al fine di acquistare una nuova moto. Vuol giungere a questa somma in 1 anno e 5 mesi. Quanto dovrà depositare ogni mese se l'interesse è dell'8%?

```

DEPOSITO REGOLARE
VALORE TOTALE DOPO Y ANNI? 2000
TASSO NOMINALE DI INTERESSE? 8
NUMERO DI DEPOSITI PER ANNO? 12
NUMERO DI ANNI,MESI? 1 , 5
DEPOSITO REGOLARE = $ 111.5
ALTRI DATI ? (1=SI,0=NO)? 0

```

Opzione

```

1 REM OPZIONE 90-105
10 PRINT "DEPOSITO REGOLARE";
:
:
90 INPUT N
90 PRINT "NUMERO DI ANNI,MESI":
100 INPUT Y0,M
104 REM - CALCOLA GLI ANNI DAGLI ANNI E MESI
105 Y=(12*Y0+M)/12
108 REM - CALCOLA IL TASSO DI INTERESSE PER IL DEPOSITO
:
:
180 END
READY.

```

PRELIEVI REGOLARI DA UN INVESTIMENTO

Questo programma calcola il massimo ammontare che può essere prelevato con regolarità da un investimento su un periodo di tempo specifico. Tutti i prelievi sono uguali. Bisogna fornire l'ammontare dell'investimento iniziale, il tasso di interesse nominale, il numero di prelievi per anno ed il numero di anni.

Il massimo ammontare di prelievi è fornito dalla seguente formula:

$$R = P \left(\frac{i/N}{(1 + i/N)^{N \cdot Y} - 1} + \frac{i}{N} \right)$$

dove:

- R = ammontare del prelievo regolare
- P = investimento iniziale
- i = tasso di interesse nominale
- N = numero di prelievi per anno
- Y = numero di anni.

Poichè questo programma calcola il totale massimo, alla fine del periodo si dovrà avere un bilancio di \$ 0. È possibile prelevare un minore ammontare sotto le stesse specifiche, in tal caso sul conto si avrà un bilancio rimanente.

Esempio:

Davide investe \$ 8000 al 9,5 %. Egli pensa di effettuare un prelievo regolare ogni mese per 10 anni, lasciando un bilancio finale nullo. Quanto dovrà prelevare ogni volta?

```
PRELIEVI REGOLARI DA UN INVESTIMENTO
INVESTIMENTO INIZIALE? 8000
TASSO NOMINALE DI INTERESSE? 9.5
NUMERO DI PRELIEVI PER ANNO? 12
NUMERO DI ANNI? 10
AMMONTARE DI OGNI PRELIEVO = $ 103.52
ALTRI DATI ? (1=SI,0=NO)? 0
```

Listing del programma

```
10 PRINT "PRELIEVI REGOLARI DA UN INVESTIMENTO";
20 PRINT
25 REM - LE ISTRUZIONI DA 30 A 100 RICHIEDONO L'INTERVENTO DELL'UTENTE
30 PRINT "INVESTIMENTO INIZIALE";
40 INPUT P
50 PRINT "TASSO NOMINALE DI INTERESSE";
60 INPUT I
```

```

70 PRINT "NUMERO DI PRELIEVI PER ANNO";
80 INPUT N
90 PRINT "TASSO DI ANNI";
100 INPUT Y
108 REM - CALCOLA IL TASSO DI INTERESSE PER IL PRELIEVO
109 REM - CONVERTI DA PERCENTO IN DECIMALI
110 I=Y/100
119 REM - CALCOLA IL PRELIEVO REGOLARE DALLA FORMULA
120 R=P*(I/((1+I)^(N*Y)-1))+I
129 REM - ARROTONDA AL CENTESIMO, STAMPA
130 PRINT "AMMONTARE DI OGNI PRELIEVO = $";INT(R*100+.5)/100
139 REM - STAMPA UNA LINEA BIANCA PER SEPARARE I DATI DALLE DOMANDE
140 PRINT
149 REM - RIPARTI O TERMINI IL PROGRAMMA? SI RICHIEDE L'INTERVENTO DELL'UTENTE
150 PRINT "ALTRI DATI ? (1=SI,0=NO)";
160 INPUT X
170 IF X=1 THEN GOTO20
180 END

```

OPZIONE

Può essere più conveniente introdurre il periodo di investimento in anni e mesi anziché solo in anni. I cambiamenti necessari sul programma sono mostrati nel successivo esempio.

Esempio:

Quanto di potrà prelevare ogni settimana se si dispone di un investimento di \$ 8000 al 9,5% di interesse e i prelievi si effettueranno per 10 anni e 5 mesi?

```

PRELIEVI REGOLARI DA UN INVESTIMENTO
INVESTIMENTO INIZIALE? 8000
TASSO NOMINALE DI INTERESSE? 9.5
NUMERO DI PRELIEVI PER ANNO? 52
NUMERO DI ANNI,MESI? 10 , 5
AMMONTARE DI OGNI PRELIEVO = $ 23.28

ALTRI DATI ? (1=SI,0=NO)? 0

```

Opzione

```

1 REM OPZIONE 90-105
10 PRINT "PRELIEVI REGOLARI DA UN INVESTIMENTO";
:
:
80 INPUT N
90 PRINT "NUMERO DI ANNI,MESI";
100 INPUT Y0,M
104 REM - CALCOLA GLI ANNI DAGLI ANNI E MESI
105 Y=(12*Y0+M)/12
108 REM - CALCOLA IL TASSO DI INTERESSE PER IL PRELIEVO
:
:
180 END

```

INVESTIMENTO INIZIALE

Questo programma calcola l'investimento necessario per fornire un determinato valore futuro in uno specifico periodo di tempo. Bisogna introdurre il valore futuro dell'investimento, il numero di anni dell'investimento, il numero di periodi composti per anno e il tasso di interesse nominale.

La formula usata per calcolare l'investimento iniziale è la seguente:

$$P = \frac{T}{(1 + i/N)^{N \cdot Y}}$$

dove: P = investimento iniziale
T = valore futuro
N = numero di periodi composti per anno
Y = numero di anni
i = tasso di interesse nominale.

Esempi:

Quanto dovrete investire all'8,5% per avere \$ 10000 dopo 10 anni se l'interesse è composto trimestralmente?

La Merchant Savings vuol vendere una obbligazione che varrà \$ 5000 dopo 5 anni della data d'acquisto. L'interesse del 7,9 % sarà composto giornalmente. Quale sarà l'investimento iniziale per l'obbligazione?

```
INVESTIMENTO INIZIALE
VALORE TOTALE DOPO Y ANNI? 10000
NUMERO DI PERIODI COMPOSTI PER ANNO? 4
NUMERO DI ANNI? 10
TASSO NOMINALE DI INTERESSE? 8.5
INVESTIMENTO INIZIALE = $ 4312.38
```

```
ALTRI DATI ? (1=SI,0=NO)? 1
```

```
VALORE TOTALE DOPO Y ANNI? 5000
NUMERO DI PERIODI COMPOSTI PER ANNO? 365
NUMERO DI ANNI? 5
TASSO NOMINALE DI INTERESSE? 7.9
INVESTIMENTO INIZIALE = $ 3368.54
```

```
ALTRI DATI ? (1=SI,0=NO)? 0
```

Listing del programma

```
10 PRINT "INVESTIMENTO INIZIALE";
20 PRINT
29 REM - LE ISTRUZIONI DA 30 A 100 RICHIEDONO L'INTERVENTO DELL'UTENTE
30 PRINT "VALORE TOTALE DOPO Y ANNI";
40 INPUT T
50 PRINT "NUMERO DI PERIODI COMPOSTI PER ANNO";
60 INPUT N
```

```

70 PRINT "NUMERO DI ANNI";
80 INPUT Y
90 PRINT "TASSO NOMINALE DI INTERESSE";
100 INPUT I
108 REM - CALCOLA LA IL TASSO INTERESSE PER PERIODO
109 REM - COVERTI DA PERCENTO A DECIMALI
110 I=I/N/100
119 REM - CALCOLA L'INTERESSE INIZIALE CON LA FORMULA
120 P=T/(1+I)N
130 PRINT "INVESTIMENTO INIZIALE = $ ";INT(P*100+.5)/100
139 REM - STAMPA UNA LINEA BIANCA PER SEPARARE I DATI DALLE DOMANDE
140 PRINT
149 REM - RIPARTI O TERMINI IL PROGRAMMA ? SI RICHIEDE L'INTERVENTO DELL'UTENTE
150 PRINT "ALTRI DATI ? (1=SI,0=NO)";
160 INPUT X
170 IF X=1 THEN GOTO 20
180 END

```

OPZIONE

Il programma precedente permette di introdurre il periodo di investimento in anni o in parti decimali di anni. Si può introdurre il periodo in anni e mesi anziché solo in anni. I cambiamenti necessari nel programma sono mostrati nell'esempio seguente.

Esempio:

Maria vuole investire una somma in un libretto di risparmio. In 3 anni e 8 mesi vuole avere \$ 4000 sul suo conto. Se l'interesse dell'8% è composto mensilmente, quanto dovrà investire?

```

INVESTIMENTO INIZIALE
VALORE TOTALE DI ANNO' ANNI? 4000
NUMERO DI PERIODI COMPOSTI PER ANNO? 12
NUMERO DI ANNI,MESI? 3 , 8
TASSO NOMINALE DI INTERESSE? 8
INVESTIMENTO INIZIALE = $ 2986

ALTRI DATI ? (1=SI,0=NO)? 0

```

Opzione

```

1 REM - OPZIONE 70-85
10 PRINT "INVESTIMENTO INIZIALE";
:
:
60 INPUT N
70 PRINT "NUMERO DI ANNI,MESI";
80 INPUT Y0,M
84 REM - CALCOLA GLI ANNI DAGLI ANNI E MESI
85 Y=(12*Y0+M)/12
90 PRINT "TASSO NOMINALE DI INTERESSE";
:
:
180 END

```

MINIMO INVESTIMENTO PER PRELIEVI

Questo programma calcola il minimo investimento richiesto per permettere dei prelievi regolari su un periodo di tempo specificato. L'ammontare calcolato dipende dall'ammontare di ogni prelievo, dal numero di prelievi per anno dal numero di anni, e dal tasso di interesse nominale dell'investimento. Tutti i prelievi sono uguali.

Solo l'ultimo ammontare necessario per l'investimento è calcolato; il programma considera un bilancio finale di \$ 0. Ogni investimento maggiore di quello calcolato lascia un bilancio restante.

Assumendo che l'interesse è composto con ogni prelievo, il calcolo è basato sulla seguente formula:

$$P = \frac{R \cdot N}{i} \left(1 - \frac{1}{(1 + i/N)^{N \cdot Y}} \right)$$

dove: P = investimento iniziale
R = ammontare del prelievo regolare
i = tasso di interesse nominale
N = numero di prelievi per anno
Y = numero di anni.

Esempio:

Quanto si deve investire al 6% d'interesse al fine di permettere un prelievo mensile di \$ 100 per 5 anni?

MINIMO INVESTIMENTO PER PRELIEVI

AMMONTARE DEL PRELIEVO? 100
TASSO NOMINALE DI INTERESSE? 6
NUMERO DI PRELIEVI PER ANNO? 12
NUMERO DI ANNI? 5
MINIMO INVESTIMENTO = \$. 5172.56

ALTRI DATI (1=SI, 0=NO)? 0

READY

Listing del programma

```
10 PRINT "INVESTIMENTO MINIMO PER PRELIEVI"  
20 PRINT  
25 REM - LE RIGHE DALLA 30 ALLA 100 RICHIEDONO L'INTRODUZIONE DI DATI DA PARTE  
29 REM - DELL'UTENTE  
30 PRINT "AMMONTARE DEL PRELIEVO";  
40 INPUT R  
50 PRINT "TASSO NOMINALE DI INTERESSE";  
60 INPUT I
```

```

70 PRINT "NUMERO DI PRELIEVI PER ANNO";
80 INPUT N
90 PRINT "NUMERO DI ANNI";
100 INPUT Y
109 REM - CONVERTI DA PERCENTUALE IN DECIMALI
110 I=I/100
119 REM - CALCOLA IL MINIMO INVESTIMENTO TRAMITE LA FORMULA
120  $P=R*N/I*(1-1/((1+I/N)^{(N*Y)}))$ 
129 REM - ARROTONDA AL CENTESIMO, STAMPA
130 PRINT "MINIMO INVESTIMENTO = $.":INT(100*P+.5)/100
139 REM - STAMPA UNA RIGA BIANCA PER SEPARARE I DATI DALLA DOMANDA
140 PRINT
149 REM - RIPARTI O TERMINI IL PROGRAMMA? SI RICHIEDE L'INTERVENTO DELL'UTENTE
150 PRINT "ALTRI DATI (1=SI,0=NO)";
160 INPUT X
170 IF X=1 THEN 20
180 END
READY.

```

OPZIONE

Può essere utile introdurre il termine dell'investimento in anni e mesi anziché in anni. Il programma andrà, in tal caso, variato come viene mostrato nel seguente esempio.

Esempio:

Tony preleva \$ 250 mensilmente per 6 anni e 5 mesi. Quanto sarà l'investimento iniziale al 6% di interesse?

INVESTIMENTO MINIMO PER PRELIEVI

```

AMMONTARE DEL PRELIEVO? 250
TASSO NOMINALE DI INTERESSE? 6
NUMERO DI PRELIEVI PER ANNO? 12
NUMERO DI ANNI, MESI? 6 , 5
MINIMO INVESTIMENTO = $ 15944.81

```

ALTRI DATI (1=SI,0=NO)? 0

Opzione

```

1 REM - OPZIONE 90-105
10 PRINT "INVESTIMENTO MINIMO PER PRELIEVI"
:
:
80 INPUT N
90 PRINT "NUMERO DI ANNI, MESI";
100 INPUT Y0,M
104 REM - CALCOLA GLI ANNI IN BASE AGLI ANNI ED AI MESI
105  $Y=(12*Y0+M)/12$ 
109 REM - CONVERTI DA PERCENTUALE IN DECIMALI
:
:
180 END

```


TASSO DI INTERESSE NOMINALE SU UN INVESTIMENTO

Questo programma calcola il tasso di interesse nominale per un dato investimento iniziale che consente un conosciuto valore futuro in un periodo di tempo conosciuto. Il tasso di interesse nominale è normalmente suddiviso al fine di permettere la sua composizione.

Il tasso di interesse nominale è basato sulla seguente formula:

$$i = N(T/P) \frac{1}{N \cdot Y} - N$$

dove: i = tasso di interesse nominale
 P = investimento iniziale
 T = valore futuro
 N = numero di periodi composti per anno
 Y = numero di anni.

Il tasso di interesse nominale è espresso come tasso annuale anche se si usa i/N come tasso d'interesse nel caso di interesse composto. Il tasso di interesse nominale sarà minore del tasso di interesse effettivo quando l'interesse è composto più di una volta all'anno. Ciò perché il tasso nominale non tiene conto dell'interesse composto sull'interesse guadagnato nei vari periodi di ogni anno. Per esempio, la scheda dell'interesse guadagnato su \$ 100 al 5% composto trimestralmente sarà:

PERIODO	BILANCIO	$\frac{i/100}{N}$	INTERESSE	NUOVO BILANCIO
1	\$ 100	0,0125 =	\$ 1,25	\$ 101,25
2	\$ 101,25	0,0125 =	\$ 1,27	\$ 102,52
3	\$ 102,52	0,0125 =	\$ 1,28	\$ 103,80
4	\$ 103,80	0,0125 =	\$ 1,30	\$ 105,10

Il tasso d'interesse effettivo nell'esempio è di 5,1 %, tuttavia il tasso nominale è del 5%.

Esempi:

Jane investe \$ 945 in un libretto di risparmio. Quattro anni e mezzo dopo il suo capitale è di \$ 1309,79. Se l'interesse è composto mensilmente quale è il tasso di interesse nominale offerto dalla banca?

Disk investe \$ 3000. Dopo 10 anni ha guadagnato \$ 1576 di interesse. Se l'interesse è composto ogni mese, qual'è il tasso di interesse nominale sul conto?

TASSO DI INTERESSE NOMINALE SU UN INVESTIMENTO

CAPITALE? 945
VALORE TOTALE? 1309.79
NUMERO DI ANNI? 4.5
NUMERO DI PERIODI COMPOSTI PER ANNO? 12
TASSO DI INTERESSE NOMINALE = 7.27613009 %

ALTRI DATI? (1=SI, 0=NO)? 1

CAPITALE? 3000
VALORE TOTALE? 4576
NUMERO DI ANNI? 10
NUMERO DI PERIODI COMPOSTI PER ANNO? 12
TASSO DI INTERESSE NOMINALE = 4.2295659 %

ALTRI DATI? (1=SI, 0=NO)? 0

Listing del programma

```
10 PRINT "TASSO NOMINALE DI INTERESSE SU UN INVESTIMENTO"
20 PRINT
28 REM - LE RIGHE DALLA 30 ALLA 100 RICHIEDONO L'INTRODUZIONE DI DATI DA PARTE
29 REM - DELL'UTENTE
30 PRINT "CAPITALE";
40 INPUT P
50 PRINT "VALORE TOTALE";
60 INPUT T
70 PRINT "NUMERO DI ANNI";
80 INPUT Y
90 PRINT "NUMERO DI PERIODI COMPOSTI PER ANNO";
100 INPUT N
109 REM - CALCOLA IL TASSO NOMINALE DI INTERESSE TRAMITE LA FORMULA, STAMPA
110 I2=N*(T/P)^(1/(N*Y))-1)*100
120 PRINT "TASSO DI INTERESSE NOMINALE =";I2;"%"
129 REM - STAMPA UNA RIGA BIANCA PER SEPARARE I DATI DALLA DOMANDA
130 PRINT
139 REM - RIPARTI O TERMINI IL PROGRAMMA? SI RICHIEDE L'INTERVENTO DELL'UTENTE
140 PRINT "ALTRI DATI? (1=SI, 0=NO)";
150 INPUT X
160 IF X=1 THEN 20
170 END
```

TASSO DI INTERESSE EFFETTIVO SU UN INVESTIMENTO

Questo programma calcola l'effettivo tasso d'interesse per un investimento iniziale conosciuto che permette un prestabilito valore futuro in un periodo di tempo specifico. Questo tasso esprime il tasso attuale di interesse guadagnato annualmente sull'investimento.

Il tasso di interesse effettivo è calcolato dalla seguente formula:

$$\text{tasso d'interesse effettivo} = \left(\frac{\text{valore futuro}}{\text{investimento iniziale}} \right)^{1/\text{anni}} - 1$$

È possibile calcolare il tasso di interesse effettivo sull'ammontare dell'investimento e dell'interesse derivato. Oppure si può calcolare il tasso di interesse effettivo necessario per far sì che il capitale raggiunga un valore ipotetico in uno specifico periodo di tempo. Per esempio, investendo \$ 5000 in una banca e desiderando \$ 6800 dopo sei anni, si può predire quale tasso di interesse effettivo dovrà pagare la banca per permettere di raggiungere il capitale prefissato.

Il tasso di interesse effettivo può anche essere usato per calcolare la percentuale effettiva di deprezzamento di un investimento. Nel caso di un'auto, ad esempio, se la si acquista per \$ 7534 e la si rivende per \$ 3555 dopo tre anni, si ricava che il deprezzamento attuale (tasso di interesse negativo) è circa del 22% all'anno.

Esempi:

Jane deposita \$ 945 su un libretto di risparmio. Quattro anni e mezzo dopo il suo conto è di \$ 1309,79. Attualmente quale percentuale del suo investimento iniziale paga la banca annualmente?

Dick ha acquistato la sua auto nel 1970 per \$ 7534,84 la vende nel 1973 per \$ 3555. Qual'è il tasso effettivo di deprezzamento?

TASSO DI INTERESSE EFFETTIVO SU UN INVESTIMENTO

INVESTIMENTO INIZIALE? 945

VALORE TOTALE DOPO Y ANNI? 1309.79

NUMERO DI ANNI? 4.5

TASSO DI INTERESSE ANNUO = 7.52375279 %

ALTRI DATI? (1=SI, 0=NO)? 1

INVESTIMENTO INIZIALE? 7534.84
VALORE TOTALE DOPO Y ANNI? 3555
NUMERO DI ANNI? 3
TASSO DI INTERESSE ANNUO = -22.1506143 %
ALTRI DATI? (1=SI, 0=NO)? 0

Listing del programma

```
10 PRINT "TASSO EFFETTIVO DI INTERESSE SU UN INVESTIMENTO"
20 PRINT
28 REM - LE RIGHE DALLA 30 ALLA 80 RICHIEDONO L'INTRODUZIONE DI DATI DA PARTE
29 REM - DELL'UTENTE
30 PRINT "INVESTIMENTO INIZIALE";
40 INPUT P
50 PRINT "VALORE TOTALE DOPO Y ANNI";
60 INPUT T
70 PRINT "NUMERO DI ANNI";
80 INPUT Y
89 REM - CALCOLA L'EFFETTIVO TASSO DI INTERESSE, STAMPA
90 PRINT "TASSO DI INTERESSE ANNUO =";((T/P)^(1/Y)-1)*100;"%"
100 PRINT
109 REM - STAMPA UNA RIGA BIANCA PER SEPARARE I DATI DALLA DOMANDA
110 PRINT "ALTRI DATI? (1=SI, 0=NO)";
120 INPUT X
130 IF X=1 THEN 20
140 END
```

TABELLA DELL'INTERESSE GUADAGNATO

Questo programma calcola e stampa una tabella degli interessi guadagnati su un investimento. La scheda contiene le seguenti uscite:

- 1) Bilancio periodico
- 2) Interesse accumulato fra due periodi
- 3) Interesse totale accumulato
- 4) Tasso di interesse effettivo

Queste uscite possono essere calcolate per un singolo investimento o per un investimento iniziale con depositi regolari o prelievi. Se la tabella è tabulata per un singolo investimento, si devono fornire l'ammontare dell'investimento iniziale, il tasso di interesse nominale, e il numero di periodi composti per anno. Il nuovo bilancio sarà stampato per un massimo di quattro volte per anno. Se l'interesse è composto per un numero di periodi all'anno minore di quattro il nuovo bilancio sarà stampato ad ogni computazione d'interesse.

Se la scheda viene tabulata per depositi o prelievi regolari, bisogna fornire l'ammontare dell'investimento iniziale, il tasso di interesse nominale, il numero di depositi o prelievi per anno ed il loro ammontare. In tal caso si assume che l'interesse venga composto giornalmente (360 giorni per anno). Il nuovo bilancio viene stampato ad ogni deposito o prelievo.

Esempi:

Sally investe \$ 2000 al 9,5% per 10 anni. L'interesse è composto mensilmente. Quale sarà il suo bilancio e l'interesse guadagnato negli ultimi due anni?

John deposita \$ 1000 all'8%. Ogni mese egli deposita su questo conto \$ 50. Quale sarà la tabella dell'interesse guadagnato su cinque anni per questo conto?

Ted deposita \$ 1000 all'8%. Trimestralmente egli effettua un prelievo di \$ 150. Quale sarà la tabella dell'interesse guadagnato su un anno per questo conto?

TABELLA DELL' INTERESSE GUADAGNATO

CAPITALE? 2000
TASSO DI INTERESSE NOMINALE? 9.5
NUMERO DI DEPOSITI/PRELIEVI PER ANNO? 0
NUMERO DI PERIODI COMPOSTI PER ANNO? 12
CON QUALE ANNO INIZI? 9
LA STAMPA TERMINA CON QUALE ANNO? 10

TABELLA DELL'INTERESSE GUADAGNATO
CAPITALE \$ 2000 AL 9.5 % NOMINALE PER 10 ANNI

TASSO DI INTERESSE EFFETTIVO 9.92 % PER ANNO

ANNO	BILANCIO	INTERESSE	INTERESSE ACCUM.
9	4365.87	2365.86	2365.87
	4470.38	104.51	2470.38
	4577.39	107.01	2577.39
	4686.97	109.58	2686.97
10	4799.17	112.2	2799.17
	4914.06	114.85	2914.06
	5031.7	117.64	3031.7
	5152.15	120.45	3152.15

CAMBI DATI E RICALCOLI? (1=SI, 0=NO)? 1

CAPITALE? 1000

TASSO DI INTERESSE NOMINALE? 8

NUMERO DI DEPOSITI/PRELIEVI PER ANNO? 12

TOTALE DEI DEPOSITI/PRELIEVI? 50

CON QUALE ANNO INIZI? 1

LA STAMPA TERMINA CON QUALE ANNO? 4

TABELLA DELL'INTERESSE GUADAGNATO
CAPITALE \$ 1000 AL 8 % NOMINALE PER 1 ANNI
DEPOSITI/PRELIEVI REGOLARI \$ 50 12 VOLTE ALL'ANNO
TASSO DI INTERESSE EFFETTIVO 8.33 % PER ANNO

ANNO	BILANCIO	INTERESSE	INTERESSE ACCUM.
1	1056.7	6.7	6.7
	1113.79	7.08	13.78
	1171.24	7.46	21.24
	1229.08	7.84	29.08
	1287.32	8.23	37.32
	1345.94	8.62	45.94
	1404.95	9.01	54.95
	1464.36	9.41	64.36
	1524.17	9.81	74.17
	1584.38	10.21	84.38
	1644.98	10.61	94.98
	1706	11.01	106

CAMBI DATI E RICALCOLI? (1=SI, 0=NO)? 1

CAPITALE? 1000

TASSO DI INTERESSE NOMINALE? 8

NUMERO DI DEPOSITI/PRELIEVI PER ANNO? 4

TOTALE DEI DEPOSITI/PRELIEVI? 150

CON QUALE ANNO INIZI? 1

LA STAMPA TERMINA CON QUALE ANNO? 1

TABELLA DELL'INTERESSE GUADAGNATO

CAPITALE \$ 1000 AL 8 % NOMINALE PER 1 ANNI
DEPOSITI/PRELIEVI REGOLARI \$-150 4 VOLTE ALL'ANNO
TASSO DI INTERESSE EFFETTIVO 8.33 % PER ANNO

ANNO	BILANCIO	INTERESSE	INTERESSE ACCUM
1	370.17	20.17	20.17
	737.71	17.54	37.71
	602.58	14.87	52.58
	464.72	12.14	64.72

CAMBI DATI E RICALCOLI? (1=SI, 0=NO)? 0

Listing del programma

```
4 REM - FUNZIONE DI ARROTONDAMENTO
5 DEFN R(X)=INT(X*100+.5)/100
10 PRINT "TABELLA DELL' INTERESSE GUADAGNATO"
20 PRINT
28 REM - LE RIGHE DALLA 30 ALLA 230 RICHIEDONO L'INTRODUZIONE DI DATI DA PARTE
29 REM - DELL'UTENTE
30 PRINT "CAPITALE";
40 INPUT P
50 PRINT "TASSO DI INTERESSE NOMINALE";
60 INPUT I
69 REM - CONVERTI DA PERCENTUALE IN DECIMALE
70 I=I/100
80 PRINT "NUMERO DI DEPOSITI/PRELIEVI PER ANNO";
90 INPUT N1
99 REM - NON DOMANDA L'AMMONTARE SE LA FREQUENZA E' NULLA
100 IF N1=0 THEN 160
108 REM - I DEPOSITI VANNO INTRODOTTI COME NUMERI POSITIVI
109 REM - I PRELIEVI VANNO INTRODOTTI COME NUMERI NEGATIVI
110 PRINT "TOTALE DEI DEPOSITI/PRELIEVI";
120 INPUT R
129 REM - L'INTERESSE E' COMPOSTO GIORNALMENTE
130 N=360
139 REM - STAMPA CIASCUN DEPOSITO/PRELIEVO
140 L2=N1
150 GOTO 200
160 PRINT "NUMERO DI PERIODI COMPOSTI PER ANNO";
170 INPUT N
180 N1=0
189 REM - STAMPA QUATTRO VOLTE PER ANNO
190 L2=4
200 PRINT "CON QUALE ANNO INIZI";
210 INPUT Y
220 PRINT "LA STAMPA TERMINA CON QUALE ANNO";
230 INPUT Y
239 REM - LA STAMPA INIZIA ALL'INIZIO DELL'ANNO
240 X=INT(X)
249 REM - INIZIALIZZA I TOTALI CORRENTI
250 B0=P
260 I1=0
270 I2=0
280 I3=0
290 K=66
300 P1=4
310 FOR J0=1 TO INT(Y)+1
319 REM - INIZIA LA STAMPA?
320 IF J0<X THEN 480
329 REM - TEST DI FINE PAGINA
330 IF K<55 THEN 470
338 REM - SPAZIA FINO ALLA NUOVA PAGINA, STAMPA L'INTERESTAZIONE (66 RIGHE PER
339 REM - PAGINA)
340 FOR K1=K TO 66
```

```

350 PRINT
360 NEXT K1
370 K=6
380 PRINT "          TABELLA DELL'INTERESSE GUADAGNATO"
390 PRINT "  CAPITALE $";P;" AL";I*100;"% NOMINALE PER";Y;"ANNI"
399 REM - SALTA DEPOSITI/PRELIEVI SE NON CE NE SONO
400 IF N1=0 THEN 430
410 PRINT "DEPOSITI/PRELIEVI REGOLARI $";R;" ";N1;" VOLTE ALL'ANNO"
419 REM - K E' IL CONTATORE AL NUMERO DI RIGHE STAMPATE
420 K=K+1
430 PRINT "          TASSO DI INTERESSE EFFETTIVO":FNR(100*((1+I/N)N-1)):
431 PRINT "% PER ANNO"
440 PRINT
450 PRINT "ANNO      BILANCIO      INTERESSE      INTERESSE ACCUM."
459 REM - CALCOLA L'INTERESSE
460 PRINT
469 REM - STAMPA IL NUMERO DELL'ANNO
470 PRINT JO;
480 L1=1
490 N2=1
500 P2=1
510 FOR J1=1 TO N
519 REM - ALTRI DEPOSITI/PRELIEVI QUEST'ANNO?
520 IF N2>N1 THEN 560
529 REM - TEMPO DI FARE DEPOSITI/PRELIEVI?
530 IF N2/N1>J1/N THEN 560
539 REM - CALCOLA IL NUOVO BILANCIO
540 B0=B0+R
549 REM - CONTA I DEPOSITI/PRELIEVI EFFETTUATI IN UN ANNO
550 N2=N2+1
560 B2=B0*(1+I/N)
569 REM - I1=TOTALE INTERESSE PER OGNI PERIODO COMPOSTO
570 I1=B2-B0
579 REM - I3=TOTALE INTERESSE ACCUMULATO TRA LE REGISTRAZIONI
580 I3=I3+I1
589 REM - I2=INTERESSE TOTALE ACCUMULATO ALLA DATA
590 I2=I2+I1
599 REM - ARROTONDA ALL'INTERESSE DEL PERIODO DELLA REGISTRAZIONE
600 IF P2/P1>J1/N THEN 640
610 I2=FNR(I2)
620 B2=FNR(B2)
630 P2=P2+1
639 REM - ANNO DI INIZIO STAMPA?
640 IF JO<X THEN 710
649 REM - TEMPO DI STAMPARE UNA RIGA?
650 IF J1/N<L1/L2 THEN 710
660 L1=L1+1
670 PRINTTAB(10)FNR(B2)TAB(23)FNR(I3)TAB(35)FNR(I2)
678 REM - INTERESSE REGISTRATO, RI-INIZIALIZZA L'INTERESSE ACCUMULATO,TRA LE
679 REM - REGISTRAZIONI
680 I3=0
690 K=K+1
698 REM - IL NUMERO DELL'ANNO VIENE STAMPATO CON LA SOLA PRIMA REGISTRAZIONE
699 REM - IN OGNI ANNO
700 PRINT\
710 B0=B2
719 REM - NESSUN'ALTRA RIGA DA STAMPARE NELL'ULTIMO ANNO
720 IF JO+J1/N-1>Y THEN 780
730 NEXT J1
739 REM - INIZIA STAMPA?
740 IF JO<X THEN 770
750 PRINT
760 K=K+1
770 NEXT JO
780 PRINT
789 REM - RIPARTI O TERMINI IL PROGRAMMA?
790 PRINT "CAMBI DATI E RICALCOLI? (1=SI, 0=NO)";
800 INPUT Z
810 PRINT
820 IF Z=1 THEN 20
840 END

```


QUOTA DI DEPREZZAMENTO

Questo programma calcola la quota di deprezzamento annuale di un investimento. Si devono fornire il prezzo originale della voce, il suo prezzo di rivendita, e la sua età in anni.

La quota di deprezzamento è calcolata dalla seguente formula:

$$\text{quota deprezzamento} = 1 - \left(\frac{\text{prezzo di rivendita}}{\text{prezzo originale}} \right)^{1/\text{età}}$$

Esempio:

Joan aveva acquistato la sua auto per \$ 4933,76 la rivende per \$ 2400 dopo tre anni. Qual'è la quota di deprezzamento attuale?

```
QUOTA DI DEPREZZAMENTO
PREZZO ORIGINALE? 4933.76
PREZZO DI VENDITA? 2400
ANNI ? 3
QUOTA DI DEPREZZAMENTO = 21.354 %
ALTRI DATI (1=SI, 0=NO)? 0
```

Listing del programma

```
10 PRINT "QUOTA DI DEPREZZAMENTO"
20 PRINT
30 PRINT "PREZZO ORIGINALE";
40 INPUT P
50 PRINT "PREZZO DI VENDITA";
60 INPUT T
70 PRINT "ANNI";
80 INPUT Y
90 REM - CALCOLA LA QUOTA DI DEPREZZAMENTO TRAMITE LA FORMULA, CONVERTE
91 REM - IN PERCENTUALE
92 D=100*(1-(T/P)^(1/Y))
93 REM - ARROTONDAMENTO, STAMPA
100 PRINT "QUOTA DI DEPREZZAMENTO =";INT(1000*D+.5)/1000;"%"
110 PRINT
111 REM - RIPARTI I TERMINI IL PROGRAMMA?
120 PRINT "ALTRI DATI (1=SI, 0=NO)";
130 INPUT X
140 IF X=1 THEN 20
150 END
```


AMMONTARE DEL DEPREZZAMENTO

Questo programma calcola il totale deprezzamento in dollari dopo un dato numero di anni per investimenti soggetti a deprezzamento. Bisogna introdurre il prezzo originale dell'investimento, la sua quota di deprezzamento, e gli anni di deprezzamento.

Il deprezzamento totale è calcolato dalla seguente formula:

$$D = P \cdot i \cdot (1 - i)^{Y - 1}$$

dove: D = totale deprezzamento
P = prezzo originale
i = quota di deprezzamento
Y = numero di anni di deprezzamento.

Esempi:

Joan acquista la sua auto per \$ 4933,76. Se la quota media di deprezzamento annuale per quel modello è del 21%, quale sarà l'ammontare del deprezzamento in ogni anno durante i primi tre anni?

Joan vuol conoscere anche il deprezzamento della cinghia di trasmissione della sua auto. Il suo prezzo due anni fa era di \$ 155, la quota di deprezzamento è del 22%. Quale sarà il suo valore nel terzo anno?

AMMONTARE DEL DEPREZZAMENTO

```
PREZZO ORIGINALE? 4933.76
QUOTA DI DEPREZZAMENTO? 21
--(DIGITARE ANNO=0 QUANDO NON DESIDERI PIU' TOTALI PER QUESTA VOCE)--
ANNO? 1
DEPREZZAMENTO = $ 1036.09

ANNO? 2
DEPREZZAMENTO = $ 818.51

ANNO? 3
DEPREZZAMENTO = $ 646.62

ANNO? 0
ALTRI DATI (1=SI, 0=NO)? 1

PREZZO ORIGINALE? 155
QUOTA DI DEPREZZAMENTO? 22
--(DIGITARE ANNO=0 QUANDO NON DESIDERI PIU' TOTALI PER QUESTA VOCE)--
ANNO? 3
DEPREZZAMENTO = $ 20.75

ANNO? 0
ALTRI DATI (1=SI, 0=NO)? 0
```

Ammontare del deprezzamento

```
10 PRINT "AMMONTARE DEL DEPREZZAMENTO"
20 PRINT
30 PRINT "PREZZO ORIGINALE";
40 INPUT P
50 PRINT "QUOTA DI DEPREZZAMENTO";
60 INPUT I
69 REM - CONVERTE DA PERCENTUALE IN DECIMALE
70 I=I/100
80 PRINT "---(DIGITARE ANNO=0 QUANDO NON DESIDERI PIU' TOTALI PER QUESTA VOCE)-
-"
90 PRINT "ANNO";
100 INPUT Y
109 REM - ALTRI CALCOLI PER QUESTA VOCE
110 IF Y=0 THEN 160
119 REM - CALCOLA L'AMMONTARE DEL DEPREZZAMENTO TRAMITE LA FORMULA
120 D=P*I*(1-I)^Y
129 REM - ARROTONDAMENTO AL CENTESIMO, STAMPA
130 PRINT "DEPREZZAMENTO = $";INT(D*100+.5)/100
140 PRINT
149 REM - RICICLA PER IL NUMERO DELL'ANNO SUCCESSIVO
150 GOTO 90
159 REM - RIPARTI O TERMINI IL PROGRAMMA?
160 PRINT "ALTRI DATI (1=SI, 0=NO)";
170 INPUT X
180 IF X=1 THEN 20
190 END
```

VALORE RISPARMIATO

Questo programma calcola il valore risparmiato di una data voce alla fine di un dato anno. È necessario fornire l'età della voce, il suo prezzo originale, e la quota di deprezzamento.

Il valore risparmiato si ottiene dalla seguente formula:

$$S = P(1 - i)^Y$$

dove: S = valore risparmiato
P = prezzo originale
i = quota di deprezzamento
Y = età in anni.

Esempio:

Qual'è il valore risparmiato dell'auto di Joan se tale auto ha tre anni, ed è stata acquistata a \$ 4933,76 con un deprezzamento annuale del 21%? Quale sarà il valore risparmiato nel prossimo anno?

La cinghia di trasmissione dell'auto di Joan è vecchia di due anni. Qual'è il valore risparmiato se essa è costata \$ 155 e se la quota di deprezzamento è del 22%?

VALORE RISPARMIATO

PREZZO ORIGINALE? 4933.76

QUOTA DI DEPREZZAMENTO? 21

--(DIGITARE ANNO=0 QUANDO NON DESIDERI PIU' VALORI PER QUESTA VOCE)-

ANNI? 3

VALORE = \$ 2432.54

ANNI? 4

VALORE = \$ 1921.7

ANNI? 0

ALTRI DATI (1=SI, 0=NO)? 1

PREZZO ORIGINALE? 155

QUOTA DI DEPREZZAMENTO? 22

--(DIGITARE ANNO=0 QUANDO NON DESIDERI PIU' VALORI PER QUESTA VOCE)-

ANNI? 3

VALORE = \$ 73.56

ANNI? 0

ALTRI DATI (1=SI, 0=NO)? 0

Listing del programma

```
10 PRINT "VALORE RISPARMIATO"
20 PRINT
30 PRINT "PREZZO ORIGINALE";
40 INPUT P
```

```

50 PRINT "QUOTA DI DEPREZZAMENTO";
60 INPUT I
70 PRINT "---(DIGITARE ANNO=0 QUANDO NON DESIDERI PIU' VALORI PER QUESTA VOCE)-
80 PRINT "ANNI";
90 INPUT Y
99 REM - CALCOLO UN ALTRO VALORE PER QUESTA VOCE
100 IF Y=0 THEN 140
108 REM - CALCOLA IL VALORE RISPARMIATO TRAMITE LA FORMULA, ARROTONDA, STAMPA
109 REM - LA QUOTA DI DEPREZZAMENTO VIENE CONVERTITA IN DECIMALI PER I CALCOLI
110 PRINT "VALORE = $";INT(100*P*(1-I/100)^Y+.5)/100
120 PRINT
129 REM - RICICLA PER IL NUMERO DELL'ANNO SUCCESSIVO
130 GOTO 80
139 REM - RIPARTI I TERMINI IL PROGRAMMA?
140 PRINT "ALTRI DATI (1=SI, 0=NO)",
150 INPUT %
160 IF %=1 THEN 20
170 END

```

SCONTO DI UNA TRATTA COMMERCIALE

Questo programma calcola l'ammontare dello sconto e il costo netto di una tratta commerciale. Si devono introdurre il valore futuro della tratta, la quota di sconto e il numero di giorni alla maturazione.

Le formule usate per calcolare lo sconto e il costo sono le seguenti:

$$\text{sconto} = T \cdot \frac{D}{100} \cdot \frac{N}{360}$$

$$\text{costo} = T - \text{sconto}$$

dove: T = valore futuro totale
 D = quota di sconto
 N = numero di giorni alla maturazione.

Esempio:

La Canning Corporation acquista una tratta commerciale di \$ 625000 pagabile in 60 giorni al 5,4%. Quali sono lo sconto ed il costo?

SCONTO DI UNA TRATTA COMMERCIALE

VALORE FUTURO? 625000
TASSO DI SCONTO? 5.4
GIORNI ALLA MATURAZIONE? 60
SCONTO = \$ 5625
COSTO = \$ 619375

ALTRI DATI (1=SI, 0=NO)? 0

Listing del programma

```
10 PRINT "SCONTO DI UNA TRATTA COMMERCIALE"
20 PRINT
29 REM - LE RIGHE DALLA 30 ALLA 90 RICHIEDONO L'INTERVENTO DELL'UTENTE
30 PRINT "VALORE FUTURO";
40 INPUT T
50 PRINT "TASSO DI SCONTO";
60 INPUT D
69 REM - CONVERTI DA PERCENTUALE IN DECIMALE
70 D=D/100
80 PRINT "GIORNI ALLA MATURAZIONE";
90 INPUT N
99 REM - CALCOLA LO SCONTO, STAMPA
100 D1=T*D*N/360
110 PRINT "SCONTO = $";D1
119 REM - CALCOLA IL COSTO, STAMPA
```

```
120 PRINT " COSTO = $";T-D1
129 REM - STAMPA UNA RIGA BIANCA PER SEPARARE I DATI DALLA DOMANDA
130 PRINT
139 REM - IL PROGRAMMA RIPARTE O TERMINA? SI RICHIEDE L'INTERVENTO DELL'UTENTE
140 PRINT "ALTRI DATI (1=SI, 0=NO)";
150 INPUT X
160 IF X=1 THEN 20
170 END
```


CAPITALE SU UN PRESTITO

Questo programma calcola l'ammontare iniziale preso in prestito. Questo ammontare è dipendente dal tasso di interesse, l'ammontare dei pagamenti regolari, il numero di pagamenti per anno ed il termine del prestito.

Il calcolo è basato sulla formula:

$$P = \frac{R \cdot N}{i} \cdot \left(1 - \frac{1}{(1 + i/N)^{N \cdot Y}} \right)$$

dove: P = capitale
R = pagamenti regolari
i = tasso di interesse annuale
N = numero di pagamenti per anno
Y = numero di anni.

Esempio:

Susan deve pagare \$ 250 ogni due mesi per tre anni al fine di restituire un prestito al 20% di interesse. Quale era l'ammontare del prestito?

Tom può pagare \$ 180 al mese per risarcire un prestito. Se egli pensa di restituire il prestito in quattro anni e mezzo, e la compagnia che fornisce il prestito richiede il 16% d'interesse, quale sarà il massimo ammontare che Tom può prendere in prestito?

CAPITALE SU DI UN PRESTITO

```
PAGAMENTO REGOLARE? 250
TERMINE IN ANNI? 3
TASSO DI INTERESSE ANNUALE? 20
NUMERO DI PAGAMENTI PER ANNO? 6
CAPITALE = $ 3343.45
```

```
ALTRI DATI (1=SI, 0=NO)? 1
```

```
PAGAMENTO REGOLARE? 180
TERMINE IN ANNI? 4.5
TASSO DI INTERESSE ANNUALE? 16
NUMERO DI PAGAMENTI PER ANNO? 12
CAPITALE = $ 6897.51
```

```
ALTRI DATI (1=SI, 0=NO)? 0
```

Listing del programma

```
10 PRINT "CAPITALE SU DI UN PRESTITO"
20 PRINT
29 REM - LE RIGHE DALLA 30 ALLA 100 RICHIEDONO L'INTERVENTO DELL'UTENTE
```

```

30 PRINT "PAGAMENTO REGOLARE";
40 INPUT R
50 PRINT "TERMINE IN ANNI";
60 INPUT Y
70 PRINT "TASSO DI INTERESSE ANNUALE";
80 INPUT I
90 PRINT "NUMERO DI PAGAMENTI PER ANNO";
100 INPUT N
108 REM - CALCOLA L'AMMONTARE DEL CAPITALE TRAMITE LA FORMULA;
109 REM - L'INTERESSE E' CONVERTITO DA PERCENTUALE IN DECIMALE PER I CALCOLI
110  $P = R * N * ((1 - 1 / ((I / 100) / N + 1) ^ (N * Y)) / (I / 100))$ 
119 REM - ARROTONDA AL CENTESIMO, STAMPA
120 PRINT "CAPITALE = $"; INT(P*100+.5)/100
129 REM - STAMPA UNA RIGA BIANCA PER SEPARARE I DATI DALLA DOMANDA
130 PRINT
139 REM - IL PROGRAMMA RIPARTE O TERMINA?
140 PRINT "ALTRI DATI (1=SI, 0=NO)";
150 INPUT X
160 IF X=1 THEN 20
170 END

```

OPZIONE

In alcuni casi è più conveniente introdurre il termine in anni e mesi anziché in anni. I necessari cambiamenti da apportare al programma sono mostrati nel successivo esempio.

Esempio:

Quale sarà l'ammontare dell'ipoteca se si pensa di pagare \$ 75 al mese per 11 mesi al 3% d'interesse?

CAPITALE SU DI UN PRESTITO

```

PAGAMENTO REGOLARE? 75
TERMINE IN ANNI, MESI? 0 , 11
TASSO DI INTERESSE ANNUALE? 3
NUMERO DI PAGAMENTI PER ANNO? 12
CAPITALE = $ 812.76

```

ALTRI DATI (1=SI, 0=NO)? 0

Opzione

```

1 REM - OPZIONE 50-65
10 PRINT "CAPITALE SU DI UN PRESTITO"
:
:
40 INPUT R
50 PRINT "TERMINE IN ANNI, MESI";
60 INPUT Y0,M
64 REM - CALCOLA GLI ANNI IN BASE AGLI ANNI ED AI MESI
65  $Y = (12 * Y0 + M) / 12$ 
70 PRINT "TASSO DI INTERESSE ANNUALE";
:
:
170 END

```

PAGAMENTO REGOLARE DI UN PRESTITO

Questo programma calcola l'ammontare richiesto come pagamenti regolari al fine di ripagare un prestito in un determinato periodo. I dati che vanno forniti sono il valore del capitale prestato, l'interesse, il numero di pagamenti fatti per anno e il numero di anni richiesti per il rimborso. Questo programma considera uguali tutti i pagamenti.

Il calcolo è basato sulla formula:

$$R = \frac{i \cdot P / N}{1 - \left(\frac{i}{N} + 1 \right)^{-N \cdot Y}}$$

dove: R = pagamento regolare
i = tasso di interesse annuale
P = capitale
N = numero di pagamenti per anno
Y = numero di anni.

Esempi:

Quanto si deve pagare per un prestito di \$ 4.000 all'8% se si effettuano pagamenti trimestrali per cinque anni?

Se Michael prende in prestito \$ 6500 al 12,5% dalla Best Rate Savings & Loan e deve rimborsarle su un periodo di 5,5 anni, quanto dovrà pagare ogni mese?

PAGAMENTO REGOLARE DI UN PRESTITO

```
TERMINE IN ANNI? 5
CAPITALE? 4000
TASSO DI INTERESSE ANNUALE? 8
NUMERO DI PAGAMENTI PER ANNO? 4
PAGAMENTO REGOLARE= $ 244.63
```

```
ALTRI DATI (1=SI, 0=NO)? 1
```

```
TERMINE IN ANNI? 5.5
CAPITALE? 6500
TASSO DI INTERESSE ANNUALE? 12.5
NUMERO DI PAGAMENTI PER ANNO? 12
PAGAMENTO REGOLARE= $ 136.68
```

```
ALTRI DATI (1=SI, 0=NO)? 0
```

Listing del programma

```
10 PRINT "PAGAMENTO REGOLARE DI UN PRESTITO"
20 PRINT
29 REM - LE RIGHE DALLA 30 ALLA 100 RICHIEDONO L'INTERVENTO DELL'UTENTE
30 PRINT "TERMINE IN ANNI";
40 INPUT Y
```

```

50 PRINT "CAPITALE";
60 INPUT P
70 PRINT "TASSO DI INTERESSE ANNUALE";
80 INPUT I
90 PRINT "NUMERO DI PAGAMENTI PER ANNO";
100 INPUT N
108 REM - CALCOLA L'AMMONTARE DEL PAGAMENTO REGOLARE TRAMITE LA FORMULA;
109 REM - L'INTERESSE E' CONVERTITO DA PERCENTUALE IN DECIMALE PER I CALCOLI
110 R=((I/100)*P/N)/((1-1/((I/100)/N+1)^(N*Y))
119 REM - ARROTONDA AL CENTESIMO, STAMPA
120 PRINT "PAGAMENTO REGOLARE= $ ";INT(R*100+.5)/100
129 REM - STAMPA UNA RIGA BIANCA PER SEPARARE I DATI DALLA DOMANDA
130 PRINT
139 REM - IL PROGRAMMA RIPARTE O TERMINA?
140 PRINT "ALTRI DATI (1=SI, 0=NO)";
150 INPUT X
160 IF X=1 THEN 20
170 END
READY.

```

OPZIONE

Può essere conveniente introdurre il termine del pagamento in anni e mesi anziché in soli anni. I cambiamenti necessari al programma sono mostrati nel successivo esempio.

Esempio:

Il sig. Tarry abbisogna di un prestito di \$ 10.000 per acquistare una nuova casa. La Best Rates offre questa cifra al 14% d'interesse se il rimborso viene effettuato in 11 anni e 5 mesi. Quale sarà l'ammontare di un pagamento regolare fatto mensilmente?

PAGAMENTO REGOLARE DI UN PRESTITO

```

TERMINE IN ANNI, MESI? 11 x 5
CAPITALE? 10000
TASSO DI INTERESSE ANNUALE? 14
NUMERO DI PAGAMENTI PER ANNO? 12
PAGAMENTO REGOLARE= $ 146.59

```

ALTRI DATI (1=SI, 0=NO)? 0

Opzione

```

1 REM - OPZIONE 30-45
10 PRINT "PAGAMENTO REGOLARE DI UN PRESTITO"
:
:
29 REM - LE RIGHE DALLA 30 ALLA 100 RICHIEDONO L'INTERVENTO DELL'UTENTE
30 PRINT "TERMINE IN ANNI, MESI";
40 INPUT Y0,M
44 REM - CALCOLA GLI ANNI IN BASE AGLI ANNI ED AI MESI
45 Y=(12*Y0+M)/12
50 PRINT "CAPITALE";
:
:
170 END

```

ULTIMO PAGAMENTO DI UN PRESTITO

Questo programma calcola il valore del pagamento finale di un prestito. Questo pagamento finale deve ammortizzare completamente il prestito alla conclusione della sua durata. Bisogna fornire l'ammontare del prestito, l'ammontare del pagamento regolare, il tasso di interesse, il numero di pagamenti per anno ed il numero di anni.

L'ammontare dell'ultimo pagamento è normalmente differente dall'ammontare dei pagamenti regolari. Il pagamento finale sarà "gonfiato" se l'ultimo pagamento è maggiore dei pagamenti regolari. Un pagamento gonfiato è necessario se applicando l'ammontare dei pagamenti regolari all'ultimo pagamento rimane un bilancio dovuto. Al fine di un completo rimborso del prestito, al termine dello stesso, il bilancio restante va sommato all'ammontare del pagamento regolare per determinare l'ammontare dell'ultimo pagamento.

D'altra parte, l'ammontare del pagamento finale è in alcuni casi il minore del pagamento regolare. Se il pagamento regolare viene effettuato come ultimo pagamento, e da luogo ad un bilancio negativo, significa che l'ultimo pagamento deve essere minore di quello regolare. In questo caso il pagamento regolare viene aggiustato tenendo conto di questo ipotetico bilancio negativo al fine di determinare l'ammontare del pagamento finale.

$$\begin{array}{rcl} \text{ammontare} & & \\ \text{pagamento} & = & \text{pagamento} + \text{ipotetico bilancio dovuto} \\ & & \text{regolare} \quad \text{dopo N.Y. pagamenti regolari} \end{array}$$

dove: N = numero di pagamenti per anno
 Y = numero di anni.

Esempi:

Lynn prende in prestito \$ 6.000 al 5% da suo padre per spese di collegio. Se paga \$ 1.000 ogni anno per sei anni, quanto varrà l'ultimo pagamento?

Lynn prende in prestito \$ 1150 all'8% di interesse, essa rimborsa questo prestito con rate di \$ 75 ogni mese. Un anno e due mesi dopo Lynn decide di recarsi in Europa. Quanto dovrà pagare il prossimo mese per estinguere totalmente il suo debito?

ULTIMO PAGAMENTO DI UN PRESTITO

PAGAMENTO REGOLARE? 1000
CAPITALE? 6000
TERMINE IN ANNI? 7
TASSO DI INTERESSE ANNUALE? 5
NUMERO DI PAGAMENTI PER ANNO? 1
ULTIMO PAGAMENTO = \$ 1300.59

```

ALTRI DATI (1=SI, 0=NO)? 1
PAGAMENTO REGOLARE? 75
CAPITALE? 1150
TERMINE IN ANNI? 1.17
TASSO DI INTERESSE ANNUALE? 8
NUMERO DI PAGAMENTI PER ANNO? 12
ULTIMO PAGAMENTO = $ 240.38

ALTRI DATI (1=SI, 0=NO)? 0

```

Listing del programma

```

10 PRINT "ULTIMO PAGAMENTO DI UN PRESTITO"
20 PRINT
29 REM - LE RIGHE DALLA 30 ALLA 130 RICHIEDONO L'INTERVENTO DELL'UTENTE
30 PRINT "PAGAMENTO REGOLARE";
40 INPUT R
50 PRINT "CAPITALE";
60 INPUT P
70 PRINT "TERMINE IN ANNI";
80 INPUT Y
90 PRINT "TASSO DI INTERESSE ANNUALE";
100 INPUT I
109 REM - L'INTERESSE E' CONVERTITO DA PERCENTUALE IN DECIMALE
110 I=I/100
120 PRINT "NUMERO DI PAGAMENTI PER ANNO";
130 INPUT N
140 B0=P
149 REM - CALCOLA TUTTI I PAGAMENTI, I BILANCI TRAMITE L'ULTIMO PAGAMENTO (R)
150 FOR J1=1 TO N*Y
159 REM - ARROTONDA L'INTERESSE PAGATO AL CENTESIMO
160 I1=INT((B0*I/N)*100+.5)/100
169 REM - CALCOLA IL TOTALE AMMORTIZZATO CON OGNI PAGAMENTO
170 A=R-I1
179 REM - IL BILANCIO RIMANENTE DECRESCA CON OGNI PAGAMENTO
180 B0=B0-A
190 NEXT J1
199 REM - CALCOLA L'ULTIMO PAGAMENTO, ARROTONDA, STAMPA
200 PRINT "ULTIMO PAGAMENTO = $";INT((R+B0)*100+.5)/100
210 PRINT
219 REM - IL PROGRAMMA RIPARTE O TERMINA?
220 PRINT "ALTRI DATI (1=SI, 0=NO)";
230 INPUT X
240 IF X=1 THEN 20
250 END

```

OPZIONE

Il programma precedente permette l'introduzione del termine di pagamento solamente in anni. Per introdurre il termine in anni e mesi il programma deve essere variato come mostra il prossimo esempio.

Esempio:

Se si pagano \$ 40 ogni mese per 2 anni e 3 mesi su un prestito di \$ 1200 al 7,5%, quale dovrà essere l'ammontare dell'ultimo pagamento?

ULTIMO PAGAMENTO DI UN PRESTITO
PAGAMENTO REGOLARE? 40
CAPITALE? 1200
TERMINE IN ANNI E MESI? 2 , 3
TASSO DI INTERESSE ANNUALE? 7.5
NUMERO DI PAGAMENTI PER ANNO? 12
ULTIMO PAGAMENTO = \$ 287.36
ALTRI DATI (1=SI, 0=NO)? 0

Opzione

```
1 REM - OPZIONE 70-85
10 PRINT "ULTIMO PAGAMENTO DI UN PRESTITO"
:
:
60 INPUT P
70 PRINT "TERMINE IN ANNI E MESI";
80 INPUT Y0,M
84 REM - CALCOLA GLI ANNI IN BASE AGLI ANNI ED AI MESI
85 Y=(12*Y0+M)/12
90 PRINT "TASSO DI INTERESSE ANNUALE";
:
:
250 END
```

BILANCIO RIMANENTE DI UN PRESTITO

Questo programma calcola il bilancio restante su un prestito dopo un numero specifico di pagamenti. È necessario fornire l'ammontare dei pagamenti regolari, il numero di pagamenti per anno, l'ammontare del capitale, il tasso di interesse annuale, e il numero di pagamento dopo il quale si vuol calcolare il bilancio restante.

Il bilancio restante è calcolato con il seguente metodo:

bilancio restante = capitale — ammontare ammortizzato dopo
N. (Y-1) + N1 pagamenti

dove: N = numero di pagamenti per anno
Y = anni su cui va calcolato il bilancio rimanente
N1 = numero di pagamenti effettuati in Y anni per calcolare il bilancio.

Esempio:

Kelly ha perso a prestito \$ 8.000 al 17,2% d'interesse. I suoi pagamenti regolari sono di \$ 200 al mese. Quanto deve ancora pagare Kelly dopo aver effettuato il pagamento della rata del decimo mese del quarto anno?

BILANCIO RIMANENTE DI UN PRESTITO

PAGAMENTO REGOLARE? 200
CAPITALE? 8000
NUMERO DI PAGAMENTI PER ANNO? 12
TASSO DI INTERESSE ANNUALE? 17.2
ULTIMO PAGAMENTO EFFETTUATO (NO. DI PAGAMENTO, ANNO)? 10 , 4
BILANCIO RIMANENTE = \$ 2496.17

ALTRI DATI (1=SI, 0=NO)? 0

Listing del programma

```
10 PRINT "BILANCIO RIMANENTE DI UN PRESTITO"
20 PRINT
29 REM - LE RIGHE DALLA 30 ALLA 130 RICHIEDONO L'INTERVENTO DELL'UTENTE
30 PRINT "PAGAMENTO REGOLARE";
40 INPUT R
50 PRINT "CAPITALE";
60 INPUT P
70 PRINT "NUMERO DI PAGAMENTI PER ANNO";
80 INPUT N
90 PRINT "TASSO DI INTERESSE ANNUALE";
100 INPUT I
109 REM - L'INTERESSE E' CONVERTITO DA PERCENTUALE IN DECIMALE
110 I=I/100
119 REM - DIGITA IL NUMERO DI PAGAMENTO ENTRO L'ANNO, I.E. N1<=N
```



```

120 PRINT "ULTIMO PAGAMENTO EFFETTUATO (NO. DI PAGAMENTO, ANNO)";
130 INPUT N1,Y
139 REM - INIZIALIZZA IL BILANCIO RIMANENTE
140 B0=P
149 REM - CICLO PER I TOTALI PAGATI FINO AD ORA
150 FOR J1=1 TO N*(Y-1)+N1
159 REM - CALCOLA L'INTERESSE PAGATO CON OGNI PAGAMENTO
160 I1=INT((B0*I/N)*100+.5)/100
169 REM - CALCOLA IL TOTALE AMMORTIZZATO CON OGNI PAGAMENTO
170 A=R-I1
179 REM - CALCOLA IL BILANCIO RIMANENTE SUL CAPITALE
180 B0=B0-A
190 NEXT J1
199 REM - ARROTONDA, STAMPA
200 PRINT "BILANCIO RIMANENTE = $";INT(B0*100+.5)/100
210 PRINT
219 REM - IL PROGRAMMA RIPARTE O TERMINA?
220 PRINT "ALTRI DATI (1=SI, 0=NO)";
230 INPUT X
240 IF X=1 THEN 20
250 END

```

OPZIONE

Si può specificare il numero dell'ultimo pagamento fatto introducendolo non come numero di pagamento in un certo anno ma come numero totale di pagamenti. Ad esempio, quando si effettuano quattro pagamenti per anno, il 3 pagamento del 3 anno può essere introdotto come pagamento numero 11. Per fare questo il programma va cambiato come viene mostrato nell'esempio successivo.

Esempio:

John fa dieci pagamenti trimestrali di \$ 550 per rimborsare un debito di \$ 6.000 al 16% d'interesse. Quale sarà il bilancio restante?

BILANCIO RIMANENTE DI UN PRESTITO

```

PAGAMENTO REGOLARE? 550
CAPITALE? 6000
NUMERO DI PAGAMENTI PER ANNO? 4
TASSO DI INTERESSE ANNUALE? 16
NUMERO DI PAGAMENTI EFFETTUATI? 10
BILANCIO RIMANENTE = $ 2278.09

```

ALTRI DATI (1=SI, 0=NO)? 0

Opzione

```

1 REM - OPZIONE 119-130, 150
10 PRINT "BILANCIO RIMANENTE DI UN PRESTITO"
:
:
110 I=I/100
119 REM - DIGITA IL NUMERO COMPLESSIVO DI PAGAMENTI EFFETTUATI ALLA DATA

```

```

120 PRINT "NUMERO DI PAGAMENTI EFFETTUATI";
130 INPUT N1
139 REM - INIZIALIZZA IL BILANCIO RIMANENTE
140 B0=P
149 REM - CICLO PER I TOTALI PAGATI FINO AD ORA
150 FOR J1=1 TO N1
159 REM - CALCOLA L'INTERESSE PAGATO CON OGNI PAGAMENTO
.
.
.
250 END

```

TERMINE DI UN PRESTITO

Questo programma calcola il periodo di tempo necessario per rimborsare un prestito. Si devono specificare l'ammontare del prestito, l'ammontare dei pagamenti, il numero di pagamenti fatti per anno e il tasso di interesse annuale. Tutti i pagamenti sono considerati uguali.

$$Y = - \frac{\log \left(1 - \frac{P \cdot i}{N \cdot R} \right)}{\log \left(1 + \frac{i}{N} \right)} \cdot \frac{1}{N}$$

dove: Y = termine del pagamento in anni
P = capitale
i = tasso di interesse annuale
N = numero di pagamenti per anno
R = ammontare dei pagamenti.

Esempio:

Quale sarà la durata di pagamento di una ipoteca di \$ 20.000 al 18% quando si effettuano dei pagamenti di \$ 1.000 ogni trimestre?

Sally prende un prestito di \$ 12.699 al 16,8%. Essa effettua dei pagamenti di \$ 512,34 ogni due mesi. Quale sarà il termine del suo debito?

```
TERMINE DI UN DEBITO
PAGAMENTO REGOLARE? 1000
CAPITALE? 20000
TASSO DI INTERESSE ANNUALE? 18
NUMERO DI PAGAMENTI PER ANNO? 4
TERMINE = 13.1 ANNI
```

```
ALTRI DATI ? (1=SI,0=NO)? 1
```

```
PAGAMENTO REGOLARE? 512.34
CAPITALE? 12669
TASSO DI INTERESSE ANNUALE? 16.8
NUMERO DI PAGAMENTI PER ANNO? 6
TERMINE = 7.1 ANNI
ALTRI DATI ? (1=SI,0=NO)? 0
```

Listing del programma

```
10 PRINT "TERMINE DI UN DEBITO";
20 PRINT
29 REM - GLI STATI DA 30 A 100 RICHIEDONO L' INTERVENTO DELL' UTENTE
```

```

30 PRINT "PAGAMENTO REGOLARE";
40 INPUT R
50 PRINT "CAPITALE";
60 INPUT P
70 PRINT "TASSO DI INTERESSE ANNUALE";
80 INPUT I
90 PRINT "NUMERO DI PAGAMENTI PER ANNO";
100 INPUT N
108 REM - CALCOLA IL TERMINE DI UN DEBITO CON LA FORMULA
109 REM - L' INTERESSE VIENE CONVERTITO DA PERCENTUALE IN DECIMALE PER I CALCOLI
110 Y=-(<math>\log(1-(P*(I/100))/(N*R))</math>)/(<math>\log(1+I/100/N)</math>*N))
112 REM - ARROTONDA AI DECIMI, STAMPA
120 PRINT "TERMINE =";INT(Y*10+.5)/10;"ANNI"
130 PRINT
139 REM - RIPARTI O TREMINI IL PROGRAMMA?
140 PRINT "ALTRI DATI ? (1=SI,0=NO)";
150 INPUT X
160 IF X=1 THEN 20
170 END

```

OPZIONE

È possibile calcolare il termine del pagamento in anni e mesi anzichè in soli anni. Per far ciò, il programma va variato come mostra il successivo esempio.

Esempio:

Disk prende in prestito \$ 8.000 al 7,5%. I suoi pagamenti regolari sono di \$ 150 mensili. Quanto tempo impiegherà per rimborsare il prestito?

```

TERMINE DI UN DEBITO
PAGAMENTO REGOLARE? 150
CAPITALE? 8000
TASSO DI INTERESSE ANNUALE? 7.5
NUMERO DI PAGAMENTI PER ANNO? 12
TERMINE = 5 ANNI 5 MESI

ALTRI DATI ? (1=SI,0=NO)? 0

```

Opzione

```

1 REM - OPZIONE 114-120
10 PRINT "TERMINE DI UN DEBITO";
:
:
110 Y=-(<math>\log(1-(P*(I/100))/(N*R))</math>)/(<math>\log(1+I/100/N)</math>*N))
114 REM - CALCOLA GLI ANNI E I MESI DAGLI ANNI
115 M=INT(Y*12+.5)
116 YO=INT(M/12)
117 M=M-YO*12
119 REM - STAMPA I RISULTATI
120 PRINT "TERMINE =";YO;"ANNI";M;"MESI"
130 PRINT
:
:
170 END

```

QUOTA DI INTERESSE ANNUALE SU UN PRESTITO

Questo programma calcola la quota di interesse caricata su un prestito. Per determinare questa quota bisogna introdurre l'ammontare del prestito, l'ammontare dei pagamenti regolari, il numero di pagamenti per anno, e il termine del prestito.

La quota di interesse viene computata col seguente metodo di approssimazione:

- 1) Si ipotizza una quota di interesse
Inizializziamo l'ultima ipotesi a 0
- 2) Si calcola il pagamento regolare usando la quota ipotizzata:

$$R_1 = \frac{i \cdot P / N}{1 - ((1 + i) / N)^{-N \cdot Y}}$$

Arrotondare il valore di R_1

- 3) Se il pagamento computato = pagamento attuale, allora l'ipotesi corrente è = alla quota di interesse approssimata
- 4) In altro caso, si osserva l'ipotesi corrente e si calcola una nuova ipotesi

$$\begin{aligned} i_2 &= i \\ i &= i \pm |(i - i_2) / 2| \quad \left\{ \begin{array}{l} + \text{ se } R_1 < R \\ - \text{ se } R_1 > R \end{array} \right. \end{aligned}$$

- 5) Si va al punto 2

dove: i = tasso d'interesse
 i_2 = precedente quota di interesse
 R = pagamento regolare introdotto
 R_1 = pagamento regolare calcolato
 P = capitale
 N = numero di pagamenti per anno
 Y = numero di anni.

Esempi:

Cindy prende a prestito \$ 3.000 da suo amico George con l'intenzione di pagare \$ 400 ogni trimestre per 2 anni. Quale sarà la quota d'interesse?

Per rimborsare un prestito di \$ 10.000 John decide di effettuare dei pagamenti mensili di \$ 120 per 9,5 anni. Qual'è la quota di interesse?

QUOTA DI INTERESSE ANNUALE SU UN DEBITO
 PAGAMENTO REGOLARE? 400
 TERMINE DI ANNI? 2
 CAPITALE? 3000
 NUMERO DI PAGAMENTI PER ANNO? 4
 QUOTA DI INTERESSE ANNUALE = 5.827 %

ALTRI DATI ? (1=SI,0=NO)? 1

PAGAMENTO REGOLARE? 120
 TERMINE DI ANNI? 9.5
 CAPITALE? 10000
 NUMERO DI PAGAMENTI PER ANNO? 12
 QUOTA DI INTERESSE ANNUALE = 6.933 %

ALTRI DATI ? (1=SI,0=NO)? 0

Listing del programma

```

10 PRINT "QUOTA DI INTERESSE ANNUALE SU UN DEBITO";
20 PRINT
29 REM - GLI STATI DA 30 A 100 RICHIEDONO L'INTERVENTO DELL'UTENTE
30 PRINT "PAGAMENTO REGOLARE";
40 INPUT R
50 PRINT "TERMINE DI ANNI";
60 INPUT Y
70 PRINT "CAPITALE";
80 INPUT P
90 PRINT "NUMERO DI PAGAMENTI PER ANNO";
100 INPUT N
109 REM - IPOTIZZA UNA QUOTA DI INTERESSE (10%) PER INIZILIZZARE IL TEST
110 I=10
119 REM - I2=ULTIMA IPOTESI O STIMA (PARTI CON 0)
120 I2=0
129 REM - COMPUTA IL PAGAMENTO REGOLARE USANDO LA QUOTA DI INTERESSE IPOTIZZATA
130 R1=(I*P/N)/((1-1/((I/N+1)^(N*Y)))
139 REM - ARROTONDA AL CENTESIMO
140 R1=INT(R1*100+.5)/100
149 REM - I3=NUMERO USATO PER AVVICINARI ALLA QUOTA DI INTERESSE
150 I3=ABS(I-I2)/2
159 REM - TIENI QUESTA IPOTESI
160 I2=I
167 REM - COMPARARE IL PAGAMENTO COMPUTATO(R1) COL PAGAMENTO INIZIALE(R);
168 REM - SE ESSI SONO UGUALI ALLORA :ULTIMA QUOTA IPOTIZZATA = TASSO DI
169 REM - INTERESSE APPROSSIMATO
170 IF R1=R THEN 230
180 IF R1>R THEN 210
189 REM - R1<R, LA QUOTA DEVE ESSERE PIU' ALTA DELL'ULTIMA IPOTIZZATA
190 I=I+I3
199 REM - RIPARTI CON UNA NUOVA IPOTESI
200 GOTO130
209 REM - R1>R, LA QUOTA DEVE ESSERE MINORE DELL' ULTIMA IPOTIZZATA
210 I=I-I3
220 GOTO130
229 REM - CALCOLA L'INTERESSE ALLE PROPRZIONI DOVUTE, ARROTONDA, STAMPA
230 I=((INT((I*1000)*100+.5))/100)/1000
240 PRINT "QUOTA DI INTERESSE ANNUALE = ";I*100;"%"
250 PRINT
260 PRINT "ALTRI DATI ? (1=SI,0=NO)";
270 INPUT X
280 IF X=1 THEN 20
290 END

```

OPZIONE

Il precedente listing permette l'introduzione del termine del prestito solo in anni. Per introdurlo come anni e mesi il programma va variato come mostrato nel successivo esempio.

Esempio:

Se Connie paga \$ 100 al mese per 11 anni e 7 mesi su un prestito di \$ 10.000, qual'è l'attuale quota di interesse per il prestito?

```
QUOTA DI INTERESSE ANNUALE SU UN DEBITO
PAGAMENTO REGOLARE? 100
TERMINE DI ANNI, MESI? 11 , 7
CAPITALE? 10000
NUMERO DI PAGAMENTI PER ANNO? 12
QUOTA DI INTERESSE ANNUALE = 6.002 %

ALTRI DATI ? (1=SI,0=NO)? 0
```

Opzione

```
1 REM - OPZIONE 50-65
10 PRINT "QUOTA DI INTERESSE ANNUALE SU UN DEBITO";
:
:
40 INPUT R
50 PRINT "TERMINE DI ANNI, MESI";
60 INPUT Y0,M
64 REM - CALCOLA GLI ANNI DAGLI ANNI E MESI
65 Y=(12*Y0+M)/12
70 PRINT "CAPITALE";
:
:
290 END
```

TABELLA DI AMMORTAMENTO DI UNA IPOTECA

Questo programma calcola e stampa la tabella del rimborso di un prestito.
La scheda prevede queste uscite:

- 1) Numero del pagamento
- 2) L'ammontare di ogni pagamento pagato con l'interesse
- 3) L'ammontare del debito ammortizzato con ogni pagamento
- 4) Bilancio rimanente sul capitale preso in prestito dopo ogni pagamento
- 5) Cumulo degli interessi pagati dopo ogni pagamento
- 6) Ammontare dell'ultimo pagamento.

In aggiunta, vengono tabulati e stampati anche i totali interessi pagati annualmente, e l'ammontare ammortizzato ogni anno.

Per usare questo programma bisogna fornire l'ammontare dei pagamenti regolari, il termine del pagamento, il numero di pagamenti per anno, l'ammontare del prestito iniziale e il tasso di interesse annuale.

La scheda è calcolata nel seguente modo:

- 1) Numero del pagamento = numero del pagamento nell'anno considerato
- 2) Ammontare di ogni pagamento pagato con l'interesse = bilancio restante $\cdot i/N$
dove: i = tasso di interesse annuale
 N = numero di pagamenti per anno
- 3) Ammontare ammortizzato con ogni pagamento = $R - I$
dove: R = ammontare del pagamento regolare
 I = ammontare di ogni pagamento pagato con l'interesse
- 4) Bilancio restante = $P - \Sigma A$
dove: P = capitale
 ΣA = somma degli ammontari ammortati con ogni pagamento a quella data
- 5) Interesse accumulato = ΣI
dove: ΣI = somma degli ammontari ammortati con ogni pagamento effettuato come interesse a quella data
- 6) Ammontare dell'ultimo pagamento = $R + (P - R \cdot N \cdot Y)$
dove: R = pagamento regolare
 P = capitale
 N = numero di pagamenti per anno
 Y = numero di anni.

Esempio:

A David occorrono \$ 2.100 per pagare alcuni debiti. Sua sorella gli offre questa somma al 6% di interesse. Quale sarà la scheda di rimborso di David se egli paga \$ 75 al mese per 2 anni e 1/2?

TABELLA DI AMMORTAMENTO DI UNA IPOTECA

PAGAMENTI REGOLARI? 75
 TERMINE IN ANNI? 2.5
 CAPITALE? 2100
 TASSO ANNUALE DI INTERESSE? 6
 NUMERO DI PAGAMENTI PER ANNO? 12
 PARTI STAMPANDO QUALE ANNO? 1

TABELLA DI AMMORTAMENTO DI UNA IPOTECA CAPITALE \$ 2100 AL 6 % PER 2.5 ANNI PAGAMENTO REGOLARE = \$ 75

N.	INTERESSE	AMMORTAMENTO	BILANCIO	INTERESSE ACC.
1	10.5	64.5	2035.5	10.5
2	10.18	64.82	1970.68	20.68
3	9.85	65.15	1905.53	30.53
4	9.53	65.47	1840.06	40.06
5	9.2	65.8	1774.26	49.26
6	8.87	66.13	1708.13	58.13
7	8.54	66.46	1641.67	66.67
8	8.21	66.79	1574.88	74.88
9	7.87	67.13	1507.75	82.75
10	7.54	67.46	1440.29	90.29
11	7.2	67.8	1372.49	97.49
12	6.86	68.14	1304.35	104.35

ANNO	1	104.35	795.65	
1	6.52	68.48	1235.87	110.87
2	6.18	68.82	1167.05	117.05
3	5.84	69.16	1097.89	122.89
4	5.49	69.51	1028.38	128.38
5	5.14	69.86	958.52	133.52
6	4.79	70.21	888.31	138.31
7	4.44	70.56	817.75	142.75
8	4.09	70.91	746.84	146.84
9	3.73	71.27	675.57	150.57
10	3.38	71.62	603.95	153.95
11	3.02	71.98	531.97	156.97
12	2.66	72.34	459.63	159.63

ANNO	2	55.28	844.720001	
1	2.3	72.7	386.93	161.93
2	1.93	73.07	313.86	163.86
3	1.57	73.43	240.43	165.43
4	1.2	73.8	166.63	166.63
5	.83	74.17	92.4599996	167.46
6	.46	92.4599992	0	167.92

ULTIMO PAGAMENTO = \$ 92.92

ANNO	3	8.29	459.629999	
------	---	------	------------	--

CAMBI DATI E COMPUTI ANCORA?(1=SI,0=NO)? 0

Listing del programma

```

10 PRINT "TABELLA DI AMMORTAMENTO DI UNA IPOTECA"
20 PRINT
29 REM - LE RIGHE DA 30 A 100 RICHIEDONO L'INTERVENTO DELL'UTENTE
30 PRINT "PAGAMENTI REGOLARI";
40 INPUT R
50 PRINT "TERMINE IN ANNI";
60 INPUT Y
70 PRINT "CAPITALE";
80 INPUT P
90 PRINT "TASSO ANNUALE DI INTERESSE";
100 INPUT I
109 REM - CONVERTI DA PERCENTALE IN DECIMALE
110 I=I/100
120 PRINT "NUMERO DI PAGAMENTI PER ANNO";
130 INPUT N
140 PRINT "PARTI STAMPANDO QUALE ANNO";
150 INPUT X :
170 C1=0
180 I2=0
190 I3=0
200 J0=0
210 N1=N
220 K=66
230 B0=P
240 A1=0
250 A2=0
259 REM - IL TERMINE E' MINORE DI UN ANNO ?
260 IF INT(Y)>=1 THEN 270
261 REM - AGGIUSTA LE VARIABILI PER STAMPARE UN ANNO PARZIALE
262 N1=((Y-INT(Y))*12)/12*N
263 J0=J0+1
264 GOTO280
269 REM - CICLO PER OGNI ANNO
270 FOR J0=1 TO INT(Y)
279 REM - INIZIA STAMPA ?
280 IF J0<N THEN410
289 REM - E' NECESSARIO INIZIARE UNA NUOVA PAGINA ?
290 IF K+N<3<58 THEN 400
298 REM - SPAZIA IL BORDO SUPERIORE DELLA PAGINA NUOVA
299 REM - (ASSUMENDO 66 LINEE PER PAGINA)
300 FOR K1=K TO 66
310 PRINT
320 NEXT K1
339 REM - STAMPA LE INTERSTAZIONI DELLA PAGINA
340 PRINT"          TABELLA DI AMMORTAMENTO DI UNA IPOTECA"
350 PRINT "          CAPITALE $";P;"          AL";I*100;"%          PER";Y;"ANNI"
360 PRINT "          PAGAMENTO REGOLARE = $";R
370 PRINT
380 PRINT "N.          INTERESSE          AMMORTAMENTO          BILANCIO          INTERESSE ACC."
389 REM - CONTA LE LINEE STAMPATE SU OGNI PAGINA IN K
390 K=7
400 K=K+N+3
410 FOR J1=1 TO N1
419 REM - CALCOLA L' INTERESSE PAGATO CON QUESTO PAGAMENTO , ARROTONDA
420 I1=INT((B0*I1/N)*100+.5)/100
429 REM - CONTA IL NUMERO DI PAGAMENTI FATTI FINO AD ORA
430 C1=C1+1
439 REM - CALCOLA IL TOTALE AMMORTIZZATO CON QUESTO PAGAMENTO
440 A=R-I1
449 REM - SOMMA IL TOTALE AMMORTIZZATO ALLA DATA
450 A1=A1+A
459 REM - CALCOLA IL BILANCIO DOVUTO
460 B=P-A1
469 REM - ULTIMO PAGAMENTO ? SE SI, CALCOLA IL TOTALE COSI' CHE
479 REM - IL BILANCIO DOVUTO UGUAGLI $.00.00 DOPO QUESTO PAGAMENTO
480 IF C1<N*N THEN 520
489 R=R+B0
490 A=A+B0

```

```

500 A1=A1+B0
510 B0=0
512 REM - SOMMA GLI INTERESSI PAGATI FINO AD ORA
520 I2=I2+I1
522 REM - SOMMA GLI INTERESSI PAGATI FINO A QUESTO ANNO
530 I3=I3+I1
532 REM - SOMMA L' AMMONTARE PAGATO FINO A QUESTO ANNO
540 A2=A2+A
542 REM - LA STAMPA E' INIZIATA? SE SI, STAMPA I VALORI COMPUTATI NELLA TABELLA
550 IF JOCK THEN 570
560 PRINT J1;TAB(10)I1;TAB(22)A;TAB(35)B0;TAB(48)I2
570 NEXT J1
572 REM - ULTIMO PAGAMENTO? SE SI', ARROTONDA, STAMPA
580 IF C1<0N*Y THEN 600
590 PRINT "          ULTIMO PAGAMENTO = $";(INT(R*100+.5))/100
595 JO=JO-1
599 REM - LA STAMPA E' INIZIATA? SE SI', STAMPA I TOTALI ANNUALI
600 IF JOCK THEN 640
610 PRINT
620 PRINT" ANNO";JO;TAB(10)I3;TAB(22)A2
630 PRINT
639 REM - TERMINE COMPLETATO?
640 IF JO=0 THEN 720
649 REM - RINIZIALIZZA LE VARIABILI ANNUALI
650 I3=0
660 A2=0
670 NEXT JO
679 REM - E' NECESSARIO STAMPARE UN ANNO AUSILIARIO ?
680 IF YC>JO THEN 262
720 PRINT
729 REM - RIPARTI O TERMINI IL PROGRAMMA?
730 PRINT "CAMBI DATI E COMPUTI ANCORA?(1=SI,0=NO)";
740 INPUT Z
750 IF Z=1 THEN 20
760 END

```

OPZIONE

È possibile introdurre il termine di pagamento in anni e mesi anzichè in anni. In tal caso il programma dovrà essere variato come viene mostrato nell'esempio successivo.

Esempio:

Avendo un debito di \$ 700 al 9% di interesse e pagando \$ 100 al mese per 8 mesi quale sarà la scheda dei rimborsi?

TABELLA DI AMMORTAMENTO DI UNA IPOTECA

```

PAGAMENTI REGOLARI? 100
TERMINE IN ANNI,MESI? 0 , 8
CAPITALE? 700
TASSO ANNUALE DI INTERESSE? 9
NUMERO DI PAGAMENTI PER ANNO? 12
PARTI STAMPANDO QUALE ANNO? 1

```

TABELLA DI AMMORTAMENTO DI UNA IPOTECA
CAPITALE \$. 700 AL 9 % PER 8 ANNI 8 MESI
PAGAMENTO REGOLARE = \$ 100

N.	INTERESSE	AMMORTAMENTO	BILANCIO	INTERESSE ACC.
1	5.25	94.75	605.25	5.25
2	4.54	95.46	509.79	9.79
3	3.82	96.18	413.61	13.61
4	3.1	96.9	316.71	16.71
5	2.38	97.62	219.09	19.09
6	1.64	98.36	120.73	20.73
7	.91	99.09	21.6399999	21.64
8	.16	21.64	0	21.8
ULTIMO PAGAMENTO = \$ 21.8				

AN. 1 21.8 700

CAMBI DATI E COMPUTI ANCORA?(1=SI,0=NO)? 0

Opzione

```

1 REM - OPZIONE 50-65.350
10 PRINT "TABELLA DI AMMORTAMENTO DI UNA IPOTECA"
:
:
40 INPUT R
50 PRINT "TERMINE IN ANNI,MESI";
60 INPUT YO,M
64 REM - CONVERTIGLI ANNI E MESI IN ANNI
65 Y=(12*YO+M)/12
70 PRINT "CAPITALE";
:
:
340 PRINT"                                TABELLA DI AMMORTAMENTO DI UNA IPOTECA"
350 PRINT "    CAPITALE $.";P;"    AL";I*100;"%    PER";YO;"ANNI";M;"MESI"
360 PRINT "    PAGAMENTO REGOLARE = $";R
:
:
760 END

```

MASSIMO COMUN DENOMINATORE

Questo programma calcola il massimo comun denominatore di due interi. Esso si basa sull'algoritmo euclideo per ricavare il M.C.D.:

- 1) entrano A, B
A = valore assoluto di A
B = valore assoluto di B
- 2) si calcola: $R = A - B \cdot (\text{intero di } (A/B))$
- 3) È: $R = 0$? Se sì, il M.C.D. è = B
 Se no, vai al punto 4
- 4) A = B
 B = R
- 5) vai al punto 2

Esempio:

Trovare il massimo comun denominatore di 50 e 18, 115 e 150.

```
MASSIMO COMUN DENOMINATORE
(DIGITA 0,0 PER TERMINARE)
DIGITA DUE NUMERI? 50 , 18
M.C.D. = 2

DIGITA DUE NUMERI? 115 , 150
M.C.D. = 5

DIGITA DUE NUMERI? 0 , 0
```

Listing del programma

```
10 PRINT "MASSIMO COMUN DENOMINATORE"
20 PRINT
30 PRINT "(DIGITA 0,0 PER TERMINARE)"
40 PRINT "DIGITA DUE NUMERI:"
50 INPUT A,B
60 IF A<0 THEN 90
70 IF B<0 THEN 90
80 GOTO 190
85 REM - CALCOLA MCD IN BASE ALL'ALGORITMO EUCLIDEO, STAMPA IL RISULTATO
90 A=ABS(A)
```

```

100 B=ABS(B)
110 R=A-B*INT(A/B)
120 IF R=0 THEN 160
130 A=B
140 B=R
150 GOTO 110
160 PRINT "M.C.D.:";B
169 REM - STAMPA UNA LINEA BIANCA PER SEPARARE LE SERIE DI DATI
170 PRINT
179 REM - IL PROGRAMMA RIPRENDE
180 GOTO 40
190 END

```

FATTORI PRIMI DI NUMERI INTERI

Questo programma indica i fattori primi di un intero. Il test non è valido per il numero intero 0.

Esempi:

Quali sono i fattori primi di —49?

Quali sono i fattori primi di 92?

FATTORI PRIMI DI NUMERI INTERI

<DIGITA 0 PER TERMINARE>

NUMERO? -49

-1
7 ↑ 2

NUMERO? 92

1
2 ↑ 2
23 ↑ 1

NUMERO? 0

Listing del programma

```
10 PRINT "FATTORI PRIMI DI NUMERI INTERI"
20 PRINT
30 PRINT "<DIGITA 0 PER TERMINARE>"
40 PRINT "NUMERO";
50 INPUT Z
59 REM - IL PROGRAMMA TERMINA?
60 IF Z=0 THEN 200
69 REM - IL SEGNO DEL NUMERO E' SEMPRE UN FATTORE
70 PRINT SGN(Z)
79 REM - UTILIZZA IL VALORE ASSOLUTO PER I CALCOLI
80 Z=ABS(Z)
88 REM - CICLO PER PROVARE TUTTI GLI INTERI (DA 2 A Z/2) COME FATTORI PRIMI
89 REM - GLI INTERI DA Z/2 A Z NON AVRANNO NUOVI FATTORI
90 FOR I=2 TO Z/2
100 S=0
110 IF Z/I<>INT(Z/I) THEN 150
120 Z=Z/I
130 S=S+1
140 GOTO 110
149 REM - TROVATO UN FATTORE PRIMO? SE SI, STAMPA
150 IF S=0 THEN 170
159 REM - STAMPA I FATTORI CON I LORO ESPONENTI; I↑S = I ALLA S-ESIMA POTENZA
160 PRINT I"↑"S
170 NEXT I
180 PRINT
189 REM - IL PROGRAMMA RIPARTE
190 GOTO 40
200 END
```

AREA DI UN POLIGONO

Questo programma calcola l'area di un poligono. Bisogna fornire le coordinate x e y di tutti i vertici. Le coordinate vanno introdotte nell'ordine dei successivi vertici.

La formula utilizzata per calcolare l'area è:

$$Area = (x_1 + x_2) \cdot (y_1 - y_2) + (x_2 + x_3) \cdot (y_2 - y_3) + \dots (x_n + x_1) \cdot (y_n - y_1) \cdot 12$$


dove: n = numero di vertici

Il massimo numero di vertici che può essere introdotto è limitato a 24. Si può variare tale massimo variando l'istruzione 30 in accordo con il seguente schema:

30 DIM X(n + 1), Y (n + 1)

Esempio:

Approssimare l'area del lago Boyer.



AREA DI UN POLIGONO

NUMERO DI VERTICI <DIGITA 0 PER TERMINARE>? 14

COORDINATE DEL VERTICE 1 ? 0 , 4
VERTICE 2 ? 1 , 7
VERTICE 3 ? 4 , 8
VERTICE 4 ? 5 , 10
VERTICE 5 ? 7 , 11
VERTICE 6 ? 9 , 10


```

    VERTICE 7 ? 12 , 9
    VERTICE 8 ? 14 , 8
    VERTICE 9 ? 13 , 4
    VERTICE 10 ? 15 , 4
    VERTICE 11 ? 15 , 1
    VERTICE 12 ? 13 , 0
    VERTICE 13 ? 5 , 1
    VERTICE 14 ? 4 , 2

```

AREA = .108

NUMERO DI VERTICI (DIGITA 0 PER TERMINARE)? 0

Listing del programma

```

10 PRINT "AREA DI UN POLIGONO"
20 PRINT
28 REM - LA MATRICE CONTENENTE LE COORDINATE DEVE ESSERE DIMENSIONATA IN BASE
29 REM - AL NUMERO DI VERTICI AUMENTATO DI UNO
30 DIM X(25),Y(25)
40 PRINT "NUMERO DI VERTICI (DIGITA 0 PER TERMINARE)";
50 INPUT N
59 REM - IL PROGRAMMA TERMINA?
60 IF N=0 THEN 230
69 REM - CICLO PER INSERIRE ORDINATAMENTE LE COORDINATE DEI VERTICI SUCCESSIVI
70 FOR I=1 TO N
80 IF I>1 THEN 110
90 PRINT "COORDINATE DEL VERTICE";I;
100 GOTO 120
110 PRINT "          VERTICE";I;
120 INPUT X(I),Y(I)
130 NEXT I
139 REM - IL PRIMO VERTICE VIENE UTILIZZATO COME ULTIMO VERTICE
140 X(N+1)=X(1)
150 Y(N+1)=Y(1)
160 A=0
169 REM - CALCOLA L'AREA. STAMPA
170 FOR I=1 TO N
180 A=A+(X(I)+X(I+1))*<math>(Y(I)-Y(I+1))</math>
190 NEXT I
200 PRINT "AREA =";ABS(A)/2
210 PRINT
219 REM - IL PROGRAMMA RIPARTE
220 GOTO 40
230 END

```

PARTI DI UN TRIANGOLO


Questo programma calcola tre parti non conosciute di un triangolo quando vengono date tre altre parti del triangolo stesso. In ogni caso, la lunghezza di un lato deve sempre essere data. Si hanno cinque possibili dati d'entrata:

- 1) Angolo, lato, angolo
- 2) Lato, angolo, lato
- 3) Angolo, angolo, lato
- 4) Lato, lato, angolo
- 5) Lato, lato, lato.

I dati vanno introdotti nell'ordine in cui appaiono nel triangolo, prendendoli in senso orario o antiorario.

Esempio:

La base di un triangolo misura 14 pollici. Gli angoli di base sono di 0,45 e 2,1 radianti. Quali sono le misure del triangolo?



PARTI DI UN TRIANGOLO

TIPI DI PROBLEMA: 1=ALA, 2=LAL, 3=AAL, 4=LLA, 5=LLL, 6=FINE PROGRAMMA

DIGITA IL TIPO DI PROBLEMA? 1

DIGITA ANGOLO, LATO, ANGOLO? .45 , 14 , 2.1

LATO 1 = 10.919

ANGOLO OPPOSTO= 0.45 RADIANTI

LATO 2 = 21.67

ANGOLO OPPOSTO= 2.1 RADIANTI

LATO 3 = 14

ANGOLO OPPOSTO= 0.592 RADIANTI

DIGITA IL TIPO DI PROBLEMA? 6

Listing del programma

```
10 PRINT "PARTI DI UN TRIANGOLO"
20 PRINT
30 DIM A(3),S(3)
31 REM - ASSEGNA UN VALORE DI PI GRECO
40 P=3.1415927
48 REM - DIGITARE UN NUMERO CORRISPONDENTE AL TIPO DI PROBLEMA IN ACCORDO CON
49 REM - LE PARTI CONOSCIUTE DEL TRIANGOLO DOVE A=ANGOLO, L=LUNGHEZZA DEL LATO
50 PRINT "TIPI DI PROBLEMA: 1=ALA,2=LAL,3=AAL,4=LLA,5=LLL,6=FINE PROGRAMMA"
60 PRINT "DIGITA IL TIPO DI PROBLEMA";
70 INPUT X
79 REM - INDIRIZZA IL PROGRAMMA AI CALCOLI OPPORTUNI
80 IF X=6 THEN 560
90 IF X=5 THEN 390
100 IF X=4 THEN 300
110 IF X=3 THEN 260
120 IF X=2 THEN 190
130 PRINT "DIGITA ANGOLO,LATO,ANGOLO";
140 INPUT A(1),S(3),A(2)
150 A(3)=P-A(1)-A(2)
160 S(1)=S(3)*SIN(A(1))/SIN(A(3))
170 S(2)=S(3)*SIN(A(2))/SIN(A(3))
180 GOTO 440
190 PRINT "DIGITA LATO,ANGOLO,LATO";
200 INPUT S(3),A(1),S(2)
210 S(1)=SQR(S(3)^2+S(2)^2-2*S(3)*S(2)*COS(A(1)))
220 A(2)=SIN(A(1))/S(1)*S(2)
225 B(2)=SQR(1-(A(2))^2)
230 A(2)=ATN(A(2)/B(2))
240 A(3)=P-A(1)-A(2)
250 GOTO 440
260 PRINT "DIGITA ANGOLO,ANGOLO,LATO";
270 INPUT A(3),A(2),S(3)
280 A(1)=P-A(2)-A(3)
290 GOTO 160
300 PRINT "DIGITA LATO,LATO,ANGOLO";
310 INPUT S(1),S(2),A(1)
320 T=S(2)*SIN(A(1))
330 IF S(1)<T THEN 520
340 S(3)=SQR(S(2)^2-T^2)
350 IF S(1)<=T THEN 380
360 Y=SQR(S(1)^2-T^2)
370 S(3)=S(3)+Y
380 GOTO 220
390 PRINT "DIGITA LATO,LATO,LATO";
400 INPUT S(1),S(2),S(3)
410 A(1)=(S(2)^2+S(3)^2-S(1)^2)/2/S(2)/S(3)
415 B(1)=SQR(1-(A(1))^2)
420 A(1)=ATN(B(1)/A(1))
430 GOTO 220
440 PRINT
449 REM - STAMPA I RISULTATI
450 FOR I=1 TO 3
459 REM - GLI ANGOLI DI UN TRIANGOLO NON POSSONO ESSERE INFERIORI A ZERO
460 IF A(I)<0 THEN 520
470 PRINT "LATO";I;"=";INT(S(I)*1000+.5)/1000
480 PRINT "ANGOLO OPPOSTO=";INT(A(I)/1000*5)/1000;"RADIANTI"
490 NEXT I
500 PRINT
510 GOTO 60
```

```

520 PRINT
530 PRINT "NON ESISTE SOLUZIONE"
540 PRINT
550 GOTO 60
560 END


```

OPZIONE


Può essere più conveniente introdurre gli angoli in gradi piuttosto che in radianti. Il programma dovrà essere variato come mostrato nell'esempio.

Esempio:

Un quadrato misura 8,76" x 8,76". Quale è la lunghezza della sua diagonale?



La scala di uno scivolo misura 10', lo scivolo 14', la distanza fra la base della scala ed il punto finale dello scivolo è 13'. Qual'è l'angolo dello scivolo?



PARTI DI UN TRIANGOLO

TIPI DI PROBLEMA: 1=ALA, 2=LAL, 3=AAL, 4=LLA, 5=LLL, 6=FINE PROGRAMMA
 DIGITA IL TIPO DI PROBLEMA? 2
 DIGITA LATO, ANGOLO, LATO? 8.76 , 90 , 8.76

LATO 1 = 12.389
 ANGOLO OPPOSTO= 90 GRADI

```

LATO 2 = 8.76
ANGOLO OPPOSTO= 45 GRADI
LATO 3 = 8.76
ANGOLO OPPOSTO= 45 GRADI

DIGITA IL TIPO DI PROBLEMA? 5
DIGITA LATO,LATO,LATO? 10 , 13 , 14

```

```

LATO 1 = 10
ANGOLO OPPOSTO= 43.279 GRADI
LATO 2 = 13
ANGOLO OPPOSTO= 63.027 GRADI
LATO 3 = 14
ANGOLO OPPOSTO= 73.694 GRADI

DIGITA IL TIPO DI PROBLEMA? 6

```

Opzione

```

1 REM - OPZIONE 44-45,145-146,205,275-276,305,480
10 PRINT "PARTI DI UN TRIANGOLO"
:
:
20 PRINT
30 DIM A(3),S(3)
31 REM - ASSEGNA UN VALORE DI PI GRECO
:
:
31 REM - ASSEGNA UN VALORE DI PI GRECO
40 P=3.1415927
:
:
140 INPUT A(1),S(3),A(2)
145 A(1)=A(1)*C
146 A(2)=A(2)*C
150 A(3)=P-A(1)-A(2)
:
:
200 INPUT S(3),A(1),S(2)
205 A(1)=A(1)*C
210 S(1)=SQR(S(3)^2+S(2)^2-2*S(3)*S(2)*COS(A(1)))
:
:
270 INPUT A(3),A(2),S(3)
275 A(3)=A(3)*C
276 A(2)=A(2)*C
280 A(1)=P-A(2)-A(3)
:
:
310 INPUT S(1),S(2),A(1)
315 A(1)=A(1)*C
320 T=S(2)*SIN(A(1))
:
:
470 PRINT "LATO";I;"=";INT(S(I)*1000+.5)/1000
480 PRINT "ANGOLO OPPOSTO=";INT(A(I)/C*1000+.5)/1000;"GRADI"
490 NEXT I


```

ANALISI DI DUE VETTORI

Questo programma calcola l'angolo fra due vettori, l'angolo fra ogni vettore e gli assi, e l'ampiezza di ogni vettore. I vettori sono dati in uno spazio tridimensionale.

Esempio:

Trovare l'angolo (θ) tra la diagonale di un cubo e la diagonale di una delle sue facce. Il cubo misura $4 \times 4 \times 4$.



ANALISI DI DUE VETTORI

VETTORE 1: X,Y,Z? 0 , 4 , 4
VETTORE 2: X,Y,Z? 4 , 4 , 4

VETTORE 1 :
AMPIEZZA: 5.65685425
ANGOLO CON L'ASSE X: 90.0000007
ANGOLO CON L'ASSE Y: 45.0000004
ANGOLO CON L'ASSE Z: 45.0000004

VETTORE 2 :
AMPIEZZA: 6.92820323
ANGOLO CON L'ASSE X: 54.7356108
ANGOLO CON L'ASSE Y: 54.7356108
ANGOLO CON L'ASSE Z: 54.7356108

ANGOLO FORMATO DAI DUE VETTORI: 35.26439

ALTRI DATI (1=SI, 0=NO)? 0

Listing del programma

```

10 PRINT "ANALISI DI DUE VETTORI"
20 PRINT
30 DIM X(2),Y(2),Z(2),M(2)
39 REM - LE RIGHE DA 40 A 70 RICHIEDONO LE COORDINATE DEI VETTORI
40 PRINT "VETTORE 1: X,Y,Z";
50 INPUT X(1),Y(1),Z(1)
60 PRINT "VETTORE 2: X,Y,Z";
70 INPUT X(2),Y(2),Z(2)
80 PRINT
89 REM - CICLO PER ANALIZZARE ENTRAMBI I VETTORI
90 FOR I=1 TO 2
99 REM - CALCOLA L'AMPIEZZA, STAMPA
100 M(I)=SQR(X(I)^2+Y(I)^2+Z(I)^2)
109 REM - IL VETTORE E' UN PUNTO? SE SI, NON SI PUO' CALCOLARE L'ANGOLO
110 IF M(I)=0 THEN 220
120 PRINT "VETTORE";I;":"
130 PRINT "AMPIEZZA:";M(I)
139 REM - FATTORE DI CONVERSIONE DA RADIANTI A GRADI
140 S=57.29578
149 REM - CALCOLA L'ANGOLO TRA IL VETTORE E L'ASSE X, STAMPA
150 J=X(I)/M(I)
153 IF J=0 THEN J=1E-9
155 W=SQR(1-J^2)
160 PRINT "ANGOLO CON L'ASSE X:";ATN(W/J)*S
169 REM - CALCOLA L'ANGOLO TRA IL VETTORE E L'ASSE Y, STAMPA
170 J=Y(I)/M(I)
173 IF J=0 THEN J=1E-9
175 W=SQR(1-J^2)
180 PRINT "ANGOLO CON L'ASSE Y:";ATN(W/J)*S
189 REM - CALCOLA L'ANGOLO TRA IL VETTORE E L'ASSE Z, STAMPA
190 J=Z(I)/M(I)
193 IF J=0 THEN J=1E-9
195 W=SQR(1-J^2)
200 PRINT "ANGOLO CON L'ASSE Z:";ATN(W/J)*S
210 PRINT
220 NEXT I
230 J=0
239 REM - SE ENTRAMBI I VETTORI SONO DEI PUNTI, NON SI PUO' CALCOLARE L'ANGOLO
240 IF M(1)=0 THEN 310
250 IF M(2)=0 THEN 310
259 REM - CALCOLA L'ANGOLO TRA I DUE VETTORI
260 J=(X(1)*X(2)+Y(1)*Y(2)+Z(1)*Z(2))/M(1)/M(2)
269 REM - I VETTORI SONO PERPENDICOLARI?
270 IF J<>0 THEN 300
280 J=90
290 GOTO 310
299 REM - CALCOLA L'ANGOLO IN GRADI, STAMPA
300 W=SQR(1-J^2);J=ATN(W/J)*S
310 PRINT "ANGOLO FORMATO DAI DUE VETTORI:";J
320 PRINT
329 REM - IL PROGRAMMA TERMINA O RIPARTE?
330 PRINT "ALTRI DATI (1=SI, 0=NO)";
340 INPUT Z
350 IF Z=1 THEN 39
360 END

```

OPERAZIONI SU DUE VETTORI

Questo programma permette quattro operazioni su due vettori dati in uno spazio tridimensionale. Le operazioni consentite sono:

- 1) Addizione
- 2) Sottrazione
- 3) Prodotto scalare
- 4) Prodotto vettoriale.

Esempio:

Due vettori vanno dall'origine ai punti A(5, — 1,2) e B(1, 4, 9). Effettuare le quattro precedenti operazioni su questi vettori.

OPERAZIONE SU DUE VETTORI

VETTORE A: COORDINATE X, Y, Z? 5 , -1 , 2

VETTORE B: COORDINATE X, Y, Z? 1 , 4 , 9

A+B= 6 , 3 , 11

A-B= 4 , -5 , -7

A.B= 19

A*B=-17 , -43 , 21

ALTRI DATI ? (1=SI,0=NO)? 0

Listing del programma

```
10 PRINT "OPERAZIONI SU DUE VETTORI"
20 PRINT
30 PRINT "VETTORE A: COORDINATE X, Y, Z";
40 INPUT X1,Y1,Z1
50 PRINT "VETTORE B: COORDINATE X, Y, Z";
60 INPUT X2,Y2,Z2
70 PRINT
78 REM - OPERA L'ADDIZIONE TRA I VETTORI, STAMPA LE COORDINATE
79 REM - DEL VETTORE RISULTANTE
80 PRINT "A+B=";X1+X2;",";Y1+Y2;",";Z1+Z2
88 REM - OPERA LA SOTTRAZIONE TRA I VETTORI, STAMPA LE COORDINATE
89 REM - DEL VETTORE RISULTANTE
90 PRINT "A-B=";X1-X2;",";Y1-Y2;",";Z1-Z2
99 REM - CALCOLA IL PRODOTTO SCALARE, STAMPA
100 PRINT "A.B=";X1*X2+Y1*Y2+Z1*Z2
109 REM - CALCOLA IL PRODOTTO VETTORIALE, STAMPA
110 PRINT "A*B=";Y1*Z2-Z1*Y2;",";Z1*X2-X1*Z2;",";X1*Y2-Y1*X2
120 PRINT
129 REM - RIPARTE O TERMINA IL PROGRAMMA ?
130 PRINT "ALTRI DATI ? (1=SI,0=NO)";
140 INPUT Z
150 IF X=1 THEN 20
160 END
```


CONVERSIONE ANGOLARE: RADIANTI IN GRADI

Questo programma permette la conversione di un angolo dato in radianti a gradi, minuti e secondi.

Esempio:

Quanti gradi, minuti, secondi corrispondono ad un angolo di 2,5 radianti? E ad uno di 118 radianti?

```
CONVERSIONE ANGOLARE: RADIANTI IN GRADI
ANGOLI IN RADIANTI<INTRODUCI 0 PER FINIRE IL PROGRAMMA>? 2.5
  GRADI = 143
  MINUTI = 14
  SECONDI = 22.01
ANGOLI IN RADIANTI? 118
  GRADI = 280
  MINUTI = 54
  SECONDI = 6.77
ANGOLI IN RADIANTI? 0
```

Listing del programma

```
10 PRINT "CONVERSIONE ANGOLARE: RADIANTI IN GRADI"
20 PRINT
30 PRINT "ANGOLI IN RADIANTI<INTRODUCI 0 PER FINIRE IL PROGRAMMA>";
40 GOTO 60
50 PRINT "ANGOLI IN RADIANTI";
60 INPUT R
69 REM - TEST PER FINIRE IL PROGRAMMA
70 IF R=0 THEN 170
79 REM - CONVERTI DA RADIANTI IN SECONDI
80 A=3600*180*R/3.1415927
89 REM - CALCOLA IL NUMERO DI GRADI INTERI
90 D=INT(A/3600)
99 REM - CALCOLA IL NUMERO DI ANGOLI GIRO INTERI
100 D1=INT(D/360)
109 REM - CALCOLA I GRADI DI OGNI ANGOLO MINORE DI 360 GRADI
110 PRINT "    GRADI =";D-360*D1
119 REM - CALCOLA I PRIMI, STAMPA
120 PRINT "    MINUTI =";INT((A-D*3600)/60)
129 REM - CALCOLA I SECONDI, ARROTONDA, STAMPA
130 S=A-D*3600-(INT((A-D*3600)/60))*60
140 PRINT "    SECONDI =";INT(100*S+.5)/100
150 PRINT
```

```
159 REM - IL PROGRAMMA RIPARTE
160 GOTO 50
170 END
```

OPZIONE

Si può fornire la risposta in gradi e decimi di grado anziché in gradi, minuti, secondi. In tal caso il programma dovrà essere variato come segue.

Esempio:

Quanti gradi vale un angolo di 2,5 radianti?

CONVERSIONE ANGOLARE: RADIANTI IN GRADI

ANGOLI IN RADIANTI<INTRODUCI 0 PER FINIRE IL PROGRAMMA>? 2.5
GRADI = 143

ANGOLI IN RADIANTI? 0

Opzione

```
1 REM - OPZIONE 110
10 PRINT "CONVERSIONE ANGOLARE: RADIANTI IN GRADI"
:
:
109 REM - CALCOLA I GRADI DI OGNI ANGOLO MINORE DI 360 GRADI
110 PRINT "    GRADI =";INT<(<D-360*D1>)*100+.5>/100
150 PRINT
:
:
170 END
```

CONVERSIONE ANGOLARE: GRADI IN RADIANTI

Il programma converte un angolo dato in gradi, minuti, secondi in radianti.

Esempio:

Un angolo misura 30 gradi, 5 minuti e 3 secondi. Quale sarà la sua misura in radianti?

Trovare la misura in radianti di due angoli di: $278^{\circ} 19' 54''$ e $721^{\circ} 0' 0''$?

```
CONVERSIONE ANGOLARE: GRADI IN RADIANTI
<PER TERMINARE IL PROGRAMMA INTRODUCI 0,0,0>
ANGOLO IN GRADI, PRIMI, SECONDI? 30 , 5 , 3
RADIANTI = .525067685

ANGOLO IN GRADI, PRIMI, SECONDI? 278 , 19 , 54
RADIANTI = 4.8578033

ANGOLO IN GRADI, PRIMI, SECONDI? 721 , 0 , 0
RADIANTI = .0174514931

ANGOLO IN GRADI, PRIMI, SECONDI? 0 , 0 , 0
```

Listing del programma

```
10 PRINT "CONVERSIONE ANGOLARE: GRADI IN RADIANTI"
20 PRINT
30 PRINT "<PER TERMINARE IL PROGRAMMA INTRODUCI 0,0,0>"
40 PRINT "ANGOLO IN GRADI, PRIMI, SECONDI";
50 INPUT D,M,S
59 REM - TEST PER TERMINARE IL PROGRAMMA
60 IF D<>0 THEN 100
70 IF M<>0 THEN 100
80 IF S<>0 THEN 100
90 GOTO 150
99 REM - CONVERTI I GRADI, PRIMI, SECONDI IN GRADI
100 A=D+M/60+S/3600
109 REM - CALCOLA IL NUMERO DEGLI ANGOLI GIRO
110 R=INT(A/360)
119 REM - CALCOLA IL NUMERO DEGLI ANGOLI MINORI DI 360 GRADI, STAMPA
120 PRINT "RADIANTI =";A*.01745329-R*6.2831853
130 PRINT
139 REM - IL PROGRAMMA RIPARTE
140 GOTO40
150 END
```

OPZIONE

Può essere più comodo introdurre l'angolo in gradi e frazione di gradi

anzichè in gradi, minuti, secondi. Il programma va allora modificato come nell'esempio.

Esempio:

Quanti radianti corrispondono ad un angolo di 33,08°? E ad uno di 90°?

CONVERSIONE ANGOLARE: GRADI IN RADIANTI

<PER TERMINARE IL PROGRAMMA INTRODUCI 0>

ANGOLO IN GRADI ? 33,08

RADIANTI = .577354833

ANGOLO IN GRADI ? 90

RADIANTI = 1.5707961

ANGOLO IN GRADI ? 0

Opzione

1 REM - OPZIONE 30-60

10 PRINT "CONVERSIONE ANGOLARE: GRADI IN RADIANTI"

20 PRINT

30 PRINT "<PER TERMINARE IL PROGRAMMA INTRODUCI 0>"

40 PRINT "ANGOLO IN GRADI ";

50 INPUT A

59 REM - TEST PER TERMINARE IL PROGRAMMA

60 IF A=0 THEN 150

⋮

109 REM - CALCOLA IL NUMERO DEGLI ANGOLI GIRO

⋮

150 END

CONVERSIONE DI COORDINATE

Questo programma converte le coordinate di un punto dato in coordinate cartesiane in coordinate polari, e viceversa.

Le formule di conversione sono le seguenti:

$$r = \sqrt{x^2 + y^2}$$

$$A = \text{arcotangente } (y/x)$$

$$x = r \cdot \cos. (A)$$

$$y = r \cdot \text{sen. } (A)$$

dove:

x	=	ascissa	} Coordinate cartesiane
y	=	ordinata	
r	=	ampiezza del raggio	} Coordinate polari
A	=	angolo (in gradi)	

Esempi:

Trovare le coordinate cartesiane del punto $(2, 30,5^\circ)$ dato in coordinate polari.

Se un punto nel sistema cartesiano ha coordinate $(7,18)$, quali sono le sue coordinate nel sistema polare?

Un punto è localizzato da $(0, -46,8)$. Come sarà localizzato in coordinate polari?

CAMBIAMENTO DI COORDINATE

```
( 1=DA CARTESIANE A POLARI)
(-1=DA POLARI A CARTESIANE)
( 0=FINE PROGRAMMA)
IN QUALE SENSO? -1
R,A? 2 , 30.5
X = 1.72 , Y = 1.02
IN QUALE SENSO? 1
X,Y? 7 , 18
R = 19.31 , A = 68.75
IN QUALE SENSO? 1
X,Y? 0 , -46.8
R = 46.8 , A = 270
IN QUALE SENSO? 0
```

Listing del programma

```
10 PRINT "CAMBIAMENTO DI COORDINATE"
20 PRINT
```

```

30 PRINT "          ( 1=DA CARTESIANE A POLARI)"
40 PRINT "          (-1=DA POLARI A CARTESIANE)"
50 PRINT "          ( 0=FINE PROGRAMMA)"
60 PRINT "IN QUALE SENSO";
70 INPUT D
79 REM - FINE PROGRAMMA
80 IF D=0 THEN 390
89 REM - INDIRIZZA IL CORSO DEL PROGRAMMA ALLA CONVERSIONE OPPORTUNA
90 IF D=-1 THEN 320
98 REM - CONVERTE LE COORDINATE CARTESIANE IN POLARI
99 REM - DIGITA LE COORDINATE CARTESIANE (ASCISSA, ORDINATA)
100 PRINT "X,Y";
110 INPUT X,Y
119 REM - IL PUNTO APPARTIENE ALL'ASSE DELLE Y ?
120 IF X=0 THEN 170
129 REM - IL PUNTO APPARTIENE ALL'ASSE DELLE X ?
139 REM - CALCOLA LE COORDINATE POLARI, ARROTONDA, STAMPA
140 PRINT "R =" ; INT(SGN(X)*SQRT(X^2+Y^2)*100+.5)/100 ; "," ;
150 PRINT "  A =" ; INT(ATN(Y/X)*180/3.1415927*100+.5)/100
160 GOTO 60
169 REM - IL PUNTO APPARTIENE ALL'ASSE DELLE Y; E' L'ORIGINE ?
170 IF Y=0 THEN 240
180 PRINT "R =" ; ABS(Y) ; "," ;
189 REM - IL PUNTO E' SOPRA O SOTTO L'ORIGINE ?
190 IF Y<0 THEN 220
200 PRINT "  A = 90"
210 GOTO 60
220 PRINT "  A = 270"
230 GOTO 60
239 REM - IL PUNTO E' L'ORIGINE
240 PRINT "R = 0,  A = 0"
250 GOTO 60
259 REM - IL PUNTO APPARTIENE ALL'ASSE DELLE X
260 PRINT "R =" ; ABS(X) ; "," ;
269 REM - IL PUNTO E' ALLA DESTRA O ALLA SINISTRA DELL'ORIGINE
270 IF X<0 THEN 300
280 PRINT "  A = 0"
290 GOTO 60
300 PRINT "  A = 180"
310 GOTO 60
318 REM - CONVERTE LE COORDINATE POLARI IN CARTESIANE
319 REM - DIGITA LE COORDINATE POLARI (AMPIEZZA DEL RAGGIO, ANGOLO)
320 PRINT "R,A";
330 INPUT R,A
339 REM - CONVERTE I GRADI IN RADIANTI
340 M=(A-INT(A/360)*360)*3.1415927/180
349 REM - CALCOLA LE COORDINATE CARTESIANE, ARROTONDA, STAMPA
350 PRINT "X =" ; INT(R*COS(M)*100+.5)/100 ; "," ;
360 PRINT "  Y =" ; INT(R*SIN(M)*100+.5)/100
370 GOTO 60
380 END

```

DISEGNO DI COORDINATE

Questo programma disegna punti su un set di assi coordinati. Bisogna fornire le coordinate x e y di tutti i punti che devono essere disegnati, il punto finale dell'asse x e quello dell'asse y, e gli incrementi fra i vari punti di ogni asse.

Il grafico non è convenzionale in quanto l'asse y è orizzontale mentre l'asse x è verticale, inoltre gli assi non si intersecano necessariamente a zero. Il punto di intersezioni degli assi viene stampato sopra il grafico.

Il limite massimo di punti che possono essere disegnati può essere variato, in più o in meno, variando l'istruzione 30 nel seguente modo:

30 DIM X(N + 1), Y(N + 1)

dove: N = massimo numero di punti da disegnare.

La lunghezza dell'asse y è limitata dalla lunghezza del sistema di uscita. Il programma controlla che la lunghezza non superi la lunghezza dello schermo che in questo caso (sistema PET) è di 40 spazi. Questo test presente nell'istruzione 90 può essere variato per tener conto di sistemi diversi. Nel caso di un sistema CBM (80 spazi) l'istruzione 90 diverrà:

90 IF B2 <= 78 THEN 120

Esempio:

La tabella seguente riporta le altezze di dieci uomini e dei loro figli. Riportate questi dati su di un grafico.

padri	65	63	67	64	68	62	70	66	68	67	69	71
figli	68	66	68	65	69	66	68	65	71	67	68	70

altezza pollici

RAPPRESENTAZIONE DI COORDINATE

ASSE DELLE X: ESTREMO SINISTRO, ESTREMO DESTRO, UNITA' DI MISURA

? 22, 33, .5

ASSE DELLE Y: ESTREMO INFERIORE, ESTREMO SUPERIORE, UNITA' DI MISURA

? 22, 33, .5

NUMERO DI PUNTI DA RAPPRESENTARE? 12

COORDINATE DEL PUNTO 1 ? 25, 28

PUNTO 2 ? 23, 26

PUNTO 3 ? 27, 28

PUNTO 4 ? 24, 25

PUNTO 5 ? 28, 29

PUNTO 6 ? 22, 26

```

PUNTO 7 ? 30 , 28
PUNTO 8 ? 26 , 25
PUNTO 9 ? 28 , 31
PUNTO 10 ? 27 , 27
PUNTO 11 ? 29 , 28
PUNTO 12 ? 31 , 30

```

INTERSEZIONE DEGLI ASSI A (22 , 22)

*****+*****Y

```

*
*      +
*
*      +
*
*          +
*
*      +
*
*          + +
*
*              +   +
*
*              +
*
*              +
*
*                  +
*
*
*
*
*
*
*

```

Listing del programma

```

10 PRINT "RAPPRESENTAZIONE DI COORDINATE"
20 PRINT
28 REM - DIMENSIONARE X() E Y() A (N+1)
29 REM - DOVE N= NUMERO DI PUNTI DA RAPPRESENTARE, LIMITE SUPERIORE
30 DIM X(100),Y(100)
39 REM - INSERIRE LE INFORMAZIONI NECESSARI AD ASSEGNARE GLI ASSI
40 PRINT "ASSE DELLE X: ESTREMO SINISTRO, ESTREMO DESTRO, UNITA' DI MISURA"
50 INPUT A1,A2,A3
50 PRINT "ASSE DELLE Y: ESTREMO INFERIORE, ESTREMO SUPERIORE, UNITA' DI MISURA"
70 INPUT B1,B2,B3
80 B2=(B2-B1)/B3
88 REM - L'ASSE DELLE Y E' TROPPO LUNGO PER LA PERIFERICA? SE SI, CAMBIA
89 REM - L'ESTREMO O AUMENTA L'UNITA' DI MISURA
90 IF B2<=70 THEN 120
100 PRINT "CAMPO Y TROPPO LARGO"
110 GOTO 60
120 PRINT "NUMERO DI PUNTI DA RAPPRESENTARE";
130 INPUT N
139 REM - NESSUN PUNTO DA RAPPRESENTARE? FINE PROGRAMMA
140 IF N=0 THEN 1070
149 REM - TROPPI PUNTI? SE SI, RIINTRODURRE IL NUMERO DI PUNTI
150 IF N<=99 THEN 180
160 PRINT "TROPPI PUNTI"
170 GOTO 120
179 REM - CICLO PER INSERIRE LE COORDINATE X,Y PER OGNI PUNTO

```



```

180 FOR I=1 TO N
190 IF I>1 THEN 220
200 PRINT "COORDINATE DEL PUNTO ";I;
210 GOTO 230
220 PRINT "          PUNTO ";I;
230 INPUT X(I),Y(I)
239 REM - ARROTONDA OGNI COORDINATA X,Y ALLA PIU' VICINA UNITA' DI MISURA
240 X(I)=INT((X(I)-A1)/A3+.5)
250 Y(I)=INT((Y(I)-B1)/B3+.5)
260 NEXT I
269 REM - CALCOLA LE COORDINATE X,Y AGGIUNTE
270 Y(N+1)=INT(B2+.5)+1
280 X(N+1)=INT((A2-A1)/A3+.5)+1
290 PRINT
299 REM - RILEVA L'INTERSEZIONE DEGLI ASSI
300 PRINT "INTERSEZIONE DEGLI ASSI A (";A1;",";B1;")"
310 PRINT
319 REM - ORDINA LE COORDINATE, RIORDINA DA X(1) A X(N) IN SENSO CRESCENTE
320 FOR J=1 TO N
330 FOR I=1 TO N-J
340 A=X(I)
350 B=Y(I)
360 C=X(I+1)
370 D=Y(I+1)
380 IF ACC THEN 430
390 X(I)=C
400 Y(I)=D
410 X(I+1)=A
420 Y(I+1)=B
430 NEXT I
440 NEXT J
449 REM - IL PROSSIMO PUNTO DA RAPPRESENTARE E' MEMORIZZATO IN T
450 T=1
459 REM - ELIMINA I PUNTI CHE NON RIENTRANO NEL SEMIASSE POSITIVO DELLE X
460 FOR P=0 TO N-1
470 IF X(P+1)>=0 THEN Q=P:P=N
480 NEXT P
485 P=Q
489 REM - CICLO PER RICHIAMARE OGNI UNITA' DI MISURA X PER LINEE DI STAMPA
490 FOR I=0 TO INT((A2-A1)/A3+.5)
500 T=T+P
509 REM - CONTA IL NUMERO DI PUNTI DA RAPPRESENTARE SU DI OGNI RIGA IN P
510 P=0
519 REM - TUTTI I PUNTI SONO RAPPRESENTATI?
520 IF T>N THEN 540
529 REM - VALORE DI X SULLA RIGA X? SE SI, VERIFICA Y
530 IF X(T)=I THEN 590
539 REM - PRIMA RIGA? SE SI, L'ASSE DELLE Y DEVE ESSERE RAPPRESENTATO
540 IF I=0 THEN 570
549 REM - RAPPRESENTAZIONE DELL'ASSE DELLE X
550 PRINT "X";
560 GOTO 1040
570 S=N+1
580 GOTO 920
590 FOR L=T TO N
599 REM - IL PROSSIMO PUNTO DA RAPPRESENTARE SI TROVA SULLA STESSA RIGA?
600 IF X(L)<=X(T) THEN P=P+1
609 REM - CONTA I PUNTI DA RAPPRESENTARE SU OGNI RIGA
620 NEXT L
629 REM - RAPPRESENTA UN PUNTO
630 IF P=1 THEN 730
638 REM - CICLO DI ORDINAMENTO DELLE COORDINATE Y CON UNA STESSA COORDINATA X
639 REM - IN SENSO CRESCENTE
640 FOR J=1 TO P
650 FOR L=1 TO P-J
660 D=Y(T+L-1)
670 B=Y(T+L)
680 IF D<B THEN 710
690 Y(T+L-1)=B
700 Y(T+L)=D

```

```

710 NEXT L
720 NEXT J
730 FOR L=0 TO P-1
740 Z=Y(T+L)
749 REM - ANALIZZA SE LA COORDINATA Y ESCE DAL PIANO
750 IF Z>=0 THEN 770
760 NEXT L
769 REM - PUNTO DA RAPPRESENTARE SULL'ASSE DELLE X?
770 IF I=0 THEN 910
779 REM - PUNTO DA RAPPRESENTARE SULL'ASSE DELLE Y?
780 IF Z=0 THEN 800
789 REM - RAPPRESENTA L'ASSE DELLE X
790 PRINT "*";
800 IF L=P-1 THEN 870
810 FOR J=L TO P-1
819 REM - ANALIZZA SE LA COORDINATA Y ESCE DAL PIANO
820 IF Z>B2 THEN 1040
829 REM - SALTA LA DUPLICAZIONE DELLE COORDINATE
830 IF Y(T+J)=Z THEN 860
839 REM - RAPPRESENTA IL PUNTO
840 PRINTAB(Z);"+";
850 Z=Y(T+J)
860 NEXT J
869 REM - ANALIZZA SE LA COORDINATA Y ESCE DAL PIANO
870 IF Z<0 THEN 1040
880 IF Z>B2 THEN 1040
889 REM - RAPPRESENTA IL PUNTO
890 PRINTAB(Z);"+";
900 GOTO 1040
910 S=T+L
919 REM - CICLO PER STABILIRE LA STAMPA DELLA PRIMA RIGA
920 FOR J=0 TO B2
929 REM - PUNTO DA RAPPRESENTARE?
930 IF Y(S)<>J THEN 1010
939 REM - RAPPRESENTA IL PUNTO
940 PRINT"+";
949 REM - SALTA LA DUPLICAZIONE DELLE COORDINATE
950 FOR K=S TO T+P-1
960 IF Y(K)=Y(S) THEN 990
970 S=K
980 GOTO 1020
990 NEXT K
1000 GOTO 1020
1009 REM - RAPPRESENTA L'ASSE DELLE Y
1010 PRINT"*";
1020 NEXT J
1029 REM - ETICHETTA L'ASSE DELLE Y
1030 PRINT "Y";
1039 REM - AVANZA LA PERIFERICA ALLA PROSSIMA RIGA
1040 PRINT
1050 NEXT I
1059 REM - ETICHETTA L'ASSE DELLE X
1060 PRINT "X";
1070 END

```

GRAFICO DI UN'EQUAZIONE POLARE

Questo programma disegna una data funzione in coordinate polari. Si possono disegnare 90 punti uno ogni quattro gradi.

Il grafico è convenzionale in quanto gli assi si intersecano allo zero e l'asse x è orizzontale mentre l'asse y è verticale. Bisogna solo specificare il valore assoluto dei punti finali.

L'incremento fra ogni punto dell'asse x e ogni punto dell'asse y è calcolato in modo che i punti degli assi risultino equidistanti dall'origine. Ciò per permettere di tracciare la funzione con la minima distorsione. L'aggiustamento di ogni incremento è necessaria a causa delle diverse dimensioni (orizzontali e verticali) dello schermo.

È necessario introdurre la funzione da rappresentare prima di far girare il programma. La funzione deve essere introdotta come una funzione di d . $f(d)$ sarà introdotta e posta uguale a F nella linea 130. Per esempio, la funzione $f(d) = 2 \cdot (1 - \cos(d))$ verrà introdotta come segue:

$$130 \text{ F} = 2 * (1 - \text{COS}(\text{D}))$$

Esempio:


Disegnare la funzione $f(d) = 2 \cdot (1 - \cos(d))$.

RAPPRESENTAZIONE DI UN'EQUAZIONE IN COORDINATE POLARI

VALORE ASSOLUTO DEGLI ESTREMI? 4

UNITA' DI MISURA DELL'ASSE X= .133333333

UNITA' DI MISURA DELL'ASSE Y= .222222222



Listing del programma

```
10 PRINT "RAPPRESENTAZIONE DI UN'EQUAZIONE IN COORDINATE POLARI"
20 PRINT
25 REM - LE DIMENSIONI DELLE MATRICI SONO SUFFICIENTI PER 90 PUNTI
29 REM - UNA COORDINATA X AGGIUNTA VIENE CALCOLATA DAL PROGRAMMA
30 DIM X(91),Y(90)
39 REM - NUMERO DI PUNTI DA CALCOLARE
40 N=90
49 REM - IL VALORE ASSOLUTO DI TUTTI GLI ESTREMI E' UGUALE
50 PRINT "VALORE ASSOLUTO DEGLI ESTREMI?";
60 INPUT Z
70 PRINT
```

```

78 REM - CALCOLA L'UNITA' DI MISURA DEGLI ASSI IN ACCORDO AL NUMERO DI
79 REM - CARATTERE PER ASSE
80 PRINT "UNITA' DI MISURA DELL'ASSE X=";Z/30
90 PRINT "UNITA' DI MISURA DELL'ASSE Y=";Z/18
100 PRINT
110 FOR I=1 TO N
119 REM - CONVERTE I GRADI IN RADIANTI
120 D=.06981317*I
130 F=2*(1-COS(D))*REM - INSERISCI QUI LA FUNZIONE (F="FUNZIONE")
138 REM - CALCOLA OGNI COORDINATA CARTESIANA, ARROTONDA ALLA PIU' VICINA UNITA'
139 REM - DI MISURA
140 X(I)=INT(((F*COS(D)/Z+1)*30)+.5)
150 Y(I)=INT(((F*SIN(D)/Z+1)*18)+.5)
160 NEXT I
169 REM - ORDINA LE COORDINATE; RIORDINA DACLOSEY(1) A Y(N) IN SENSO CRESCENTE
170 FOR J=1 TO N
180 FOR I=1 TO N-J
190 A=X(I)
200 B=Y(I)
210 IF B<Y(I+1) THEN 260
220 X(I)=X(I+1)
230 Y(I)=Y(I+1)
240 X(I+1)=A
250 Y(I+1)=B
260 NEXT I
270 NEXT J
279 REM - IL PROSSIMO PUNTO DA RAPPRESENTARE E' MEMORIZZATO IN T
280 T=1
289 REM - SALTA I PUNTI CHE ESCONO DAL SEMIASSE POSITIVO DELLE Y
290 FOR P=0 TO N-1
300 IF Y(P+1)>=0 THEN 320
310 NEXT P
319 REM - CICLO PER RICHIAMARE OGNI UNITA' DI MISURA Y PER RIGA DI STAMPA
320 FOR I=0 TO 36
330 T=T+P
339 REM - IL NUMERO DI PUNTI DA RAPPRESENTARE IN OGNI RIGA E' MEMORIZZATO IN P
340 P=0
349 REM - TUTTI I PUNTI SONO STATI RAPPRESENTATI?
350 IF T>N THEN 370
359 REM - VALORE DI Y SULLA RIGA Y?
360 IF Y(T)=I THEN 420
369 REM - STAMPA ASSE DELLE X?
370 IF I=18 THEN 400
379 REM - STAMPA ASSE DELLE Y?
380 PRINTTAB(30);"*";
390 GOTO 860
400 S=N+1
410 GOTO 740
420 FOR L=T TO N
429 REM - IL PROSSIMO PUNTO DEVE ESSERE RAPPRESENTATO SULLA STESSA RIGA
430 IF Y(L)>Y(T) THEN 450
440 P=P+1
450 NEXT L
460 IF P=1 THEN 560
469 REM - CICLO PER ORDINARE LE COORDINATE X CON UNA STESSA COORDINATA Y;
470 FOR J=1 TO P
480 FOR L=1 TO P-J
490 C=X(T+L-1)
500 A=X(T+L)
510 IF C<A THEN 540
520 X(T+L-1)=A
530 X(T+L)=C
540 NEXT L
550 NEXT J
559 REM - STAMPA L'ASSE DELLE X
560 IF I=18 THEN 730
570 L=-1
580 S=0
590 FOR K=0 TO P-1
598 REM - DEVE ESSERE RAPPRESENTATO PIU' DI UN PUNTO ALLO STESSO PUNTO

```

```

599 REM - DEL GRAFICO
600 IF X(T+K)=L THEN 690
610 L=X(T+K)
619 REM - RAPPRESENTA UN PUNTO SULL'ASSE Y?
620 IF L=30 THEN 660
629 REM - RAPPRESENTA UN PUNTO ALLA SINISTRA DELL'ASSE Y?
630 IF L<30 THEN 670
640 IF S=1 THEN 670
649 REM - STAMPA L'ASSE DELLE Y
650 PRINTTAB(30);"*";
660 S=1
669 REM - IL PUNTO NON E' CONTENUTO NEL SEMIASSE POSITIVO DELLE X
670 IF L>60 THEN 860
679 REM - RAPPRESENTA IL PUNTO
680 PRINTTAB(L);"+";
690 NEXT K
700 IF S=1 THEN 860
709 REM - STAMPA L'ASSE DELLE Y
710 PRINTTAB(30);"*";
720 GOTO 860
730 S=T
739 REM - CICLO PER STAMPARE LA RIGA DELL'ASSE DELLE X
740 FOR J=0 TO 60
750 IF X(S)<J THEN 830
759 REM - RAPPRESENTA UN PUNTO SULL'ASSE DELLE X
760 PRINT"+";
770 FOR K=S TO T+P-1
780 IF X(K)=X(S) THEN 810
790 S=K
800 GOTO 840
810 NEXT K
820 GOTO 840
829 REM - STAMPA L'ASSE DELLE X
830 PRINT "*";
840 NEXT J
849 REM - ETICHETTA L'ASSE DELLE X
850 PRINT "X";
860 PRINT
870 NEXT I
879 REM - ETICHETTA L'ASSE DELLE Y
880 PRINT TAB(30);"Y"
890 END

```

RAPPRESENTAZIONE DI FUNZIONI

Questo programma calcola e traccia fino ad un massimo di nove funzioni. Tutte le funzioni sono funzioni di x , e tutte vengono tracciate sullo stesso set di assi.

Per tracciare gli assi si devono introdurre i punti finali dell'asse x e dell'asse y . Bisogna anche fornire l'incremento che c'è fra un punto e l'altro di ogni asse.

Il grafico non è convenzionale, poichè l'asse x è verticale mentre l'asse y è orizzontale. Per leggere il grafico si devono ruotare le uscite di 90° in senso antiorario, oppure aggiustare mentalmente il grafico stesso.

Il grafico risulta anticonvenzionale anche perchè gli assi non devono intersecarsi necessariamente a zero. L'effettivo punto di intersezione viene stampato sopra il grafico stesso.

Si devono introdurre le funzioni da disegnare come istruzione prioritaria per eseguire il programma. Le linee che vanno dal 221 al 229 sono destinate a questo scopo.

Le funzioni devono essere introdotte in modo sequenziale $Y(1), Y(2) \dots Y(9)$. Per esempio volendo tracciare le funzioni: $f(x) = 2x + 1$ e $f(x) = \sqrt{x}$ si avrà:

```
221 Y(1) = 2 * X + 1
```

```
222 Y(2) = SQR (X)
```

La lunghezza dell'asse y è limitata dalla lunghezza del vostro sistema di uscita. Il programma compie un test per accertarsi che la lunghezza non sia maggiore della lunghezza dello schermo PET (40 spazi). Il test posto nella riga 140 può essere variato per tener conto di schermi di 80 spazi (schermo CBM). Per esempio, per uno schermo di 80 spazi si può prestabilire un grafico di 78 spazi variando la riga 140 nel modo seguente:

```
140 IF Y2 <= 78 THEN 170
```

Esempio:

Tracciare le funzioni $f(x) = \cos(x)$ e $f(x) = \sin(x)$.

```
221 Y(1)=COS(X)
```

```
222 Y(2)=SIN(X)
```

RAPPRESENTAZIONE DI FUNZIONI

```
NUMERO DI FUNZIONI DA RAPPRESENTARE? 2
```

```
ASSE DELLE X: ESTREMO SINISTRO, ESTREMO DESTRO, UNITA' DI MISURA? -5 , 5 , .25
```

```
ASSE DELLE Y: ESTREMO INFERIORE, ESTREMO SUPERIORE, UNITA' DI MISURA? -2 , 2 , .25
```

[illegible]

Listing del programma

76


```

129 REM - CALCOLA IL NUMERO DI SPAZI SULL'ASSE DELLE Y
130 Y2=(Y2-Y1)/Y3
138 REM - VERIFICA CHE L'ASSE DELLE Y NON SIA TROPPO LUNGO PER LA PERIFERICA
139 REM - SE LO E' DIMINUISCI IL CAMPO O AUMENTA L'UNITA' DI MISURA
140 IF Y<=70 THEN 170
150 PRINT "IL CAMPO DELL'ASSE Y E' TROPPO LARGO"
160 GOTO 110
170 PRINT
180 PRINT
189 REM - INDICA IL PUNTO IN CUI GLI ASSI SI INCONTRANO
190 PRINT "L'ASSE DELLE X INCONTRA L'ASSE DELLE Y A Y="Y1
200 PRINT "L'ASSE DELLE Y INCONTRA L'ASSE DELLE X A X="X1
210 PRINT
218 REM - ASSEGNA GLI INDICI DEL CICLO IN MODO DA LEGGERE I VALORI AD OGNI
219 REM - INCREMENTO DELL'ASSE DELLE X
220 FOR X=X1 TO X2 STEP X3
221 REM - LE FUNZIONI DA Y(1) A Y(9) DEVONO ESSERE INSERITE ALLE RIGHE 221-229
230 FOR I=1 TO N
239 REM - STABILISCE IL VALORE ARROTONDATO DI Y PER OGNI INCREMENTO DI X
240 Y(I)=INT((Y(I)-Y1)/Y3+.5)
250 NEXT I
259 REM - CICLO PER LEGGERE I VALORI DI OGNI INCREMENTO DI Y
260 FOR I=0 TO Y2
269 REM - S CONTA IL NUMERO DI VALORI AD OGNI INCREMENTO Y PER OGNI X
270 S=0
280 FOR J=1 TO N
288 REM - RAPPRESENTA UN PUNTO QUI? SE SI, MEMORIZZA IL NUMERO DELLA FUNZIONE
289 REM - IN T
290 IF Y(J)>I THEN 320
300 S=S+1
310 T=J
320 NEXT J
327 REM - VERIFICA IL NUMERO DI PUNTI DA RAPPRESENTARE IN OGNI PUNTO
328 REM - SE 0 STAMPA "+" (SOLO LA PRIMA LINEA); SE 1 STAMPA IL NUMERO DELLA
329 REM - FUNZIONE CORRISPONDENTE, SE 2 O PIU' STAMPA "*"
330 IF S>0 THEN 360
340 PRINT$(SGN(I)+10);
350 GOTO 400
360 IF S>1 THEN 390
370 PRINT$(T);
380 GOTO 400
390 PRINT"*";
400 NEXT I
409 REM - ETICHETTA GLI ASSI ALL'ESTRIMITA' DI OGNI ASSE
410 IF X>X1 THEN 430
420 PRINT "Y";
429 REM - AVANZA LA STAMPANTE ALLA PROSSIMA LINEA
430 PRINT
439 REM - STAMPA SPAZIO INVECE DI "+" DOPO LA PRIMA LINEA DI STAMPA (ASSE Y)
440 A$(1)=" "
450 NEXT X
460 PRINT "X"
470 DATA "1","2","3","4","5","6","7","8","9","+", "+"
480 END

```


INTERPOLAZIONE LINEARE

Questo programma calcola le coordinate y di punti di una data linea avendo le coordinate x. È necessario conoscere le coordinate di due punti che si trovano sulla stessa linea.

Il punto è interpolato usando la seguente formula:

$$y = y_1 + \frac{(y_2 - y_1) \cdot (x - x_1)}{(x_2 - x_1)}$$

dove: x_1, y_1 = coordinate del primo punto sulla linea
 x_2, y_2 = coordinate del secondo punto sulla linea
 x = ascissa del punto che deve essere interpolato
 y = ordinata del punto della linea corrispondente a x .

Esempio:

Una tavola di conversione assegna 60 °F a 15,56 °C e 90 °F a 32,22 °C. Calcolare i gradi Celsius corrispondenti a 73 °F e 85,6 °F.

Una nuova tassa di vendita del 17,5% è stata imposta. Quale sarà la tassa su un sofà che viene venduto a \$ 455,68?

INTERPOLAZIONE LINEARE

```
%Y DEL PRIMO PUNTO? 60 , 15.56
%Y DEL SECONDO PUNTO? 90 , 32.22
INTERPOLAZIONE: X=? 73
                Y = 22.779
```

```
ALTRI PUNTI SU QUESTA LINEA (1=SI, 0=NO)? 1
```

```
INTERPOLAZIONE: X=? 85.6
                Y = 29.777
```

```
ALTRI PUNTI SU QUESTA LINEA (1=SI, 0=NO)? 0
```

```
NUOVA LINEA (1=SI, 0=NO)? 1
```

```
%Y DEL PRIMO PUNTO? 0 , 0
%Y DEL SECONDO PUNTO? 100 , 17.5
INTERPOLAZIONE: X=? 455.68
                Y = 79.744
```

```
ALTRI PUNTI SU QUESTA LINEA (1=SI, 0=NO)? 0
```

```
NUOVA LINEA (1=SI, 0=NO)? 0
```

Listing del programma

```
10 PRINT "INTERPOLAZIONE LINEARE"
20 PRINT
29 REM - INSERIRE LE COORDINATE X ED Y DI DUE PUNTI APPARTENENTI ALLA LINEA
30 PRINT "X,Y DEL PRIMO PUNTO";
40 INPUT X1,Y1
50 PRINT "X,Y DEL SECONDO PUNTO";
60 INPUT X2,Y2
69 REM - INSERIRE LA COORDINATA X DEL PUNTO DA INTERPOLARE
70 PRINT "INTERPOLAZIONE: X =";
80 INPUT X
89 REM - CALCOLA LA CORRISPONDENTE COORDINATA Y
90  $Y=Y1+(Y2-Y1)/(X2-X1)*(X-X1)$ 
99 REM - ARROTONDA, STAMPA
100 PRINT "          Y =";INT(Y*1000+.5)/1000
110 PRINT
120 PRINT "ALTRI PUNTI SU QUESTA LINEA (1=SI, 0=NO)";
130 INPUT Z
140 PRINT
150 IF Z=1 THEN 70
159 REM - INTERPOLAZIONE DI UN'ALTRA LINEA?
160 PRINT "NUOVA LINEA (1=SI, 0=NO)";
170 INPUT Z
180 IF Z=1 THEN 20
190 END
```

INTERPOLAZIONE CURVILINEA

Questo programma calcola le coordinate y dei punti di una curva quando sono assegnate le coordinate x dei punti stessi. Bisogna introdurre le coordinate di punti conosciuti della curva, non si devono introdurre due punti aventi la stessa ascissa.

Il computo è eseguito applicando il metodo di interpolazione Lagrangiano.

Il numero di punti conosciuti della curva che può essere introdotto nel programma è limitato a 50. È possibile variare tale numero variando la linea 30 come segue:

30 DIM X(P), Y(P)

dove: P = numero di punti conosciuti della curva.

Esempi:

Consideriamo la curva: $y = x^3 - 3x + 3$. Conoscendo i punti: (-3, -15), (-2, 1), (-1, 5), (0, 3), (1, 1), (2, 5), (3, 21). Quale sarà il valore di y quando $x = -1,65$ e quando $x = 0,2$?

Dati i seguenti punti della curva del seno, quali saranno il seno di -2,47 e quello di 1,5?

(-5,.958)	(0,0)
(-4,.757)	(1,.841)
(-3,-.141)	(2,.909)
(-2,-.909)	(3,.141)
(-1,-.841)	(4,-.757)
	(5,-.959)

INTERPOLAZIONE CURVILINEA

NUMERO DI PUNTI NOTI? 7

X,Y DEL PUNTO 1 ? -3 , -15

X,Y DEL PUNTO 2 ? -2 , 1

X,Y DEL PUNTO 3 ? -1 , 5

X,Y DEL PUNTO 4 ? 0 , 3

X,Y DEL PUNTO 5 ? 1 , 1

X,Y DEL PUNTO 6 ? 2 , 5

X,Y DEL PUNTO 7 ? 3 , 21

INTERPOLAZIONE: X=?-1.65

Y= 3.457875

ALTRE COORDINATE X SU QUESTA CURVA

<1=SI, 0=NO>? 1

INTERPOLAZIONE: X=? .2

Y= 2.408

ALTRE COORDINATE X SU QUESTA CURVA

<1=SI, 0=NO>? 0

```

ALTRE COORDINATE X SU DI UN'ALTRA CURVA (1=SI, 0=NO)? 1
NUMERO DI PUNTI NOTI? 11
X,Y DEL PUNTO 1 ?-5 , .958
X,Y DEL PUNTO 2 ?-4 , .757
X,Y DEL PUNTO 3 ?-3 , -.141
X,Y DEL PUNTO 4 ?-2 , -.909
X,Y DEL PUNTO 5 ?-1 , -.841
X,Y DEL PUNTO 6 ? 0 , 0
X,Y DEL PUNTO 7 ? 1 , .841
X,Y DEL PUNTO 8 ? 2 , .909
X,Y DEL PUNTO 9 ? 3 , .141
X,Y DEL PUNTO 10 ? 4 , -.757
X,Y DEL PUNTO 11 ? 5 , -.959

```

```

INTERPOLAZIONE: X=?-2.47
Y=-.621839596

```

```

ALTRE COORDINATE X SU QUESTA CURVA (1=SI, 0=NO)? 1

```

```

INTERPOLAZIONE: X=? 1.5
Y=.9971638

```

```

ALTRE COORDINATE X SU QUESTA CURVA (1=SI, 0=NO)? 0
ALTRE COORDINATE X SU DI UN'ALTRA CURVA (1=SI, 0=NO)? 0

```

Listing del programma

```

10 PRINT "INTERPOLAZIONE CURVILINEA"
20 PRINT
28 REM - LIMITARE X() ED Y() IN BASE AL NUMERO MASSIMO DI PUNTI NOTI SU DI
29 REM - OGNI CURVA DA INSERIRE
30 DIM X(50),Y(50)
40 PRINT "NUMERO DI PUNTI NOTI";
50 INPUT P
60 FOR I=1 TO P
69 REM - INSERIRE LE COORDINATE DEI PUNTI NOTI SULLA CURVA
70 PRINT "X,Y DEL PUNTO";I;
80 INPUT X(I),Y(I)
90 NEXT I
100 PRINT
109 REM - INSERIRE LA COORDINATA X DEL PUNTO DA INTERPOLARE
110 PRINT "INTERPOLAZIONE: X=";
120 INPUT A
130 B=0
138 REM - CALCOLA LA CORRISPONDENTE COORDINATA Y COL METODO LAGRANGIANO DI
139 REM - INTERPOLAZIONE
140 FOR J=1 TO P
150 T=1
160 FOR I=1 TO P
170 IF I=J THEN 190
180 T=T*(A-X(I))/(X(J)-X(I))
190 NEXT I
200 B=B+T*Y(J)
210 NEXT J
219 REM - STAMPA I RISULTATI
220 PRINT " Y=";B
230 PRINT
239 REM - DEVI INTERPOLARE ALTRI PUNTI SULLA STESSA CURVA?
240 PRINT "ALTRE COORDINATE X SU QUESTA CURVA (1=SI, 0=NO)";
250 INPUT C
260 IF C=1 THEN 100
269 REM - IL PROGRAMMA RIPARTE O TERMINA?
270 PRINT "ALTRE COORDINATE X SU DI UN'ALTRA CURVA (1=SI, 0=NO";
280 INPUT C
290 IF C=1 THEN 40
300 END

```

INTEGRAZIONE: REGOLA DI SIMPSON

Questo programma approssima l'integrale definito di una funzione. L'integrale è calcolato usando la regola di Simpson.

Il metodo che il programma usa è opzionale; si può introdurre la funzione della curva oppure il valore della funzione in uno specifico intervallo. Per entrambi i metodi vanno introdotti i limiti di integrazione e l'incremento fra i punti contenuti nei limiti.

Se la funzione che deve essere integrata è conosciuta, essa va introdotta, prima di eseguire il programma. La funzione sarà definita alla linea 50. Per esempio, la funzione $f(x) = x^3$ verrà introdotta come segue:

50 DEFFNC (X) = x³

Esempi:

Trovare l'integrale definito della funzione $f(x) = x^3$ fra 0 e 2 con incrementi di 0,2 e 0,1.

Quale è l'integrale di una curva i cui punti conosciuti sono riportati in seguito se i limiti di integrazione sono -1 e 1?

(-1,.54)	(.25,.969)
(-.75,.73)	(.5,.878)
(-.5,.878)	(.75,.73)
(-.25,.969)	(1,.54)
(0,1)	

```
50 DEFFNC(X)=X^3
READY.
```

INTEGRAZIONE: REGOLA DI SIMPSON

```
SELEZIONE: 1=FORMULA NOTA, 0=FORMULA SCONOSCIUTA? 1
LIMITE INFERIORE, SUPERIORE DI INTEGRAZIONE? 0 , 2
UNITA' DI MISURA DI X? .2
L'INTEGRALE E' 4
```

READY.

INTEGRAZIONE: REGOLA DI SIMPSON

```
SELEZIONE: 1=FORMULA NOTA, 0=FORMULA SCONOSCIUTA? 1
LIMITE INFERIORE, SUPERIORE DI INTEGRAZIONE? 0 , 2
UNITA' DI MISURA DI X? .1
L'INTEGRALE E' 4
```

READY.

INTEGRAZIONE: REGOLA DI SIMPSON

```
SELEZIONE: 1=FORMULA NOTA, 0=FORMULA SCONOSCIUTA? 0
LIMITE INFERIORE, SUPERIORE DI INTEGRAZIONE?-1 , 1
UNITA' DI MISURA DI X? .25
```

```

VALORE INIZIALE, FINALE DI F(X)? .54 , .54
VALORE DI F(X) ALL'INTERVALLO 1 (X=-.75 )? .73
VALORE DI F(X) ALL'INTERVALLO 2 (X=-.5 )? .878
VALORE DI F(X) ALL'INTERVALLO 3 (X=-.25 )? .969
VALORE DI F(X) ALL'INTERVALLO 4 (X= 0 )? 1
VALORE DI F(X) ALL'INTERVALLO 5 (X= .25 )? .969
VALORE DI F(X) ALL'INTERVALLO 6 (X= .5 )? .878
VALORE DI F(X) ALL'INTERVALLO 7 (X= .75 )? .73
L'INTEGRALE E' 1.682

```

Listing del programma

```

10 PRINT"INTEGRAZIONE: REGOLA DI SIMPSON"
20 PRINT
30 PRINT "SELEZIONE: 1=FORMULA NOTA, 0=FORMULA SCONOSCIUTA";
40 INPUT S
49 REM - SE LA FUNZIONE E' NOTA INSERIRE ALLA RIGA 50 (DEFFNC(X)="FUNZIONE")
50 DEFFNC(X)=X^3
60 PRINT "LIMITE INFERIORE, SUPERIORE DI INTEGRAZIONE";
70 INPUT A,B
80 PRINT "UNITA' DI MISURA DI X";
90 INPUT X1
98 REM - L'UNITA' DI MISURA DEVE DIVIDERE L'INTERVALLO IN SOTTOINTERVALLI
99 REM - UGUALI ALTRIMENTI, CAMBIA UNITA' DI MISURA
100 T=(B-A)/X1:U=INT(T):IF T>U THEN 80
110 IF S=1 THEN 150
118 REM - FORMULA SCONOSCIUTA; INSERIRE IL VALORE DELLA FUNZIONE AI LIMITI DI
119 REM - INTEGRAZIONE
120 PRINT "VALORE INIZIALE, FINALE DI F(X)";
130 INPUT Y1,Y2
140 GOTO 170
149 REM - FORMULA NOTA; CALCOLA F(X) AI LIMITI DI INTEGRAZIONE
150 Y1=FNC(A)
160 Y2=FNC(B)
170 C=0
180 D=0
189 REM - CICLO PER OGNI SOTTOINTERVALLO
190 FOR I=1 TO (B-A)/X1-.5
200 IF S=1 THEN 240
209 REM - INSERIRE LA FUNZIONE NOTA AD OGNI INTERVALLO
210 PRINT "VALORE DI F(X) ALL'INTERVALLO";I;"(X=";A+I*X1;")";
220 INPUT Y
230 GOTO 250
239 REM - CALCOLA F(X) AD OGNI SOTTOINTERVALLO
240 Y=FNC(A+I*X1)
249 REM - INTERVALLI PARI O DISPARI
250 IF I=INT(I/2)*2 THEN 280
259 REM - SOMMA TUTTI I VALORI DELLA FUNZIONE DEGLI INTERVALLI DISPARI
260 C=C+Y
270 GOTO 290
279 REM - SOMMA TUTTI I VALORI DELLA FUNZIONE DEGLI INTERVALLI PARI
280 D=D+Y
290 NEXT I
299 REM - CALCOLA L'INTEGRALE, STAMPA
300 PRINT "L'INTEGRALE E'";X1/3*(Y1+4*C+2*D+Y2)
310 END

```


INTEGRAZIONE: REGOLA TRAPEZOIDALE

Questo programma approssima l'integrale definito di una funzione. L'integrale è calcolato usando la regola trapezoidale. Si devono fornire i limiti di integrazione e il numero di intervalli dentro i limiti.

La funzione da integrare va introdotta prima di eseguire il programma. La funzione di x va definita nella linea 30. Per esempio, la funzione $f(x) = x^3$ verrà introdotta come segue:

30 DEFFNC (X) = X \uparrow 3

Esempi:

Trovare l'integrale definito di $f(x) = x^3$ tra 0 e 2 usando 10 e 20 intervalli.

Trovare l'integrale definito della funzione $f(x) = x^{-2}$ tra 1 e 2, e tra 2 e 3 usando 10 sottointervalli.

```
30 DEFFNC(X)=X3
```

```
INTEGRAZIONE: REGOLA TRAPEZOIDALE
```

```
<INSERIRE 0,0 PER TERMINARE>
```

```
LIMITI DI INTEGRAZIONE <INFERIORE, SUPERIORE>? 0 , 2
```

```
NUMERO DI INTERVALLI? 10
```

```
INTEGRALE = 2.44000001
```

```
LIMITI DI INTEGRAZIONE <INFERIORE, SUPERIORE>? 0 , 2
```

```
NUMERO DI INTERVALLI? 20
```

```
INTEGRALE = 3.21000001
```

```
LIMITI DI INTEGRAZIONE <INFERIORE, SUPERIORE>? 0 , 0
```

```
30 DEFFNC(X)=1/X2
```

```
:
```

```
INTEGRAZIONE: REGOLA TRAPEZOIDALE
```

```
<INSERIRE 0,0 PER TERMINARE>
```

```
LIMITI DI INTEGRAZIONE <INFERIORE, SUPERIORE>? 1 , 2
```

```
NUMERO DI INTERVALLI? 10
```

```
INTEGRALE = .476455127
```

```
LIMITI DI INTEGRAZIONE <INFERIORE, SUPERIORE>? 2 , 3
```

```
NUMERO DI INTERVALLI? 10
```

```
INTEGRALE = .166813181
```

```
LIMITI DI INTEGRAZIONE <INFERIORE, SUPERIORE>? 0 , 0
```

Listing del programma

```
10 PRINT"INTEGRAZIONE: REGOLA TRAPEZOIDALE"
20 PRINT
30 DEF FNC(X)=1/X↑2: REM - INSERIRE QUI LA FUNZIONE DESIDERATA
40 PRINT "(INSERIRE 0.0 PER TERMINARE)"
50 PRINT "LIMITI DI INTEGRAZIONE (INFERIORE, SUPERIORE)";
60 INPUT A,B
69 REM - FINE PROGRAMMA
70 IF A=B THEN 190
80 PRINT "NUMERO DI INTERVALLI";
90 INPUT N
100 I=0
109 REM - D E' LA GRANDEZZA DI OGNI INTERVALLO
110 D=(B-A)/N
119 REM - SOMMA L'AREA DI OGNI TRAPEZIO
120 FOR J=A TO B STEP D
130 I=I+FNC(J)
140 NEXT J
149 REM - CALCOLA L'INTEGRALE
150 I=(I-(FNC(A)+FNC(B))/2)*D
160 PRINT "INTEGRALE =";I
170 PRINT
180 GOTO 50
190 END
```

INTEGRAZIONE: QUADRATURA GAUSSIANA

Questo programma approssima l'integrale definito di una funzione. Si devono fornire i limiti di integrazione e il numero di intervalli tra questi limiti.

L'intervallo di integrazione è diviso in uguali sottointervalli. L'integrale definito è computato su ogni sottointervallo usando la formula di Gauss. Gli integrali dei sottointervalli sono sommati per fornire l'integrale definito di pieno intervallo.

La funzione da integrare va introdotta prima di eseguire il programma. La funzione di x viene definita nella linea 30. Per esempio, la funzione $f(x) = x^3$ va introdotta come segue:

```
30 DEFFNC (X) = X ^ 3
```

Esempi:

Trovare l'integrale definito della funzione $f(x) = x^3$ fra 0 e 2 usando 10 e 20 sottointervalli.

Trovare l'integrale definito della funzione $f(x) = x^{-2}$ tra 1 e 2, e tra 2 e 3 usando 10 sottointervalli.

```
30 DEFFNC(X)=X^3
```

READY.

INTEGRAZIONE: QUADRATURA GAUSSIANA

LIMITI DI INTEGRAZIONE (INFERIORE, SUPERIORE)? 0 , 2

NUMERO DI INTERVALLI? 10

INTEGRALE = 4.00000004

CAMBI DATI O RICALCOLI?

(0=NO,1=NUOVI LIMITI DI INTEGRAZIONE,2=NUOVO NUMERO DI INTERVALLI)? 2

NUMERO DI INTERVALLI? 20

INTEGRALE = 4.00000004

CAMBI DATI O RICALCOLI?

(0=NO,1=NUOVI LIMITI DI INTEGRAZIONE,2=NUOVO NUMERO DI INTERVALLI)? 0

```
30 DEFFNC(X)=1/X^2
```

READY.

INTEGRAZIONE: QUADRATURA GAUSSIANA

LIMITI DI INTEGRAZIONE (INFERIORE, SUPERIORE)? 1 , 2

NUMERO DI INTERVALLI? 10

INTEGRALE = .500000002

CAMBI DATI O RICALCOLI?

(0=NO,1=NUOVI LIMITI DI INTEGRAZIONE,2=NUOVO NUMERO DI INTERVALLI)? 1

LIMITI DI INTEGRAZIONE (INFERIORE, SUPERIORE)? 2 , 3

NUMERO DI INTERVALLI? 10
INTEGRALE = .166666668

CAMBI DATI O RICALCOLI?
'0=NO,1=NUOVI LIMITI DI INTEGRAZIONE,2=NUOVO NUMERO DI INTERVALLI'? 0

Listing del programma

```
10 PRINT"INTEGRAZIONE: QUADRATURA GAUSSIANA"
20 PRINT
30 DEFNOC(X)=1/X^2
39 REM - ASCIISSE E FATTORI DI PESO PER INTEGRAZIONE GAUSSIANA A 20 PUNTI
40 DATA .076526521,.15275339,.22778585,.14917299,.37370609
50 DATA .14209611,.510867,.13168864,.63605368,.11819453
60 DATA .74633191,.10193012,.83911697,.083276742,.91223443
70 DATA .962672048,.96397193,.04060143,.9931286,.017614007
80 PRINT "LIMITI DI INTEGRAZIONE (INFERIORE, SUPERIORE)";
90 INPUT X,Y
100 PRINT "NUMERO DI INTERVALLI";
110 INPUT N
120 S=(Y-X)/N/2
130 T=X+S
140 P=0
149 REM - CALCOLA L'INTEGRALE PER CIASCUN SOTTOINTERVALLO
150 FOR I=1 TO N
160 P=0
169 REM - CALCOLA IL FATTORE DI SOMMA PER OGNI SOTTOINTERVALLO
170 FOR J=1 TO 10
180 READ A,B
190 P=P+B*(FNC(S*A+T)+FNC(T-S*A))
200 NEXT J
210 RESTORE
220 R=R+P*S
230 T=T+2*S
240 NEXT I
250 PRINT "INTEGRALE =";R
260 PRINT
270 PRINT "CAMBI DATI O RICALCOLI?"
280 PRINT "(0=NO,1=NUOVI LIMITI DI INTEGRAZIONE,2=NUOVO NUMERO DI INTERVALLI)";
290 INPUT S
300 IF S=1 THEN 80
310 IF S=2 THEN 100
320 END
```

DERIVATA

Questo programma calcola la derivata di una data funzione in un punto specifico. La funzione va introdotta prima di eseguire il programma. La funzione va introdotta nella linea 30. Per esempio, per valutare la funzione $f(x) = x^2 + \cos(x)$ essa va introdotta nel seguente modo:

30 DEFFNC (X) = X \uparrow 2 + COS (X)

Esempio:

Calcolare la derivata dell'equazione $x^2 + \cos(x) = 0$ quando $x = -1, x = 0, x = 1$.

```
30 DEFFNC(X)=X2+COS(X)
```

DERIVATA

```
(INTRODURRE X=99999 PER TERMINARE IL PROGRAMMA)
```

```
DERIVATA A X =? -1
```

```
E' -1.15852881
```

```
DERIVATA A X =? 0
```

```
E' 2.43075192E-07
```

```
DERIVATA A X =? 1
```

```
E' 1.15852833
```

```
DERIVATA A X =? 99999
```

Listing del programma

```
10 PRINT "DERIVATA"
20 PRINT
30 DEFFNC(X)=X2+COS(X)
40 PRINT "(INTRODURRE X=99999 PER TERMINARE IL PROGRAMMA)"
50 PRINT "DERIVATA A X =";
60 INPUT X1
69 REM - TEST PER TERMINARE IL PROGRAMMA
70 IF X1=99999 THEN 160
80 D=0
89 REM - CALCOLA I DIFFERENTI QUOZIENTI PER PUNTI VICINI
90 FOR N=1 TO 10
100 D1=D
110 X=X1+.51/N
120 D=(FNC(X)-FNC(X1))/(X-X1)
130 NEXT N
139 REM -APPROSSIMA LA DERIVATA DELLA FUNZIONE A X, STAMPA
140 PRINT " E'";2*D-D1
149 REM -IL PROGRAMMA RIPARTE
150 GOTO50
160 END
```

RADICI DI EQUAZIONI QUADRATICHE

Questo programma calcola le radici di una equazione quadratica. L'equazione deve essere nella seguente forma:

$$ax^2 + bx + c = 0$$

dove a, b, c sono i coefficienti reali.

La formula usata per calcolare le radici è:

$$\text{radice} = \frac{-b \pm \sqrt{b^2 - 4 \cdot a \cdot c}}{2 \cdot a}$$

Esempio:

Calcolare le radici delle seguenti equazioni:

$$2x^2 + x - 1 = 0$$

$$x^2 + 4x + 6 = 0$$

RADICI DI EQUAZIONI QUADRATICHE

COEFFICIENTI A, B, C? 2, 1, -1
RADICI (REALI): -1, .5

ALTRI DATI (1=SI, 0=NO)? 1

COEFFICIENTI A, B, C? 1, 4, 6
RADICI (COMPLESSE): -2 + 0 - 1.41421356 I

ALTRI DATI (1=SI, 0=NO)? 0

Listing del programma

```
10 PRINT "RADICI DI EQUAZIONI QUADRATICHE"
20 PRINT
29 REM - INTRODUCI I COEFFICIENTI
30 PRINT "COEFFICIENTI A, B, C";
40 INPUT A,B,C
50 S=B^2-4*A*C
60 R=SQR(ABS(S))
69 REM -RADICI COMPLESSE
70 IF S<0 THEN 100
80 PRINT "RADICI (REALI): ";(-B-R)/(2*A);"; ";(-B+R)/(2*A)
90 GOTO110
100 PRINT "RADICI (COMPLESSE): ";-B/(2*A);"+ 0 -";R/(2*A);" I"
110 PRINT
119 REM - IL PROGRAMMA RIPARTE O TERMINA
120 PRINT "ALTRI DATI (1=SI, 0=NO)";
130 INPUT X
140 IF X=1 THEN 20
150 END
```

RADICI REALI DI UN POLINOMIO: NEWTON

Questo programma calcola le radici reali di un polinomio con coefficienti reali. Si deve fornire una stima di ogni radice.

I calcoli vengono effettuati utilizzando il metodo di Newton per approssimare le radici dell'equazione. Il valore dell'errore e della derivata sono incluse per ogni radice calcolata.

Le equazioni che possono essere introdotte devono avere come grado massimo 10. Tale numero può essere variato cambiando le linee 30 e 40 in accordo con il seguente schema:

```
30 DIM A (N + 1), B(N+1)
40 FOR I = 1 TO N + 1
```

dove: N = grado dell'equazione.

Esempio:

Trovare le radici di $4x^4 - 2,5x^2 - x + 0,5$.

RADICI REALI DI UN POLINOMIO: NEWTON

```
GRADO DELL'EQUAZIONE? 4
COEFFICIENTE A( 0 )? .5
COEFFICIENTE A( 1 )? -1
COEFFICIENTE A( 2 )? -2.5
COEFFICIENTE A( 3 )? 0
COEFFICIENTE A( 4 )? 4
```

IPOTESI?-.8

RADICE	ERRORE	DERIVATA
.30357634	-2.91038305E-11	-2.070247

```
UN ALTRO VALORE (1=SI, 0=NO)? 0
UN'ALTRA FUNZIONE (1=SI, 0=NO)? 0
```

Listing del programma

```
10 PRINT "RADICI REALI DI UN POLINOMIO: NEWTON"
20 PRINT
25 REM - LIMITARE A( ) E B( ) A N+1; QUANDO CIO' E' FATTO, CICLO ALLA LINEA 40
29 REM - DEVE ESSERE INDICIZZATO PER ANALISI DA 1 A N+1
30 DIM A(11),B(11)
40 FOR I=1 TO 11
50 A(I)=0
60 B(I)=0
70 NEXT I
80 PRINT "GRADO DELL'EQUAZIONE";
90 INPUT N
100 FOR I=1 TO N+1
109 REM - INSERIRE I COEFFICIENTI ORDINATAMENTE DAL GRADO INFERIORE AL SUPERIORE
```

```

110 PRINT "COEFFICIENTE A("&I-1;&")";
120 INPUT A(I)
130 NEXT I
140 FOR I=1 TO 10
145 REM - CALCOLATE I COEFFICIENTI DELLA DERIVATA DEL POLINOMIO
150 B(I)=A(I+1)*I
160 NEXT I
170 PRINT
175 REM - INIZIALIZZA L'IPOTESI
180 PRINT "IPOTESI";
190 INPUT X
200 Q=0
210 S=1
220 F1=0
230 F0=0
235 REM - CONTA LE ITERAZIONI
240 Q=Q+1
250 FOR I=1 TO N+1
255 REM - CALCOLA IL VALORE DELLA FUNZIONE
260 F0=F0+A(I)*S
265 REM - CALCOLA IL VALORE DELLA DERIVATA
270 F1=F1+B(I)*S
280 S=S*X
290 NEXT I
295 REM - TEST PER DERIVATA ZERO; SE SI, INTERROMPI LA RICERCA, STAMPA
300 REM - OTTIENI UNA NUOVA IPOTESI USANDO L'IPOTESI PRECEDENTE
310 S=X-F0/F1
315 REM - SE LA NUOVA IPOTESI=VECCHIA IPOTESI ALLORA INTERROMPI LA RICERCA,
318 REM - STAMPA
320 IF X=S THEN 380
325 REM - MEMORIZZA L'ULTIMA IPOTESI
330 X=S
340 IF Q>100 THEN 490
350 GOTO 210
360 PRINT "DERIVATA = 0 A X ="&X
370 GOTO 180
380 PRINT
390 PRINT " RADICE ", " ERRORE ", " DERIVATA"
400 PRINT X,F0,F1
410 PRINT
415 REM - RICICLA PER OTTENERE UN'ALTRA RADICE NELLA STESSA FUNZIONE?
420 PRINT "UN ALTRO VALORE (1=SI, 0=NO)";
430 INPUT A
440 IF A=1 THEN 170
445 REM - IL PROGRAMMA TERMINA O RIPARTE?
450 PRINT "UN'ALTRA FUNZIONE (1=SI, 0=NO)";
460 INPUT A
470 IF A=1 THEN 30
480 GOTO 550
485 REM - STAMPA IL VALORE CALCOLATO DOPO 100 ITERAZIONI; ALTRE 100?
490 PRINT "100 ITERAZIONI COMPLETATE:"
500 PRINT " X ="&X;" F(X) ="&F0
510 PRINT " CONTINUA (1=SI, 0=NO)";
520 INPUT A
530 IF A=1 THEN 200
540 GOTO 420
550 END

```


RADICI DI POLINOMI: RICERCA A MEZZO INTERVALLO

Questo programma calcola radici di polinomi dentro un dato intervallo. Il programma esegue per prima cosa una ricerca casuale dentro un dato intervallo per trovare due punti di segno opposto. Se si trova un cambio di segno, la radice viene calcolata con il metodo della ricerca a mezzo intervallo. Se non si trova un cambio di segno viene richiesto un altro intervallo.

In questo, programma possono aversi degli errori per una serie di cause. Primo, una radice può essere calcolata anche quando essa non esiste. Ciò può verificarsi quando il punto più basso è così vicino a zero che la radice viene calcolata come errore di arrotondamento. Secondo, due radici possono trovarsi così vicine che il programma non trova un cambio di segno fra di esse. In tal caso nessuna radice viene calcolata.

È necessario introdurre l'equazione prima di eseguire il programma. L'equazione sarà definita come una funzione di x alla linea 30. Per esempio, se si vogliono trovare le radici della funzione $f(x) = 4x^4 - 2,5x^2 - x + 0,5$ essa andrà introdotta nel modo seguente:

$$30 \text{ DEFFNC } (X) = 4 \cdot X^4 - 2,5 \cdot X^2 - X + 0,5$$

Esempio:

Trovare una radice della funzione: $f(x) = 4x^4 - 2,5x^2 - x + 0,5$.

```
30 DEFFNR(X)=4*X^4-2.5*X^2-X+.5
```

```
RADICI DI UN POLINOMIO: RICERCA A MEZZI INTERVALLI
```

```
(PER INTERROMPERE LA RICERCA INSERIRE 0,0)  
INTERVALLO (LIMITE INFERIORE, SUPERIORE)?-1 , 0  
NESSUN CAMBIAMENTO DI SEGNO RILEVATO  
INTERVALLO (LIMITE INFERIORE, SUPERIORE)? 0 , 1  
RADICE = .303575526
```

```
INTERVALLO (LIMITE INFERIORE, SUPERIORE)? 0 , 0
```

Listing del programma

```
10 PRINT "RADICI DI UN POLINOMIO: RICERCA A MEZZI INTERVALLI"  
20 PRINT  
30 DEFFNR(X)=4*X^4-2.5*X^2-X+.5  
40 DIM D(3)  
50 PRINT "(PER INTERROMPERE LA RICERCA INSERIRE 0,0)"  
59 REM - STABILISCI GLI INTERVALLI DI RICERCA CASUALE
```

```

60 PRINT "INTERVALLO (LIMITE INFERIORE, SUPERIORE)";
70 INPUT A,B
79 REM - TEST PER UTILIZZABILITA' DEI LIMITI INSERITI
80 IF A<>B THEN 120
89 REM - FINE PROGRAMMA?
90 IF A=0 THEN 430
100 PRINT "-- I LIMITI DELL'INTERVALLO NON POSSONO ESSERE UGUALI--"
110 GOTO 60
120 IF A<B THEN 150
130 PRINT "-- IL LIMITE INFERIORE VA INSERITO PER PRIMO--"
140 GOTO 60
150 A1=SGN(FNR(A))
160 B1=SGN(FNR(B))
169 REM - TEST PER RADICI AD ENTRAMBI I LIMITI
170 IF A1*B1=0 THEN 360
179 REM - TEST PER SEGNI OPPOSTI AI LIMITI DELL'INTERVALLO
180 IF A1*B1<0 THEN 280
189 REM - CICLO DI RICERCA DI 1000 NUMERI PER SEGNI OPPOSTI NELLA FUNZIONE
190 FOR I=1 TO 1000
200 X=A+RND(2)*(B-A)
210 X1=SGN(FNR(X))
218 REM - TEST PER RADICE AL NUMERO CASUALE; SE SI, INTERROMPI LA RICERCA,
219 REM - STAMPA
220 IF X1=0 THEN 400
229 REM - TEST PER SEGNI OPPOSTI AL NUMERO CASUALE ED AL LIMITE INFERIORE
230 IF A1*X1<0 THEN 270
239 REM - PROVA UN ALTRO NUMERO CASUALE
240 NEXT I
250 PRINT "NESSUN CAMBIAMENTO DI SEGNO RILEVATO"
260 GOTO 60
269 REM - CAMBIAMENTO DI SEGNO TROVATO; CALCOLA LA RADICE
270 B=X
278 REM - MEMORIZZA IL PUNTO POSITIVO IN D(3), PUNTO NEGATIVO IN D(1)
279 REM - D(1) E D(3) DIVENGONO LIMITI DELL'INTERVALLO
280 D(2+A1)=A
290 D(2-A1)=B
299 REM - CALCOLA IL PUNTO MEDIO TRA I DUE LIMITI
300 Y=(D(1)+D(3))/2
310 Y1=SGN(FNR(Y))
319 REM - TEST PER RADICE AL PUNTO INTERMEDIO
320 IF Y1=0 THEN 400
329 REM - RICERCA UN NUOVO LIMITE PER AVVICINARTI ALLA RADICE
330 D(2+Y1)=Y
338 REM - TEST PER UN VALORE SUFFICIENTEMENTE VICINO A ZERO PER ASSUMERE UNA
339 REM RADICE
340 IF ABS(D(1)-D(3))/ABS(D(1)+ABS(D(3)))<5E-6 THEN 400
349 REM - RIPROVA CON NUOVI LIMITI
350 GOTO 300
359 REM - RADICE AD UN LIMITE DELL'INTERVALLO; TROVA QUALE LIMITE, STAMPA
360 IF A1=0 THEN 390
370 Y=B1
380 GOTO 400
390 Y=A1
400 PRINT "RADICE =" ;Y
410 PRINT
419 REM - IL PROGRAMMA RIPARTE
420 GOTO 60
430 END

```

POLINOMIO TRIGONOMETRICO

Questo programma risolve una funzione trigonometrica per un dato angolo. La funzione deve essere nella seguente forma:

$$f(x) = A_1 \sin(x) + B_1 \cos(x) + A_2 \sin(2x) + B_2 \cos(2x) \dots + A_n \sin(n \cdot x) + B_n \cos(n \cdot x)$$

dove: $n \hat{=}$ numero di coppie di coefficienti.

I coefficienti della funzione devono essere introdotti nello stato dati 30. Lo stato dati dovrà includere il numero di coppie di coefficienti (n) e i coefficienti del polinomio. Essi vanno introdotti come segue:

30 DATA n, A₁, B₁, A₂, B₂ ... A_n, B_n

Esempio:

Risolvere la seguente equazione per gli angoli 45°, 90° e 105°:

$$f(x) = \sin(x) + 2 \cdot \cos(x) - 2 \cdot \sin(2x) + \cos(2x) + 5 \cdot \sin(3x) - 3 \cdot \cos(3x)$$

30 DATA 3, 1, 2, -2, 1, 5, -3

POLINOMIO TRIGONOMETRICO

COEFFICIENTI ALLA LINEA 30!

<INSERISCI ANGOLO 99999 PER TERMINARE>

ANGOLO? 45
F(45)= 3.09558755

ANGOLO? 90
F(90)=-2.83168119

ANGOLO? 105
F(105)=-1.54684808

ANGOLO? 99999

Listing del programma

```
10 PRINT "POLINOMIO TRIGONOMETRICO"
20 PRINT
28 REM-INTRODUCI IL NUMERO DI PAIA DI TERMINI E I COEFFICIENTI
29 REM-CON L'ISTRUZIONE DATA
30 DATA 3, 1, 2, -2, 1, 5, -3
31 PRINT "COEFFICIENTI ALLA LINEA 30!":PRINT
40 PRINT "<INSERISCI ANGOLO 99999 PER TERMINARE>":PRINT
50 PRINT "ANGOLO";
60 INPUT R
```

```

69 REM-FINE PROGRAMMA?
70 IF R=99999 THEN 180
79 REM-PRENDI IL NUMERO DI PAIA DI TERMINI NEL POLINOMIO
80 READ N
89 REM-RIPETI PER PRENDERE I VALORI DEI COEFFICIENTI DALLA TABELLA DEI DATI
90 FOR I=1 TO N
100 READ A,B
109 REM-CALCOLA IL VALORE DELLA FUNZIONE PER ANGOLO X
110 Z=Z+A*SIN(I*R)+B*COS(I*R)
120 NEXT I
129 REM-STAMPA I RISULTATI
130 PRINT"F(";R;")=";Z
139 REM-PREPARATI A RILEGGERE I COEFFICIENTI DELLA FUNZIONE
140 RESTORE
150 PRINT
160 Z=0
169 REM-IL PROGRAMMA RIPARTE
170 GOTO50
180 END

```

SISTEMI DI EQUAZIONE

Il programma risolve un sistema di equazioni lineari. Il numero di coefficienti incogniti in ogni equazione deve essere uguale al numero di equazioni da risolvere. È necessario inserire i coefficienti di ogni equazione.

La dimensione dello statement alla linea 30 limita il numero delle equazioni che possono essere risolte. È possibile cambiare questo limite mediante il seguente schema:

30 DIM A(R, R + 1)

in cui R è il massimo numero di equazioni.

Esempio:

Risolvere il seguente sistema di equazioni:

$$\begin{aligned}x + 2x + 3x &= 4 \\ 3x + 6x &= 1 \\ -3x + 4x - 2x &= 0\end{aligned}$$

30 DIM A(3,4)

SISTEMI DI EQUAZIONE

NUMERO DELLE EQUAZIONI? 3

MATRICE DEI COEFFICIENTI:

EQUAZIONE 1

COEFFICIENTE 1 ? 1

COEFFICIENTE 2 ? 2

COEFFICIENTE 3 ? 3

COSTANTE? 4

EQUAZIONE 2

COEFFICIENTE 1 ? 3

COEFFICIENTE 2 ? 6

COEFFICIENTE 3 ? 0

COSTANTE? 1

EQUAZIONE 3

COEFFICIENTE 1 ? -3

COEFFICIENTE 2 ? 4

COEFFICIENTE 3 ? -2

COSTANTE? 0

X 1 = -.356

X 2 = .344

X 3 = 1.222

Listing del programma

```
10 PRINT"SISTEMI DI EQUAZIONE"
20 PRINT
25 REM-LIMITA A(> A(R+1) CON R=MASSIMO NUMERO DI EQUAZIONI
30 DIMA(3,4)
40 PRINT"NUMERO DELLE EQUAZIONI";
50 INPUT R
60 PRINT"MATRICE DEI COEFFICIENTI:"
70 FORJ=1TOR
80 PRINT
85 PRINT"EQUAZIONE";J
90 FORI=1TOR+1
100 IF I=R+1THEN130
110 PRINT"    COEFFICIENTE";I;
120 GOTO140
130 PRINT"COSTANTE";
140 INPUT A(J,I)
150 NEXTI
160 NEXTJ
170 FORJ=1TOR
175 REM-LE ISTRUZIONI DAL 180 AL 220 TROVANO LA PRIMA EQUAZIONE
179 REM-CON COEFFICIENTI DIVERSI DA 0 NELLA COLONNA CORRENTE
180 FORI=JTOR
190 IF A(I,J)≠0 THEN 230
200 NEXTI
210 PRINT"SOLUZIONE NON UNICA"
220 GOTO440
225 REM-LE ISTRUZIONI DAL 230 AL 270 SPOSTANO L'EQUAZIONE IN SU PER
229 REM-LA RIGA CORRENTE
230 FORK=1TOR+1
240 X=A(J,K)
250 A(J,K)=A(I,K)
260 A(I,K)=X
270 NEXTK
275 REM-LE ISTRUZIONI DAL 280 AL 310 GENERANO UN VALORE DI UNO NELLA PRIMA
279 REM-COLONNA DIVERSA DA 0 NELLA RIGA CORRENTE
280 Y=1/A(J,J)
290 FORK=1TOR+1
300 A(J,K)=Y*A(J,K)
310 NEXTK
320 FORI=1TOR
330 IFI=JTHEN380
340 Y=-A(I,J)
350 FORK=1TOR+1
360 A(I,K)=A(I,K)+Y*A(J,K)
370 NEXTK
380 NEXTI
385 REM-IL PROCEDIMENTO VIENE RIPETUTO PER TUTTE LE EQUAZIONI
390 NEXTJ
400 PRINT
405 REM-STAMPA LE SOLUZIONI
410 FORI=1TOR
420 PRINT"X";I;"=";INT(A(I,R+1)*1000+.5)/1000
430 NEXTI
440 END
```

PROGRAMMAZIONE LINEARE

dovuto alla cortesia di: Harold Hanes; Earlham College; Richmond, Indiana

Questo programma utilizza il metodo dei semplici per risolvere un problema di programmazione lineare. Si debbono inserire i coefficienti della funzione obiettivo e i coefficienti, le relazioni e le costanti di ciascun vincolo. Queste informazioni vengono inserite nello stato dei dati prima di far partire il programma.

Nel momento in cui si carica il programma, i dati vanno inseriti secondo le seguenti istruzioni. Desiderando far girare più di un problema, ci si deve ricordare di cancellare tutto lo stato dei dati del precedente problema, prima di partire con uno nuovo.

La specifica dei dati si presenta dalla linea 30 alla 35.

- 1) Si ordinino i vincoli del problema a seconda delle loro relazioni, in modo che le relazioni di disuguaglianza "minore di" precedano quelle di uguaglianza, le quali, a loro volta, precedano le relazioni di disuguaglianza "maggiore di".
- 2) Si inseriscano nei dati i coefficienti dei vincoli, nell'ordine in cui i vincoli stessi sono stati ordinati nello step 1. Non si devono includere i coefficienti inattivi o in eccedenza, o le variabili artificiali. Non si deve includere il coefficiente 0 per ogni variabile, se questo non compare in vincoli particolari.
- 3) Si inseriscano nei dati le costanti dei vincoli (situate in ogni equazione alla destra dell'uguale) nello stesso ordine in cui sono state inserite le righe dei coefficienti. Questi valori non possono essere negativi.
- 4) Si inseriscano nei dati i coefficienti della funzione obiettivo.

È necessario decidere se la soluzione del problema deve avere un valore massimo o minimo. Il programma richiede anche che si inseriscano il totale numero di vincoli e il numero di variabili permesse per ciascuno di essi, oltre al numero di vincoli del tipo "minore di", "uguale", "maggiore di" che si stanno considerando.

La dimensione dello stato alla linea 180 limita il numero delle variabili e delle costanti che si possono inserire. Questi limiti possono essere variati mediante il seguente schema:

180 DIM A(C + 2, V + C + G + 1), B(C + 2)

in cui C = numero di vincoli
 V = numero di variabili
 G = numero dei vincoli "maggiore di"

Esempio:

Una ditta desidera produrre 100 tonnellate di una lega composta da 83% di piombo, 14% di ferro e 3% di antimonio.

La ditta ha la possibilità di produrre cinque leghe con le seguenti composizioni e i seguenti prezzi:

lega 1	lega 2	lega 3	lega 4	lega 5
90	80	95	70	30
5	5	2	30	70
5	15	3	0	0
\$6,13	\$7,12	\$5,85	\$4,57	\$3,96

Come è possibile combinare queste leghe in modo da combinare il prodotto desiderato al minimo costo?

Si noti che questo problema è determinato dalla risoluzione del seguente sistema di equazioni:

$$\begin{array}{rclclcl} x_1 + & x_2 + & x_3 + & x_4 + & x_5 = & 100 \\ 0,90x_1 + & 0,80x_2 + & 0,95x_3 + & 0,70x_4 + & 0,30x_5 = & 83 \\ 0,05x_1 + & 0,05x_2 + & 0,02x_3 + & 0,30x_4 + & 0,70x_5 = & 14 \\ 0,05x_1 + & 0,15x_2 + & 0,03x_3 + & & & = 3 \\ 6,13x_1 + & 7,12x_2 + & 5,85x_3 + & 4,57x_4 + & 3,96x_5 = & Z \text{ (min)} \end{array}$$

PROGRAMMAZIONE LINEARE -
METODO DEI SIMPLESSI

1=MASSIMIZZA -1=MINIMIZZA?-1
DI VINCOLI, # DI VARIABILI? 4 , 5
DI < , = , > VINCOLI ? 0 , 4 , 0

LE VOSTRE VARIABILI DA 1 A 5
VARIABILI ARTIFICIALI DA 6 A 9

RISPOSTE:
VARIABILI PRIMITIVE:
VARIABILE VALORE
2 10.4347826
3 47.826087
4 41.7391304
VALORE DELLA FUNZIONE OBIETTIVO = 544.826087

Listing del programma

```
1 REM - PROGRAMMAZIONE LINEARE
20 REM - *** ESEGUIRE I SEGUENTI PASSI PRIMA DI FAR PARTIRE IL PROGRAMMA ***
21 REM - INSERIRE I COEFFICIENTI '<','=',>' DEI VINCOLI NELLE ISTRUZIONI DATA
```



```

22 REM - PARTENDO DALLA LINEA 30, UN'ISTRUZIONE DATA PER OGNI VINCOLO ;
23 REM - (LINEE 30-35 NEL NOSTRO ESEMPIO)
24 REM - INSERIRE LE COSTANTI DEI VINCOLI NELL'ISTRUZIONE DATA SUCCESSIVO;
25 REM - IL COEFFICIENTE DEI DATI, VENGONO INSERITI NELLO STESSO ORDINE
26 REM - DEI VINCOLI DATI (LINEA 34 NEL NOSTRO ESEMPIO)
27 REM - INSERIRE I COEFFICIENTI DELLA FUNZIONE OBIETTIVO NELL'ISTRUZIONE DATA
28 REM - (LINEA 35 NEL NOSTRO ESEMPIO) SECONDO LE COSTANTI DATE
29 DATA 1,1,1,1,1
30 DATA .9,.8,.95,.7,.3
31 DATA .05,.05,.02,.3,.7
32 DATA .05,.15,.03,0,0
33 DATA 100,83,14,3
34 DATA 6.13,7.12,5.85,4.57,3.96
100 PRINT "I"
170 PRINT "PROGRAMMAZIONE LINEARE - "
175 PRINT "METODO DEI SIMPLESSI"
180 DIM A(6,10),B(6)
200 PRINT
210 PRINT "1=MASSIMIZZA -1=MINIMIZZA";
220 INPUT Z
230 Z=-Z
240 PRINT "# DI VINCOLI, # DI ";
245 PRINT "VARIABILI";
250 INPUT M,N:PRINT?"M;",";N
260 PRINT "# DI <,> VINCOLI ";
270 INPUT L,E,G
280 IF M=L+E+G THEN 320
290 PRINT "DATI INCONSISTENTI- ";
295 PRINT "PROVA ANCORA"
300 GOTO 260
310 REM - QUESTA E' UNA ROUTINE DI INIZIALIZZAZIONE
320 C=N+M+G
330 C1=C+1
340 C2=N+L+G
350 M1=M+1
360 M2=M+2
380 PRINT
390 FOR I=1 TO M2
400 FOR J=1 TO C1
410 A(I,J)=0
420 NEXT J
430 NEXT I
440 FOR I=1 TO M
450 B(I)=0
460 NEXT I
470 FOR I=1 TO M
480 FOR J=1 TO N
490 READ A(I,J)
500 IF I<L THEN 520
510 A(M1,J)=A(M1,J)-A(I,J)
520 NEXT J
530 IF I>L THEN 570
540 B(I)=N+1
550 A(I,N+1)=1
560 GOTO 630
570 B(I)=N+G+1
580 A(I,N+G+1)=1
590 IF I>L+E THEN 610
600 GOTO 630
610 A(I,N+1+E)=-1
620 A(M1,N+1+E)=1
630 NEXT I
640 FOR I=1 TO M
650 READ A(I,C1)
660 NEXT I
670 FOR J=1 TO N
680 READ A(M2,J)
690 A(M2,J)=Z*A(M2,J)
700 NEXT J
710 PRINT

```

```

730 PRINT "LE VOSTRE VARIABILI ";
731 PRINT "DA";1;"A";N
740 IF L=0 THEN 760
750 PRINT "VARIABILI INATTIVE ";
751 PRINT "DA";N+1;"A";N+L
760 IF G=0 THEN 780
770 PRINT "VARIABILI IN ECCESSO ";
771 PRINT "DA";N+L+1;"A";C2
780 IF L=M THEN 970
790 PRINT "VARIABILI ARTIFICIALI ";
791 PRINT "DA";C2+1;"A";C
800 M3=M1
810 GOSUB 1240
820 PRINT
830 FOR I1=1 TO M
840 IF B(I1)<=C2 THEN 950
850 IF A(I1,C1)<=.00001 THEN 880
860 PRINT "SOLUZIONE NON REALIZZABILE"
870 GOTO 1700
880 FOR J1=1 TO C2
890 IF ABS(A(I1,J1))<=.00001 THEN 940
900 R=I1
910 S=J1
920 GOSUB 1490
930 J1=C2
940 NEXT J1
950 NEXT I1
970 PRINT
980 M3=M2
990 GOSUB 1240
1020 PRINT
1030 PRINT "RISPOSTE:"
1040 PRINT "VARIABILI PRIMITIVE:"
1050 PRINT "VARIABILE","VALORE"
1060 FOR J=1 TO C2
1070 FOR I=1 TO M
1080 IF B(I)<>J THEN 1110
1090 PRINT J,A(I,C1)
1100 I=M
1110 NEXT I
1120 NEXT J
1130 IF L=0 THEN 1190
1140 PRINT "VARIABILI DUALI:"
1150 PRINT "VARIABILE","VALORE"
1160 FOR I=1 TO L
1170 PRINT I,-Z*A(M2,N+I)
1180 NEXT I
1190 PRINT "VALORE DELLA FUNZIONE ";
1191 PRINT "OBIETTIVO = ";-Z*A(M2,C1)
1200 PRINT
1210 PRINT
1230 GOTO 1700
1240 REM - ROUTINE DI OTTIMIZZAZIONE
1241 REM - PRIMO PREZZO FUORI DALLE COLONNE
1260 P=-.00001
1270 FOR J=1 TO C2
1280 IF A(M3,J)>=P THEN 1310
1290 S=J
1300 P=A(M3,J)
1310 NEXT J
1320 IF P=-.00001 THEN 1680
1330 GOSUB 1350
1340 GOSUB 1440
1345 GOTO 1260
1350 REM - ORA SI TROVA QUALE VARIABILE LASCIA LA BASE
1360 Q=1E38
1370 FOR I=1 TO M
1380 IF A(I,S)<=.00001 THEN 1420
1390 IF A(I,C1)/A(I,S)>=Q THEN 1420
1400 R=I

```

```

1410 Q=A(I,C1)/A(I,S)
1420 NEXT I
1430 RETURN
1440 IF Q=1E38 THEN 1470
1450 GOSUB 1490
1460 RETURN
1470 PRINT "INFINITE SOLUZIONI"
1480 GOTO 1700
1490 REM - ESEGUIRE RUOTANDO
1500 P=A(R,S)
1510 FOR I=1 TO M2
1520 IF I=R THEN 1590
1530 FOR J=1 TO C1
1540 IF J=S THEN 1580
1550 A(I,J)=A(I,J)-A(I,S)*A(R,J)/P
1560 IF ABS(A(I,J))>=.00001 THEN 1580
1570 A(I,J)=0
1580 NEXT J
1590 NEXT I
1600 FOR J=1 TO C1
1610 A(R,J)=A(R,J)/P
1620 NEXT J
1630 FOR I=1 TO M2
1640 A(I,S)=0
1650 NEXT I
1660 A(R,S)=1
1670 B(R)=S
1680 RETURN
1700 END

```

ADDIZIONE, SOTTRAZIONE E MOLTIPLICAZIONE PER UNO SCALARE DI MATRICI

Questo programma aggiunge o sottrae due matrici, oppure moltiplica una matrice per un dato scalare. È possibile inserire i valori di ogni elemento per ciascuna matrice. L'addizione e la sottrazione delle due matrici sono possibili solo se le dimensioni di queste sono uguali.

La dimensione delle matrici può essere incrementata o decrementata in funzione della grandezza della memoria del sistema adoperato. Lo statement 30 può essere variato in:

30 DIM A(X,Y), B(X,Y)

in cui X e Y rappresentano i limiti delle dimensioni delle matrici.

Esempio:

Sommare le seguenti matrici e, successivamente, moltiplicare la matrice risultante per 3.

$$\begin{bmatrix} 1 & 0 & -1 \\ 5 & 8 & 0,5 \\ -1 & 2 & 0 \end{bmatrix}$$

$$\begin{bmatrix} -5 & -1 & 2 \\ 6 & -0,1 & 0 \\ 3 & 4 & -2 \end{bmatrix}$$

ADDIZIONE, SOTTRAZIONE
MOLTIPLICAZIONE SCALARE
DI MATRICI

1=ADDIZIONE
2=SOTTRAZIONE
3=MOLTIPLICAZIONE SCALARE

QUALE OPERAZIONE? 1
DIMENSIONI DELLA MATRICE (R,C):
? 3 , 3
MATRICE 1:
RIGA 1
VALORE DELLA COLONNA 1 ? 1
VALORE DELLA COLONNA 2 ? 0
VALORE DELLA COLONNA 3 ? -1
RIGA 2
VALORE DELLA COLONNA 1 ? 5
VALORE DELLA COLONNA 2 ? 8
VALORE DELLA COLONNA 3 ? .5
RIGA 3
VALORE DELLA COLONNA 1 ? -1
VALORE DELLA COLONNA 2 ? 2
VALORE DELLA COLONNA 3 ? 0
MATRICE 2:

```

RIGA 1
VALORE DELLA COLONNA 1 ?-5
VALORE DELLA COLONNA 2 ?-1
VALORE DELLA COLONNA 3 ? 2
RIGA 2
VALORE DELLA COLONNA 1 ? 6
VALORE DELLA COLONNA 2 ?-1.1
VALORE DELLA COLONNA 3 ? 0
RIGA 3
VALORE DELLA COLONNA 1 ? 3
VALORE DELLA COLONNA 2 ? 4
VALORE DELLA COLONNA 3 ?-2

```

```

-4  -1  1
11  7.9  .5
2   6  -2

```

```

ALTRI DATI (1=SI, 0=NO)? 1
1=ADDIZIONE
2=SOTTRAZIONE
3=MOLTIPLICAZIONE SCALARE

```

```

QUALE OPERAZIONE? 3
VALORE SCALARE
? 3
DIMENSIONI DELLA MATRICE (R,C)
? 3 , 3
MATRICE 1:
RIGA 1
VALORE DELLA COLONNA 1 ?-4
VALORE DELLA COLONNA 2 ?-1
VALORE DELLA COLONNA 3 ? 1
RIGA 2
VALORE DELLA COLONNA 1 ? 11
VALORE DELLA COLONNA 2 ? 7.9
VALORE DELLA COLONNA 3 ? .5
RIGA 3
VALORE DELLA COLONNA 1 ? 2
VALORE DELLA COLONNA 2 ? 6
VALORE DELLA COLONNA 3 ?-2
-12  -3  3

```

```

33  23.7  1.5
6   18  -6

```

```

ALTRI DATI (1=SI, 0=NO)
? 0

```

Listing del programma

```

20 PRINT "1=ADDIZIONE, 2=SOTTRAZIONE"
21 PRINT "3=MOLTIPLICAZIONE SCALARE"
22 PRINT "DI MATRICI"
25 PRINT
28 REM - L'ORDINE DEI VETTORI DEVE ESSERE ASSEGNATO IN BASE ALLE DIMENSIONI
29 REM - DELLE MATRICI
30 DIM A(3,3),B(3,3)
40 PRINT "1=ADDIZIONE"
50 PRINT "2=SOTTRAZIONE"
60 PRINT "3=MOLTIPLICAZIONE SCALARE"
65 PRINT

```

```

69 REM - SCEGLIERE L'OPERAZIONE INSERENDO IL NUMERO CORRISPONDENTE
70 PRINT"QUALE OPERAZIONE";
80 INPUT D
89 REM - TEST PER ADDIZIONE O SOTTRAZIONE
90 IF D<>3 THEN 120
100 PRINT"VALORE SCALARE"
110 INPUT S
120 PRINT"DIMENSIONI DELLA MATRICE (R,C)"
130 INPUT R,C
138 REM - CICLO PER INTRODURRE I VALORI DELLA MATRICE
139 REM - PER LA SOTTRAZIONE LA MATRICE 2 VIENE SOTTRATTA DALLA MATRICE 1
140 FOR K=1 TO 2
150 IF K=2 THEN 180
160 PRINT "MATRICE 1:"
170 GOTO 190
180 PRINT "MATRICE 2:"
190 FOR J=1 TO R
200 PRINT "RIGA";J
210 FOR I=1 TO C
220 PRINT "VALORE DELLA COLONNA";I;
230 IF K=2 THEN 260
240 INPUT A(J,I)
250 GOTO 270
260 INPUT B(J,I)
270 NEXT I
280 NEXT J
289 REM - UNA SOLA MATRICE VIENE USATA PER LA MOLTIPLICAZIONE SCALARE
290 IF D=3 THEN 310
300 NEXT K
305 PRINT
308 REM - LE ISTRUZIONI DALLA RIGA 310 ALLA 410 SVOLGONO LE OPERAZIONI
309 REM - RICHIESTE E STAMPANO LA MATRICE RISULTANTE
310 FOR J=1 TO R
320 FOR I=1 TO C
330 IF D<>2 THEN 350
340 B(J,I)=-B(J,I)
350 IF D=3 THEN 380
360 PRINT A(J,I)+B(J,I);" ";
370 GOTO 390
380 PRINT A(J,I)*S;" ";
390 NEXT I
399 REM - AVANZARE PER STAMPARE LA RIGA SUCCESSIVA
400 PRINT:PRINT
410 NEXT J
420 PRINT
428 REM - IL PROGRAMMA RI-INIZIA O TERMINA? VIENE RICHIESTO L'INTERVENTO
429 REM - DELL'UTENTE
430 PRINT "ALTRI DATI (1=SI, 0=NO)";
440 INPUT D
450 IF D=1 THEN 40
460 END

```

MOLTIPLICAZIONE TRA MATRICI

Questo programma moltiplica due matrici. Per l'esattezza, è la prima matrice che moltiplica la seconda. È necessario inserire gli elementi di ciascuna matrice.

Per quanto riguarda questa operazione, è necessario che il numero di righe della prima matrice sia uguale al numero di righe della seconda.

Nel caso specifico, la dimensione delle matrici è limitata a 20x20. Questo limite può essere incrementato, o decrementato, variando la linea 30 secondo il seguente schema:

30 DIM A(X,Y), B(Z,X)

dove: (x,y) = dimensioni della matrice 1
(z,x) = dimensioni della matrice 2

Esempio:

Moltiplicare la matrice 1 per la matrice 2.

$$\begin{array}{l} 1 \\ 2 \end{array} \left\{ \begin{array}{ccccc} 2 & -1 & 4 & 1 & 2 \\ 1 & 0 & 1 & 2 & -1 \\ 2 & 3 & -1 & 0 & -2 \end{array} \right.$$
$$\left\{ \begin{array}{ccc} -2 & -1 & 2 \\ 0 & 2 & 1 \\ -1 & 1 & 4 \\ 3 & 0 & -1 \\ 2 & 1 & 2 \end{array} \right.$$

```
30 DIM A(3,5),B(5,3)
READY.
```

```
DIMENSIONI DELLA MATRICE 1 (R,C)? 3 , 5
DIMENSIONI DELLA MATRICE 2 (R,C)? 5 , 3
MATRICE 1:
RIGA 1
VALORE DELLA COLONNA 1 ? 2
VALORE DELLA COLONNA 2 ? -1
VALORE DELLA COLONNA 3 ? 4
VALORE DELLA COLONNA 4 ? 1
VALORE DELLA COLONNA 5 ? 2
RIGA 2
VALORE DELLA COLONNA 1 ? 1
VALORE DELLA COLONNA 2 ? 0
VALORE DELLA COLONNA 3 ? 1
VALORE DELLA COLONNA 4 ? 2
VALORE DELLA COLONNA 5 ? -1
RIGA 3
VALORE DELLA COLONNA 1 ? 2
```

```

VALORE DELLA COLONNA 2 ? 3
VALORE DELLA COLONNA 3 ?-1
VALORE DELLA COLONNA 4 ? 0
VALORE DELLA COLONNA 5 ?-2

```

MATRICE 2:

```

RIGA 1
VALORE DELLA COLONNA 1 ?-2
VALORE DELLA COLONNA 2 ?-1
VALORE DELLA COLONNA 3 ? 2
RIGA 2
VALORE DELLA COLONNA 1 ? 0
VALORE DELLA COLONNA 2 ? 2
VALORE DELLA COLONNA 3 ? 1
RIGA 3
VALORE DELLA COLONNA 1 ?-1
VALORE DELLA COLONNA 2 ? 1
VALORE DELLA COLONNA 3 ? 4
RIGA 4
VALORE DELLA COLONNA 1 ? 3
VALORE DELLA COLONNA 2 ? 0
VALORE DELLA COLONNA 3 ?-1
RIGA 5
VALORE DELLA COLONNA 1 ? 2
VALORE DELLA COLONNA 2 ? 1
VALORE DELLA COLONNA 3 ? 2

```

```

-1   2   22

```

```

1   -1   2

```

```

-7   1   -1

```

Listing del programma

```

10 PRINT "MOLTIPLICAZIONE DI MATRICI"
20 PRINT
28 REM - LE DIMENSIONI DELLE MATRICI A E B DEVONO ESSERE ASSEGNATE IN BASE
29 REM - ALLE DIMENSIONI DELLE MATRICI
30 DIM A(20,20),B(20,20)
40 PRINT "DIMENSIONI DELLA MATRICE 1 (R,C)";
50 INPUT R1,C1
60 PRINT "DIMENSIONI DELLA MATRICE 2 (R,C)";
70 INPUT R2,C2
78 REM - IL NUMERO DELLE COLONNE DELLA MATRICE 1 DEVE ESSERE UGUALE AL NUMERO
79 REM - DI RIGHE DELLA MATRICE 2
80 IF C1=R2 THEN 110
90 PRINT "NON POSSONO ESSERE MOLTIPLICATE"
100 GOTO 40
109 REM - INSERIRE I VALORI DELLE MATRICI
110 PRINT "MATRICE 1:"
120 FOR J=1 TO R1
130 PRINT "RIGA";J
140 FOR I=1 TO C1
150 PRINT "VALORE DELLA COLONNA";I;
160 INPUT A(J,I)
170 NEXT I
180 NEXT J
190 PRINT
200 PRINT "MATRICE 2:"
210 FOR J=1 TO R2
220 PRINT "RIGA";J
230 FOR I=1 TO C2
240 PRINT "VALORE DELLA COLONNA";I;
250 INPUT B(J,I)
260 NEXT I
270 NEXT J
280 PRINT

```



```

289 REM - MOLTIPLICARE LE MATRICI E STAMPARE LA MATRICE RISULTANTE
290 FOR I=1 TO R1
300 FOR J=1 TO C2
310 S=0
320 FOR K=1 TO C1
330 S=S+A(I,K)*B(K,J)
340 NEXT K
350 PRINT S;" ";
360 NEXT J
369 REM - AVANZARE PER STAMPARE LA PROSSIMA RIGA
370 PRINT:PRINT
380 NEXT I
390 END

```

INVERSIONE DI MATRICI

Questo programma inverte una matrice quadrata. Questa inversione è compiuta mediante una modifica del metodo per eliminazione di Gauss-Jordan.

Nel caso specifico, la dimensione delle matrici è limitata a 20 x 20. Il limite può essere incrementato o decrementato variando la linea 30 secondo il seguente schema:

30 DIM A(R,R), B(R,R)

dove R = numero di righe (o di colonne) della matrice.

Esempio:

Invertire la matrice A.

$$A = \begin{Bmatrix} 3 & 5 & -1 & -4 \\ 1 & 4 & -0,7 & -3 \\ 0 & -2 & 0 & -1 \\ -2 & 6 & 0 & 0,3 \end{Bmatrix}$$

INVERSIONE DI MATRICI

DIMENSIONE DELLA MATRICE? 4

ELEMENTI DELLA MATRICE:

RIGA 1

VALORE DI COLONNA 1 ? 3

VALORE DI COLONNA 2 ? 5

VALORE DI COLONNA 3 ? -1

VALORE DI COLONNA 4 ? -4

RIGA 2

VALORE DI COLONNA 1 ? 1

VALORE DI COLONNA 2 ? 4

VALORE DI COLONNA 3 ? -0,7

VALORE DI COLONNA 4 ? -3

RIGA 3

VALORE DI COLONNA 1 ? 0

VALORE DI COLONNA 2 ? -2

VALORE DI COLONNA 3 ? 0

VALORE DI COLONNA 4 ? 1

RIGA 4

VALORE DI COLONNA 1 ? -2

VALORE DI COLONNA 2 ? 6

VALORE DI COLONNA 3 ? 0

VALORE DI COLONNA 4 ? 0,3

.654	-.935	-.191	.014
.198	-.283	-.103	.156
.368	-1.955	-4.263	-.425
.397	-.567	.793	.312

Listing del programma

```
10 PRINT "INVERSIONE DI MATRICI"
20 PRINT
25 REM - A() E B() DEVONO ENTRAMBE ESSERE DIMENSIONATE IN BASE ALLA MATRICE
30 DIM A(20,20), B(20,20)
35 REM - LA MATRICE E' QUADRATA QUINDI E' RICHIESTA UNA SOLA DIMENSIONE
40 PRINT "DIMENSIONE DELLA MATRICE";
50 INPUT R
60 PRINT "ELEMENTI DELLA MATRICE:"
65 REM - INSERIRE GLI ELEMENTI DELLA MATRICE
70 FOR J=1 TO R
80 PRINT "RIGA"/J
90 FOR I=1 TO R
100 PRINT "VALORE DI COLONNA"/I;
110 INPUT A(J,I)
120 NEXT I
130 B(J,J)=1
140 NEXT J
145 REM - LE ISTRUZIONI DALLA RIGA 150 ALLA 420 INVERTONO LA MATRICE
150 FOR J=1 TO R
160 FOR I=J TO R
170 IF A(I,J) <> 0 THEN 210
180 NEXT I
190 PRINT "MATRICE SINGOLARE"
200 GOTO 500
210 FOR K=1 TO R
220 S=A(J,K)
230 A(J,K)=A(I,K)
240 A(I,K)=S
250 S=B(J,K)
260 B(J,K)=B(I,K)
270 B(I,K)=S
280 NEXT K
290 T=1/A(J,J)
300 FOR K=1 TO R
310 A(J,K)=T*A(J,K)
320 B(J,K)=T*B(J,K)
330 NEXT K
340 FOR L=1 TO R
350 IF L=J THEN 410
360 T=-A(L,J)
370 FOR K=1 TO R
380 A(L,K)=A(L,K)+T*A(J,K)
390 B(L,K)=B(L,K)+T*B(J,K)
400 NEXT K
410 NEXT L
420 NEXT J
430 PRINT
435 REM - STAMPA LA MATRICE RISULTANTE
440 FOR I=1 TO R
450 FOR J=1 TO R
455 REM - ARROTONDA, STAMPA
460 PRINT INT(B(I,J)*1000+.5)/1000,
461 PRINT " ";
470 NEXT J
475 REM - AVANZARE PER STAMPARE LA RIGA SEGUENTE
480 PRINT:PRINT
490 NEXT I
500 END
```


PERMUTAZIONI E COMBINAZIONI

Questo programma calcola il numero di permutazioni e di combinazioni di N oggetti, presi a D alla volta.

Esempi:

Quante combinazioni e permutazioni possono essere ottenute con le 26 lettere dell'alfabeto, prendendone cinque alla volta?

In quanti modi differenti possono sedersi dodici persone in un sedile, se c'è posto solo per due?

PERMUTAZIONI E COMBINAZIONI

<INSERISCI 0 PER TERMINARE IL PROGRAMMA>

NUMERO TOTALE DI OGGETTI? 26

DIMENSIONE DEL SOTTOGRUPPO? 5

7893600 PERMUTAZIONI

65780 COMBINAZIONI

NUMERO TOTALE DI OGGETTI? 12

DIMENSIONE DEL SOTTOGRUPPO? 2

132 PERMUTAZIONI

66 COMBINAZIONI

NUMERO TOTALE DI OGGETTI? 0

Listing del programma

```
20 PRINT "PERMUTAZIONI E COMBINAZIONI"
25 PRINT
30 PRINT "<INSERISCI 0 PER TERMINARE IL PROGRAMMA>"
40 PRINT "NUMERO TOTALE DI OGGETTI";
50 INPUT N
59 REM - TEST PER FINE PROGRAMMA
60 IF N=0 THEN 280
70 PRINT "DIMENSIONE DEL SOTTOGRUPPO";
80 INPUT D
88 REM - LA DIMENSIONE DEL SOTTOGRUPPO NON PUO' ESSERE MAGGIORE DI QUELLA DEL
89 REM - SOTTOGRUPPO
90 IF D<=N THEN 130
100 PRINT "SOTTOGRUPPO TROPPO GRANDE"
110 PRINT
120 GOTO 40
129 REM - LE RIGHE DA 130 A 200 CALCOLANO LE PERMUTAZIONI
130 P=1
140 C=1
150 FOR I=N-D+1 TO N
158 REM - NON CONSENTE ALLA DIMENSIONE DEL NUMERO DI SUPERARE LE CAPACITA'
159 REM - DELLA MACCHINA
160 IF 1.7E38/I>=P THEN 190
170 PRINT"> 1.7E38 PERMUTAZIONI"
180 GOTO 280
```

```
190 P=P*I
200 NEXT I
209 REM - CALCOLA IL FATTORIALE INTERMEDIO PER LE COMBINAZIONI
210 FOR J=2 TO D
220 C=C*J
230 NEXT J
240 PRINT P; "PERMUTAZIONI"
250 PRINT P/C; "COMBINAZIONI"
260 PRINT
269 REM - IL PROGRAMMA RIPART
270 GOTO 40
280 END
```

IL TEST U DI MANN-WHITNEY

Questo programma utilizza il criterio U di Mann-Whitney su un campione preso da due popolazioni.

Le dimensioni dello statement sulla linea 30 delimita la grandezza dei campioni. È possibile incrementare o decrementare questi limiti, secondo il seguente schema:

30 DIM X(M), Y(N)

dove: M = massima grandezza del primo campione
N = massima grandezza del secondo campione.

Esempio:

Ad un gruppo di dieci donne e ad uno di dieci uomini viene chiesto di valutare il gusto di un piatto freddo, mediante una votazione da uno a dieci. La tabella sottostante elenca i punteggi ottenuti. Calcolare il numero delle volte in cui il punteggio delle donne è inferiore al punteggio degli uomini e viceversa.

donne
uomini

1	3	4	3	6	8	9	7	8	4
7	9	8	5	10	9	10	6	5	2

```
30 DIM X(10),Y(10)
READY.
```

```
TEST U DI MANN-WHITNEY
```

```
CAMPIONE 1 :
GRANDEZZA? 10
DATI 1 ? 1
DATI 2 ? 3
DATI 3 ? 4
DATI 4 ? 3
DATI 5 ? 6
DATI 6 ? 8
DATI 7 ? 9
DATI 8 ? 7
DATI 9 ? 8
DATI 10 ? 4
```

```
CAMPIONE 2 :
GRANDEZZA? 10
DATI 1 ? 7
DATI 2 ? 9
DATI 3 ? 8
DATI 4 ? 5
DATI 5 ? 10
DATI 6 ? 9
DATI 7 ? 10
```

DATI 3 ? 6
DATI 9 ? 5
DATI 10 ? 2

PRIMO PRECEDENTE, U = 71.5
SECONDO PRECEDENTE, U = 28.5

Listing del programma

```
10 PRINT"TEST U DI MANN-WHITNEY"  
20 PRINT  
27 REM-ASSEGNA LA GRANDEZZA MASSIMA DEL CAMPIONE A X(M),Y(N) (IN CUI M E' LA  
28 REM-GRANDEZZA MASSIMA DEL CAMPIONE 1,N E' LA GRANDEZZA MASSIMA  
29 REM-DEL CAMPIONE 2)  
30 DIM X(25),Y(25)  
40 DIM N(2)  
49 REM-SI INSERISCONO I DUE CAMPIONI  
50 FOR I=1 TO 2  
60 PRINT"CAMPIONE";I;" "  
70 PRINT"GRANDEZZA";  
80 INPUT N(I)  
90 FOR J=1 TO N(I)  
100 PRINT"  DATI";J;  
110 INPUT Y(J)  
120 NEXT J  
129 REM-RAGGRUPPARE OGNI CAMPIONE  
130 FOR J=1 TO N(1)  
140 FOR K=1 TO N(1)-J  
150 C=Y(K)  
170 IF Y(K)<Y(K+1) THEN 200  
180 Y(K)=Y(K+1)  
190 Y(K+1)=C  
200 NEXT K  
210 NEXT J  
220 PRINT  
229 REM-TRASFERIRE IL PRIMO CAMPIONE ALLA MATRICE X  
230 IF I=2 THEN 270  
240 FOR J=1 TO N(1)  
250 X(J)=Y(J)  
260 NEXT J  
270 NEXT I  
279 REM-SOMMA LE FILE  
280 R=1  
290 I=0  
300 J=0  
310 I=I+1  
320 J=J+1  
330 IF I>N(1) THEN 580  
340 IF J>N(2) THEN 620  
350 IF X(I)<Y(J) THEN 620  
360 IF Y(J)<X(I) THEN 590  
368 REM-LE LINEE DAL 370 AL 570 MANEGGIANO I PUNTEGGI UGUALI DA ENTRAMBI  
369 REM-I CAMPIONI  
370 K=2  
380 M=I  
390 L=J  
400 R1=2*R+1  
410 R=R+2  
420 I=I+1  
430 J=J+1  
440 IF I>N(1) THEN 480  
450 IF X(I)<X(I-1) THEN 480  
460 I=I+1  
470 GOTO 510  
480 IF J>N(2) THEN 550  
490 IF Y(J)<Y(J-1) THEN 550
```



```

500 J=J+1
510 R1=R1+R
520 R=R+1
530 K=K+1
540 GOTO 440
550 X=X+(I-M)*R1/K
560 Y=Y+(J-L)*R1/K
570 GOTO 330
580 IF J>N(2) THEN 660
590 Y=Y+R
600 J=J+1
610 GOTO 640
620 X=X+R
630 I=I+1
640 R=R+1
650 GOTO 330
658 REM-U1=NUMERO DI VOLTE IN CUI I PUNTEGGI DEL CAMPIONE 1 PRECEDONO I
659 REM-PUNTEGGI DEL CAMPIONE 2
660  $U1 = N(1) * N(2) + N(1) * (N(1) + 1) / 2 - X$ 
668 REM-U2=NUMERO DI VOLTE IN CUI I PUNTEGGI DEL CAMPIONE 2 PRECEDONO I
669 REM-PUNTEGGI DEL CAMPIONE 1
670  $U2 = N(1) * N(2) + N(2) * (N(2) + 1) / 2 - Y$ 
680 PRINT
690 PRINT"PRIMO PRECEDENTE, U =";U1
700 PRINT"SECONDO PRECEDENTE; U =";U2
710 END

```

MEDIA, VARIANZA, DEVIAZIONE STANDARD

Questo programma calcola la media aritmetica, la varianza e la deviazione standard di un certo numero di dati, raggruppati o meno. I dati possono rappresentare una intera popolazione o semplicemente un campione.

Esempi:

Alla ricezione di un hotel vi sono dieci persone, la cui età è di 87, 53, 35, 42, 9, 48, 51, 60, 39 e 44 anni. Quale sarà la media, la varianza e la deviazione standard dell'età di tutti i clienti dell'hotel, considerando come campione i clienti alla ricezione?

Trovare la media, la varianza e la deviazione standard dell'età di una crema di formaggio posta in un ripiano di un negozio. La tabella seguente indica la distribuzione delle età di 50 confezioni. Si ponga che la tabella mostri l'intero inventario della quantità. Cosa si avrà se questa rappresenta solo un campione dell'intero inventario?

età	1	2	3	4	5	6
quantità	15	10	9	6	7	3

crema di formaggio

MEDIA, VARIANZA, DEVIAZIONE STANDARD

```
METODO (0=POPOLAZIONE, 1=CAMPIONE)? 1
DATI (0=RAGGRUPPATI, 1=NON RAGGRUPPATI)? 1
NUMERO DI OSSERVAZIONI? 10
VOCE 1 ? 87
VOCE 2 ? 53
VOCE 3 ? 35
VOCE 4 ? 42
VOCE 5 ? 9
VOCE 6 ? 48
VOCE 7 ? 51
VOCE 8 ? 60
VOCE 9 ? 39
VOCE 10 ? 44
MEDIA          VARIANZA
46.8           389.733332
DEVIAZIONE STANDARD
19.7416649
ALTRI DATI (1=SI, 0=NO)? 1
```

```
METODO (0=POPOLAZIONE, 1=CAMPIONE)? 0
DATI (0=RAGGRUPPATI, 1=NON RAGGRUPPATI)? 0
NUMERO DI OSSERVAZIONI? 6
VOCE, FREQUENZA 1 ? 1 , 15
VOCE, FREQUENZA 2 ? 2 , 10
VOCE, FREQUENZA 3 ? 3 , 9
VOCE, FREQUENZA 4 ? 4 , 6
```

```
VOCE, FREQUENZA 5 ? 5 , 7
VOCE, FREQUENZA 6 ? 6 , 3
```

```
MEDIA          VARIANZA
2.78           2.57159999
```

```
DEVIAZIONE STANDARD
1.6036209
```

```
ALTRI DATI (1=SI,0=NO)? 1
```

```
METODO (0=POPOLAZIONE, 1=CAMPIONE)? 1
DATI (0=RAGGRUPPATI, 1=NON RAGGRUPPATI)? 0
NUMERO DI OSSERVAZIONI? 6
VOCE, FREQUENZA 1 ? 1 , 15
VOCE, FREQUENZA 2 ? 2 , 10
VOCE, FREQUENZA 3 ? 3 , 9
VOCE, FREQUENZA 4 ? 4 , 6
VOCE, FREQUENZA 5 ? 5 , 7
VOCE, FREQUENZA 6 ? 6 , 3
```

```
MEDIA          VARIANZA
2.78           2.62408162
```

```
DEVIAZIONE STANDARD
1.61990173
```

```
ALTRI DATI (1=SI,0=NO)? 0
```

Listing del programma

```
10 PRINT"MEDIA, VARIANZA, ";
20 PRINT"DEVIAZIONE STANDARD"
25 PRINT
30 PRINT"METODO (0=POPOLAZIONE, ";
31 PRINT"1=CAMPIONE)";
40 INPUT S
50 PRINT"DATI (0=RAGGRUPPATI, ";
51 PRINT"1=NON RAGGRUPPATI)";
60 INPUT K
70 PRINT"NUMERO DI OSSERVAZIONI";
80 INPUT N
90 R=0
100 M=0
110 P=0
120 IF K=1 THEN 230
129 REM-PER DATI RAGGRUPPATI
130 FOR I=1 TO N
140 PRINT"VOCE, FREQUENZA";I;
150 INPUT A,B
159 REM-ACCUMULARE I DATI INSERITI
160 R=R+B*A
169 REM-ACCUMULARE I VALORI INTERMEDI PER LA VARIANZA
170 P=P+B
180 M=M+B*A^2
190 NEXT I
199 REM-CALCOLARE LA MEDIA E LA VARIANZA
200 R=R/P
210 V=(M-P*R^2)/(P-S)
219 REM-STAMPA DEI RISULTATI
220 GOTO 310
229 REM-PER DATI NON RAGGRUPPATI
230 FOR I=1 TO N
240 PRINT"VOCE";I;
250 INPUT D
```

```

259 REM-ACCUMULARE I DATI INSERITI
260 P=P+D
269 REM-ACCUMULARE I VALORI INTEREDI PER LA VARIANZA
270 M=M+D^2
280 NEXT I
289 REM-CALCOLO DELLA MEDIA E DELLA VARIANZA E STAMPA
290 R=P/N
300 V=(M-N*R^2)/(N-1)
310 PRINT
319 REM-STAMPA RISULTATI
320 PRINT"MEDIA", "VARIANZA"
330 PRINTR, V
331 PRINT
332 PRINT"DEVIATIONE STANDARD"
333 PRINTSQR(V)
340 PRINT
349 REM-IL PROGRAMMA DEVE RIPARTIRE? SI RICHIEDE LA RISPOSTA DELL'UTENTE
350 PRINT"ALTRI DATI (1=SI,0=NO)";
360 INPUT S
370 IF S=1 THEN 25
380 END

```

MEDIA E DEVIAZIONE GEOMETRICA

Questo programma calcola la media e la deviazione geometrica standard di una serie di dati.

Esempio:

Trovare la media geometrica e la deviazione standard di 3, 5, 8, 3, 7, 2.

```
MEDIA E DEVIAZIONE GEOMETRICA
( PER TERMINARE IL PROGRAMMA INSERIRE 0 )
NUMERO DI OSSERVAZIONI? 6
VOCE 1 ? 3
VOCE 2 ? 5
VOCE 3 ? 8
VOCE 4 ? 3
VOCE 5 ? 7
VOCE 6 ? 2
MEDIA GEOMETRICA= 4.14068084
DEVIAZIONE GEOMETRICA= 1.72368956
NUMERO DI OSSERVAZIONI? 0
```

Listing del programma

```
10 PRINT"MEDIA E DEVIAZIONE GEOMETRICA"
20 PRINT
30 PRINT"( PER TERMINARE IL PROGRAMMA INSERIRE 0 )"
40 PRINT"NUMERO DI OSSERVAZIONI";
50 INPUT N
59 REM-TEST PER LA FINE DEL PROGRAMMA
60 IF N=0 THEN 200
59 REM-CALCOLO DELLA RADICE DA USARE
70 P=1/N
80 M=1
90 FOR I=1 TO N
100 PRINT"VOCE";I;
110 INPUT D
119 REM-SI CALCOLA LA MEDIA ITERATIVAMENTE
120 M=M*D/P
129 REM-SI ACCUMULANO I TERMINI INTERMEDI PER LA DEVIAZIONE
130 Q=Q+LOG(D)^2
140 NEXT I
149 REM-SI CALCOLA LA DEVIAZIONE
150 R=EXP(SQR(Q/(N-1)-(N/(N-1))*(LOG(M))^2)))
160 PRINT"MEDIA GEOMETRICA=";M
170 PRINT"DEVIAZIONE GEOMETRICA=";R
180 PRINT
189 REM-IL PROGRAMMA RIPARTE
190 GOTO 40
200 END
```

DISTRIBUZIONE BINOMIALE

Questo programma calcola la possibilità di ottenere un dato numero di eventi favorevoli in un determinato numero di prove di Bernulli.

Esempi:

Qual'è la possibilità di ottenere tre teste con cinque lanci di una moneta?

Qual'è la possibilità di ottenere due volte il numero uno, lanciando cinque volte un dado?

DISTRIBUZIONE BINOMIALE

```
(PER TERMINARE IL PROGRAMMA INSERIRE 0)
NUMERO DI PROVE? 5
NUMERO ESATTO DI SUCCESSI? 3
PROBABILITA' DI SUCCESSO? .5
```

```
PROBABILITA' DI 3 SUCCESSI
SU 5 PROVE = .3125
```

```
NUMERO DI PROVE? 5
NUMERO ESATTO DI SUCCESSI? 2
PROBABILITA' DI SUCCESSO? .166666667
```

```
PROBABILITA' DI 2 SUCCESSI
SU 5 PROVE = .160751029
```

```
NUMERO DI PROVE? 0
```

Listing del programma

```
10 PRINT"DISTRIBUZIONE BINOMIALE"
20 PRINT
30 DIM M(3)
40 PRINT"(PER TERMINARE IL PROGRAMMA INSERIRE 0)"
50 PRINT"NUMERO DI PROVE";
60 INPUT N
70 IF N=0 THEN 270
80 PRINT"NUMERO ESATTO DI SUCCESSI";
90 INPUT X
100 PRINT"PROBABILITA' DI SUCCESSO";
110 INPUT P
115 REM-CALCOLO DEI FATTORIALI
120 M(1)=N
130 M(2)=X
140 M(3)=N-X
150 FOR I=1 TO 3
160 IF M(I)=0 THEN 220
170 A=1
180 FOR J=1 TO M(I)
190 A=A*I
200 NEXT J
210 M(I)=LOG(A)
```

```

220 NEXT I
229 REM-USANDO I FATTORIALI CALCOLATI, CALCOLA LE PROBABILITA'
230 R=EXP(M(1)-M(2)-M(3)+X*LOG(P)+(N-X)*LOG(1-P))
235 PRINT
240 PRINT"PROBABILITA' DI";X:
245 PRINT"SUCCESSI"
247 PRINT"SU";N"PROVE ="":R
250 PRINT
259 REM-IL PROGRAMMA RIPARTE
260 GOTO 50
270 END

```

DISTRIBUZIONE DI POISSON

Il programma, mediante la distribuzione di Poisson, calcola la probabilità che un evento succeda un dato numero di volte. È necessario conoscere la frequenza che ci si aspetta dall'evento.

Esempio:

Si inietta un siero a 2.000 persone. La probabilità che qualche persona abbia una reazione negativa è dello 0,001. In questo caso ci si aspetta che due ($0,001 \times 2.000 = 2$) individui avranno una reazione negativa. Qual'è la probabilità che quattro persone abbiano una reazione negativa? E che l'abbia una persona sola?

DISTRIBUZIONE DI POISSON

<INSERIRE 0 PER TERMINARE IL PROGRAMMA>

FREQUENZA CALCOLATA? 2
FREQUENZA RILEVATA? 4
PROBABILITA' DI 4
EVENTI = .0902235222

FREQUENZA CALCOLATA? 2
FREQUENZA RILEVATA? 1
PROBABILITA' DI 1
EVENTO = .270670566


FREQUENZA CALCOLATA? 0

Listing del programma

```
10 PRINT"DISTRIBUZIONE DI POISSON"
15 PRINT
20 PRINT"<INSERIRE 0 PER TERMINARE IL PROGRAMMA>"
30 PRINT
40 PRINT"FREQUENZA CALCOLATA";
50 INPUT L
59 REM-FINE PROGRAMMA
60 IF L=0 THEN 180
70 PRINT"FREQUENZA RILEVATA";
80 INPUT X
89 REM-CALCOLA I FATTORIALI
90 A=1
100 FOR I=1 TO X
110 A=A*I
120 NEXT I
129 REM-CALCOLO DELLA PROBABILITA'
130 A=LOG(A)
140 A=EXP(-L+X*LOG(L)-A)
150 PRINT"PROBABILITA' DI";X
155 IF X=1 THEN PRINT"EVENTO =" ;A:GOTO169
160 PRINT"EVENTI =" ;A
169 REM-IL PROGRAMMA RIPARTE
170 GOTO 30
180 END
```


DISTRIBUZIONE NORMALE

Questo programma calcola la probabilità e la frequenza di determinati valori su una curva di normale distribuzione standard. È necessario utilizzare variabili non standard se si vogliono conoscere la media e la deviazione standard.



Normale distribuzione standard

L'area più scura rappresenta la probabilità di x . y corrisponde alla frequenza di x .

La probabilità normale è approssimata usando la seguente formula:

$$\text{probabilità} = 1 - r(a_1 t + a_2 t^2 + a_3 t^3) + \epsilon(x)$$

$$\begin{aligned} \text{con: } a_1 &= 0,4361836 \\ a_2 &= -0,1201676 \\ a_3 &= 0,9372980 \\ r &= (e^{-x^2/2}) (2\pi)^{-1/2} \\ t &= (1 + 0,3326x)^{-1} \\ |\epsilon(x)| &< 10^{-5} \end{aligned}$$

Esempio:

Il peso medio degli studenti in un college è di 75 Kg. La deviazione standard è di 7,5 Kg. Se i pesi sono normalmente distribuiti, qual'è la probabilità che uno studente abbia un peso compreso tra i 75 e i 90 chili? E quale tra i 65 e i 75?

DISTRIBUZIONE NORMALE

<0=STANDARD, 1=NON STANDARD>

CHE TIPO DI VARIABILE? 1

MEDIA? 150

DEVIATION STANDARD? 15

<PER TERMINARE IL PROGRAMMA X=99999>

X=? 180

FREQUENZA= .0539909665

PROBABILITA' = .977241178

X=? 130

FREQUENZA= .164010074

PROBABILITA' = .908798028


X=? 99999

Listing del programma

```
10 PRINT"DISTRIBUZIONE NORMALE"
20 PRINT
30 PRINT"(0=STANDARD, 1=NON STANDARD)"
40 PRINT"CHE TIPO DI VARIABILE";
50 INPUT S
60 IF S=0 THEN 120
69 REM-LE LINEE DAL 70 AL 110 RICHIEDONO DATI VARIABILI NON STANDARD
70 PRINT"MEDIA";
80 INPUT M
90 PRINT"DEVIAZIONE STANDARD";
100 INPUT S
110 GOTO 130
120 S=1
130 PRINT
140 PRINT"(PER TERMINARE IL PROGRAMMA X=99999)"
150 PRINT"X=";
160 INPUT X
170 IF X=99999 THEN 290
180 X=ABS((X-M)/S)
190 R=EXP(-(X^2)/2)/2.5066282746
200 PRINT"FREQUENZA= ";R
210 Z=X
220 T=1/(1+.33267*ABS(X))
230 T=1-R*(.4361836*T-.120167*T^2+.937298*T^3)
240 IF Z>=.0 THEN 260
250 T=1-T
260 PRINT"PROBABILITA' =";T
270 PRINT
279 REM-IL PROGRAMMA RIPARTE
280 GOTO 150
290 END
```

DISTRIBUZIONE DEL CHI-QUADRO

Il programma calcola il valore della parte terminale per punti su una curva di distribuzione del chi-quadro X^2 .



Distribuzione del chi-quadro

L'area più scura rappresenta il valore della parte terminale del chi-quadro (X^2).

La funzione di distribuzione chi-quadro è calcolata utilizzando la seguente formula:

$$\text{con } v \text{ dispari, valore della parte terminale} = 1 - \frac{(X^2)(v+1)/2 \cdot e^{-X^2/2}}{1 \cdot 3 \cdot 5 \dots \cdot v} \cdot \left(\frac{2}{X^2 \pi} \right)^{1/2} \cdot Z$$

$$\text{con } v \text{ pari, valore della parte terminale} = 1 - \frac{(X^2)v/2 \cdot e^{-X^2/2}}{2 \cdot 4 \dots \cdot v} \cdot Z$$

in cui v = gradi di libertà

$$Z = 1 + \sum_{m=1}^{\infty} \frac{(X^2)^m}{m! (v+2) \cdot (v+4) \cdot \dots \cdot (v+2m)}$$

Poichè la sommatoria per il calcolo di Z non può essere estesa all'infinito, questa si arresterà quando il termine successivo sarà inferiore ad un prescelto livello di precisione. La precisione di calcolo è limitata approssimativamente a 10^{-7} .

Esempio:

Sono state date delle pillole di sonnifero a 54 persone di un gruppo di 168 individui che soffre di insonnia. Alle rimanenti persone sono stati dati dei placebo, cioè delle pillole senza nessun effetto terapeutico, se non quello psicologico.

Successivamente è stato chiesto se le pillole hanno aiutato le persone ad

addormentarsi. La statistica X^2 è stata calcolata a 2,571108 con un grado di libertà. Qual'è il valore della parte terminale?

DISTRIBUZIONE DEL CHI-QUADRO

```
(PER TERMINARE IL PROGRAMMA INSERIRE 0)
GRADI DI LIBERTA'? 1
CHI-QUADRO? 2.571108
VALORE DELLA PARTE TERMINALE = .108831483

GRADI DI LIBERTA'? 0
```

Listing del programma

```
10 PRINT "DISTRIBUZIONE DEL CHI-QUADRO"
20 PRINT
30 PRINT "(PER TERMINARE IL PROGRAMMA INSERIRE 0)"
40 PRINT "GRADI DI LIBERTA' ";
50 INPUT V
60 IF V=0 THEN 280
70 PRINT "CHI-QUADRO ";
80 INPUT W
89 REM- R=PRODOTTO DEL DENOMINATORE
90 R=1
100 FOR I=V TO 2 STEP-2
110 R=R*I
120 NEXT I
129 REM- K=PRODOTTO AL NUMERATORE
130 K=W*(INT((V+1)/2))*EXP(-W/2)/R
138 REM- IL FATTORE PI, CIOE' PI GRECO, E' USATO SOLO QUANDO I GRADI DI
139 REM- LIBERTA' SONO DISPARI
140 IF INT(W/2)=W/2 THEN 170
150 J=SQR(2/W/PI)
160 GOTO 180
169 REM- L, (FATTORE DI SOMMA), E' CALCOLATO CON LE LINEE DAL 170 AL 240
170 J=1
180 L=1
190 M=1
200 V=V+2
210 M=M*W/V
219 REM- VERIFICA PER LA FINE DELLA SOMMA
220 IF M<.0000001 THEN 250
230 L=L+M
240 GOTO 200
250 PRINT "VALORE DELLA PARTE TERMINALE = "; 1-J*K*L
260 PRINT
269 REM- IL PROGRAMMA RIPARTE
270 GOTO 40
280 END
```

OPZIONE

È possibile calcolare il percentile, cioè l'altra parte dell'area della curva, invece del valore della parte terminale. I necessari cambiamenti di programma sono elencati nell'esempio seguente.

Esempio:

Qual'è il percentile nell'esempio precedente?

DISTRIBUZIONE DEL CHI-QUADRO

(PER TERMINARE IL PROGRAMMA INSERIRE 0)

GRADI DI LIBERTA'? 1

CHI-QUADRO? 2.571108

PERCENTUALE = .891168517

GRADI DI LIBERTA'? 0

Opzione

1 REM - OPZIONE 250

10 PRINT"DISTRIBUZIONE DEL CHI-QUADRO"

:

:

240 GOTO 200

250 PRINT"PERCENTUALE =" ; J*K*L

260 PRINT

269 REM- IL PROGRAMMA RIPARTE

270 GOTO 40

280 END

TEST DEL CHI-QUADRO

Questo programma calcola la statistica del chi-quadro (X^2) e i gradi di libertà associati ad una data tabella di eventi. Verranno stampati sia il valore cercato per ogni casella, sia il contributo al chi-quadro da ogni casella.

La dimensione dello statement alla linea 30 limita la grandezza della tabella di eventi. È possibile cambiarne le dimensioni secondo il seguente schema:

30 DIM V1(R·C), V2(C), A(R)

dove: R = numero di righe della tabella di eventualità
C = numero di colonne della tabella di eventualità

Esempio:

Ad alcune persone, appartenenti ad un gruppo di sofferenti di insonnia, vengono date delle pillole di sonnifero, mentre alle altre vengono somministrati dei placebo. Più tardi a tutti viene chiesto se le pillole li hanno aiutati a dormire.

I risultati sono riportati nella tabella seguente. Qual'è il valore del chi-quadro statistico?

	dormito bene	dormito male
Sonnifero	44	10
Placebo	81	35

```
30 DIM V1(4),V(2),A(2)
READY.
```

```
NUMERO DI RIGHE? 2
NUMERO DI COLONNE? 2
TABELLA DEGLI EVENTI:
```

```
RIGA 1
ELEMENTO 1 ? 44
ELEMENTO 2 ? 10
RIGA 2
ELEMENTO 1 ? 81
ELEMENTO 2 ? 35
```

VALORE OSSERVATO	VALORE PREVISTO	CONTRIBUTO CHI ²
COLONNA 1		
44	39.7058824	.362549019
81	85.2941177	.168772821
COLONNA 2		
10	14.2941176	1.00708061
35	30.7058824	.468813387

```
CHI QUADRO = 2.00721584
GRADI DI LIBERTA' = 1
```

Listing del programma


```

10 PRINT"TEST DEL CHI QUADRO"
20 PRINT
28 REM - SI LIMITANO LE DIMENSIONI DELLA TABELLA DEGLI EVENTI A V1(R*C),
29 REM - V2(C), A(R); DOVE R=NUMERO DI RIGHE, C=NUMERO DI COLONNE
30 DIM V1(25),V2(2),A(2)
40 PRINT"NUMERO DI RIGHE";
49 REM - LA TABELLA DEGLI EVENTI VIENE INTRODOTTA DALLA LINEA 50 ALLA 150
50 INPUT R
50 PRINT"NUMERO DI COLONNE";
70 INPUT C
80 PRINT"TABELLA DELLE EVENTUALITA':"
90 FOR I=1 TO R
100 PRINT"RIGA";I
110 FOR J=1 TO C
120 PRINT"    ELEMENTO";J;
130 INPUT V1((I-1)*C+J)
140 NEXT J
150 NEXT I
160 PRINT
169 REM - SOMMA DELLE FREQUENZE PARZIALI PER OGNI RIGA
170 L=0
180 M=1
190 FOR I=1 TO R
200 FOR J=1 TO C
210 A(I)=A(I)+V1(M)
220 M=M+1
230 NEXT J
240 L=L+A(I)
250 NEXT I
260 N=R*C
269 REM - SOMMA DELLE FREQUENZE PARZIALI PER OGNI COLONNA
270 FOR I=1 TO C
280 FOR J=1 TO N STEP C
290 V2(I)=V2(I)+V1(J)
300 NEXT J
310 NEXT I
320 Z=0
330 PRINT"OSSERVATO    PREVISTO    CONTR. CHI 12"
340 FOR I=1 TO C
350 PRINT" COLONNA";I
360 FOR J=1 TO R
369 REM - P=VALORE ATTESO DELLA CASELLA
370 P=A(J)*V2(I)/L
375 X=I+(J-1)*C
378 REM - SI USA LA CORREZIONE DI YATES PER LA CONTINUITA' IN UN TEST DEL CHI
379 REM - QUADRO FORMATO 2 X 2
380 IF R<>2 THEN 390
381 IF C<>2 THEN 390
382 Y=(ABS(V1(X)-P)-.5)*12/P
383 GOTO 400
389 REM - Y=CONTRIBUTO DEL CHI QUADRO DA QUESTA CASELLA
390 Y=(V1(X)-P)*12/P
399 REM - Z=VALORE TOTALE DEL CHI QUADRO
400 Z=Z+Y
410 PRINTTAB(3);V1(X),P;TAB(23)Y
420 NEXT J
430 NEXT I
440 PRINT
450 PRINT"CHI QUADRO =" ;Z
460 PRINT"GRADI DI LIBERTA' =" ;
461 PRINT (C-1)*(R-1)
470 END

```

DISTRIBUZIONE t DI STUDENT

Il programma calcola il valore della coda di destra per punti di una curva di distribuzione t. È necessario assegnare il valore di t e i gradi di libertà.



Distribuzione t di student

L'area più scura rappresenta il valore della coda di destra per t.

Il valore della coda di destra è approssimato usando la seguente formula:

valore
della coda di destra $= \frac{1}{2} (1 + a_1 x + a_2 x^2 + a_3 x^3 + a_4 x^4)^{-4} + \epsilon(x)$

dove: $a_1 = 0,196854$

$a_2 = 0,115194$

$a_3 = 0,000344$

$a_4 = 0,019527$

$x = t^{2/3} \left(1 - \frac{2}{9d}\right) - \frac{7}{9} \left(\frac{2}{9} + t^{4/3} \cdot \frac{2}{9d}\right)^{-1/2}$

$d = \text{gradi di libertà}$

$|\epsilon(x)| < 2,5 \cdot 10^{-4}$

Esempi:

Qual'è il valore della coda di destra, quando il valore di t è 2,921 ed esistono 16 gradi di libertà?

Qual'è il valore della coda di destra, quando il valore di t è 11,178 ed esistono 5 gradi di libertà?

DISTRIBUZIONE T DI STUDENT

<INSERIRE 0 PER TERMINARE IL PROGRAMMA>

VALORE DI T? 2.921

GRADI DI LIBERTÀ? 16

VALORE DELLA CODA DI DESTRA = 9.7E-04

VALORE DI T? 11.178

GRADI DI LIBERTÀ? 5

VALORE DELLA CODA DI DESTRA = 4E-05

VALORE DI T? 0

Listing del programma

```

10 PRINT"DISTRIBUZIONE T DI STUDENT"
20 PRINT
30 PRINT"(INSERIRE 0 PER TERMINARE IL PROGRAMMA)"
40 PRINT"VALORE DI T";
50 INPUT T
60 IF T=0 THEN 340
70 PRINT "GRADI DI LIBERTA'";
80 INPUT D
90 X=1
100 Y=1
110 T=T+2
119 REM - CALCOLARE USANDO L'INVERSO PER PICCOLI VALORI DI T
120 IF T<1 THEN 170
130 S=Y
140 R=D
150 Z=T
160 GOTO 200
170 S=D
180 R=Y
190 Z=1/T
200 J=2/9/S
210 K=2/9/R
219 REM - SI USANO FORMULE DI APPROSSIMAZIONE
220 L=ABS((1-K)*Z+(1/3)-1+J)/SQR(K*Z+(2/3)+J)
230 IF R<4 THEN 270
240 X=,5/(1+L*(,196854+L*(,115194+L*(,000344+L*(,019527))))+4
250 X=INT(X*10000+,5)/10000
260 GOTO 290
270 L=L*(1+,08*L+4/R+3)
280 GOTO 240
289 REM - APPROSSIMARE SE E' STATO CALCOLATO L'INVERSO
290 IF T>=1 THEN 310
300 X=1-X
310 PRINT "VALORE DELLA CODA DI DESTRA =" ;X
320 PRINT
329 REM - IL PROGRAMMA RIPARTE
330 GOTO 40
340 END

```

TEST DELLA DISTRIBUZIONE t DI STUDENT

Questo programma calcola il t statistico e i gradi di libertà di una distribuzione di Student. Il calcolo è basato su una delle tre seguenti ipotesi.

La prima prende come ipotesi che una popolazione media sia uguale ad un determinato valore. È necessario introdurre gli elementi del campione ed il valore della media.

Le rimanenti ipotesi paragonano due popolazioni. In entrambi i test le medie delle due popolazioni sono uguali, mentre la deviazione standard può anche non esserlo. Con queste ipotesi è necessario introdurre gli elementi di ciascun campione.

La dimensione dello statement alla linea 30 limita la grandezza dei campioni che si vogliono inserire. Tuttavia è possibile cambiare il limite mediante il seguente schema:

30 DIM P(N,2)

dove N è la massima grandezza del campione.

Esempi:

Considerando il quoziente di intelligenza QI di un campione di ragazzi, si sono avuti i seguenti risultati:

101, 99, 120, 79, 111, 98, 106, 112, 87, 97.

Calcolare il t statistico, con l'ipotesi che la media della popolazione sia 100. In un secondo campione si sono avuti i seguenti risultati:

101, 95, 130, 150, 75, 79, 111, 100, 98, 91.

Calcolare il t statistico con l'ipotesi che i due campioni hanno la stessa media e la stessa deviazione standard.

TEST DELLA DISTRIBUZIONE T DI STUDENT

```
TEST 1: MEDIA=X
TEST 2: MEDIA=MEDIA, DS=DS
TEST 3: MEDIA=MEDIA, DS<>DS
QUALE IPOTESI? 1
```

```
CAMPIONE 1 :
NUMERO DI ELEMENTI ? 10
ELEMENTI 1 ? 101
ELEMENTI 2 ? 99
ELEMENTI 3 ? 120
ELEMENTI 4 ? 79
ELEMENTI 5 ? 111
ELEMENTI 6 ? 98
ELEMENTI 7 ? 106
ELEMENTI 8 ? 112
```

ELEMENTI 9 ? 87
ELEMENTI 10 ? 97

VALORE DELLA MEDIA? 100

VALORE DI T= .26151303
GRADI DI LIBERTA' = 9

TEST DELLA DISTRIBUZIONE T DI STUDENT

TEST 1: MEDIA=X
TEST 2: MEDIA=MEDIA, DS=DS
TEST 3: MEDIA=MEDIA, DS<>DS
QUALE IPOTESI? 2

CAMPIONE 1 :
NUMERO DI ELEMENTI ? 10
ELEMENTI 1 ? 101
ELEMENTI 2 ? 99
ELEMENTI 3 ? 120
ELEMENTI 4 ? 79
ELEMENTI 5 ? 111
ELEMENTI 6 ? 98
ELEMENTI 7 ? 106
ELEMENTI 8 ? 112
ELEMENTI 9 ? 87
ELEMENTI 10 ? 97

CAMPIONE 2 :
NUMERO DI ELEMENTI ? 10
ELEMENTI 1 ? 101
ELEMENTI 2 ? 95
ELEMENTI 3 ? 130
ELEMENTI 4 ? 150
ELEMENTI 5 ? 75
ELEMENTI 6 ? 79
ELEMENTI 7 ? 111
ELEMENTI 8 ? 100
ELEMENTI 9 ? 98
ELEMENTI 10 ? 91

VALORE DI T= .246515218
GRADI DI LIBERTA' = 18

Listing del programma

```
10 PRINT"TEST DELLA DISTRIBUZIONE T DI STUDENT"
20 PRINT
28 REM - SI LIMITA LA GRANDEZZA DEL CAMPIONE A P(N,2) IN CUI N=DIMENSIONE
29 REM - MASSIMA DEL CAMPIONE
30 DIM P(10,2)
40 DIM V(2),R(2),M(2),D(2)
50 PRINT"TEST 1: MEDIA=X"
60 PRINT"TEST 2: MEDIA=MEDIA, DS=DS"
70 PRINT"TEST 3: MEDIA=MEDIA, DS<>DS"
80 PRINT"QUALE IPOTESI";
90 INPUT T
100 PRINT
109 REM~- INSERIRE 1 O 2 CAMPIONI A SECONDA DELLE IPOTESI
110 FOR I=1 TO SGN(T-1)+1
120 V(I)=0
130 D(I)=0
140 PRINT"CAMPIONE";I;":":
```


```

150 PRINT"NUMERO DI ELEMENTI ";
160 INPUT R(I)
170 FOR J=1 TO R(I)
180 PRINT" ELEMENTI";J;
190 INPUT P(J,I)
199 REM - SI ACCUMULANO I CAMPIONI
200 V(I)=V(I)+P(J,I)
210 D(I)=D(I)+P(J,I)^2
220 NEXT J
229 REM - SI CALCOLANO I VALORI INTERMEDI
230 M(I)=V(I)/R(I)
240 V(I)=(D(I)-V(I)^2/R(I))/(R(I)-1)
250 NEXT I
260 PRINT
270 IF T=2 THEN 340
280 IF T=3 THEN 380
289 REM - SI INSERISCONO I VALORI DATI PER LA PRIMA IPOTESI
290 PRINT "VALORE DELLA MEDIA";
300 INPUT M
309 REM - CALCOLO DI T E GRADI DI LIBERTA' PER LA PRIMA IPOTESI
310 A=(M(1)-M)*SQR(R(1)/V(1))
320 B=R(1)-1
330 GOTO 420
339 REM - CALCOLO DI T E GRADI DI LIBERTA' PER LA SECONDA IPOTESI
340 A=(M(1)-M(2))/SQR(1/R(1)+1/R(2))
350 B=R(1)+R(2)-2
360 A=A/SQR(((R(1)-1)*V(1)+(R(2)-1)*V(2))/B)
370 GOTO 420
379 REM - CALCOLO DI T E GRADI DI LIBERTA' PER LA TERZA IPOTESI
380 A=(M(1)-M(2))/SQR(V(1)/R(1)+V(2)/R(2))
390 B=(V(1)/R(1)+V(2)/R(2))^2
400 B=B/((V(1)/R(1))^2/(R(1)+1)+(V(2)/R(2))^2/(R(2)+1))-2
410 B=INT(B+.5)
420 PRINT
430 PRINT"VALORE DI T=";ABS(A)
440 PRINT"GRADI DI LIBERTA' =" ;B
450 END

```

DISTRIBUZIONE F

Il programma calcola i valori percentili per determinati valori di una curva di distribuzione F. È necessario inserire i valori di F, i gradi di libertà nel numeratore e quelli nel denominatore.



La distribuzione F

L'area più scura rappresenta il percentile.

La funzione di distribuzione F è approssimata mediante la seguente formula:

$$\text{percentile} = 1 - \frac{1}{2} (1 + a_1 y + a_2 y^2 + a_3 y^3 + a_4 y^4)^{-4} + \epsilon(y)$$

$$\text{dove: } a_1 = 0,196854$$

$$a_2 = 0,115194$$

$$a_3 = 0,000344$$

$$a_4 = 0,019527$$

$$y = (F^{1/3} (1 - \frac{2}{9d_2}) - (1 - \frac{2}{9d_1})) (\frac{2}{9d_1} + F^{2/3} \cdot \frac{2}{9d_2})^{-1/2}$$

$$d_1 = \text{gradi di libertà nel numeratore}$$

$$d_2 = \text{gradi di libertà nel denominatore}$$

$$|\epsilon(y)| < 2,5 \times 10^{-4}$$

Esempi:

Qual'è il percentile di una curva di distribuzione F, quando il valore di F è 0,474 e i gradi di libertà sono 1 e 18?

Qual'è il percentile quando il valore di F è 23,7 e i gradi di libertà sono 3 e 6?

DISTRIBUZIONE F

<INSERIRE 0 PER TERMINARE IL PROGRAMMA>

VALORE DI F? .474

GRADI DI LIBERTÀ' NEL NUMERATORE? 1

GRADI DI LIBERTÀ' NEL DENOMINATORE? 18

PERCENTILE = .4937

VALORE DI F? 23.7

GRADI DI LIBERTÀ' NEL NUMERATORE? 3

GRADI DI LIBERTA' NEL DENOMINATORE? 6
PERCENTILE = .9984

VALORE DI F? 0

Listing del programma

```
10 PRINT"DISTRIBUZIONE F"
20 PRINT
30 PRINT"<INSERIRE 0 PER TERMINARE IL PROGRAMMA>"
40 PRINT "VALORE DI F";
50 INPUT F
60 IF F=0 THEN 340
70 PRINT"GRADI DI LIBERTA' ";
71 PRINT"NEL NUMERATORE";
80 INPUT D1
90 PRINT"GRADI DI LIBERTA' ";
91 PRINT"NEL DENOMINATORE";
100 INPUT D2
110 X=1
119 REM - CALCOLA USANDO L' INVERSO PER PICCOLI VALORI DI F
120 IF F<1 THEN 170
130 S=D1
140 T=D2
150 Z=F
160 GOTO 200
170 S=D2
180 T=D1
190 Z=1/F
200 J=2/3/S
210 K=2/3/T
219 REM - CALCOLA CON L' USO DI FORMULE DI APPROSSIMAZIONE
220 Y=ABS((1-K)*Z↑(1/3)-1+J)/SQR(K*Z↑(2/3)+J)
230 IF T<4 THEN 270
240 X=.5/(1+Y*(.196854+Y*(.115194+Y*(.000344+Y*(.019527))))↑4
250 X=INT(X*10000+.5)/10000
260 GOTO 290
270 Y=Y*(1+.08*Y↑4/T↑3)
280 GOTO240
289 REM - APPROSSIMA SE E' STATO CALCOLATO L' INVERSO
290 IF F>=1 THEN 310
300 X=1-X
310 PRINT"PERCENTILE =" ;1-X
320 PRINT
329 REM - IL PROGRAMMA RIPARTE
330 GOTO 40
340 END
```

OPZIONE

È possibile anche calcolare il valore di fine coda (l'area chiara nella figura precedente). Le variazioni necessarie di programma sono elencate negli esempi seguenti.

Esempi:

Qual'è il valore di fine coda di una curva di distribuzione F, in cui il valore di F è 0,474 e i gradi di libertà sono 1 e 18?

Qual'è il valore di fine coda, quando il valore di F è 23,7 e i gradi di libertà sono 3 e 6?

DISTRIBUZIONE DI F

```
<INSERIRE 0 PER TERMINARE IL PROGRAMMA>  
VALORE DI F? .474  
GRADI DI LIBERTA' NEL NUMERATORE? 1  
GRADI DI LIBERTA' NEL DENOMINATORE? 18  
VALORE DI FINE CODA = .5063
```

```
VALORE DI F? 23.7  
GRADI DI LIBERTA' NEL NUMERATORE? 3  
GRADI DI LIBERTA' NEL DENOMINATORE? 6  
VALORE DI FINE CODA = 1.6E-03
```

```
VALORE DI F? 0  
:  
:
```

Opzione

```
1 REM - OPZIONE 310  
10 PRINT"DISTRIBUZIONE DI F"  
:  
:  
300 X=1-X  
310 PRINT"VALORE DI FINE CODA =" ;X  
320 PRINT  
329 REM - IL PROGRAMMA RIPARTE  
330 GOTO 40  
340 END  
:  
:
```

COEFFICIENTE DI CORRELAZIONE LINEARE

Questo programma calcola il coefficiente di correlazione tra due variabili. Si presume che tra le due variabili esista una correlazione lineare. È necessario inserire le coordinate del gruppo di punti dati che forma la linea di regressione.

Esempio:

La tabella che compare più sotto riporta l'altezza di dodici uomini e dei loro dodici figli. Qual'è il coefficiente di correlazione tra le altezze dei padri e quella dei figli?

padri	65	63	67	64	68	62	70	66	68	67	69	71
figli	68	66	68	65	69	66	68	65	71	67	68	70

altezze in pollici (= 2,54 cm)

COEFFICIENTE DI CORRELAZIONE LINEARE

```
NUMERO DI PUNTI? 12
X,Y DEL PUNTO 1 ? 65 , 68
X,Y DEL PUNTO 2 ? 63 , 66
X,Y DEL PUNTO 3 ? 67 , 68
X,Y DEL PUNTO 4 ? 64 , 65
X,Y DEL PUNTO 5 ? 68 , 69
X,Y DEL PUNTO 6 ? 62 , 66
X,Y DEL PUNTO 7 ? 70 , 68
X,Y DEL PUNTO 8 ? 66 , 65
X,Y DEL PUNTO 9 ? 68 , 71
X,Y DEL PUNTO 10 ? 67 , 67
X,Y DEL PUNTO 11 ? 69 , 68
X,Y DEL PUNTO 12 ? 71 , 70
```

COEFFICIENTE DI CORRELAZIONE = .702652258

Listing del programma

```
10 PRINT"COEFFICIENTE DI CORRELAZIONE LINEARE"
20 PRINT
30 PRINT"NUMERO DI PUNTI";
40 INPUT N
99 REM - INSERIRE LE COORDINATE DEI PUNTI DATI
100 FOR I=1 TO N
110 PRINT"X,Y DEL PUNTO";I;
120 INPUT X,Y
```



```

129 REM - SI ACCUMULANO I VALORI INTERMEDI
130 J=J+X
140 K=K+Y
150 L=L+X^2
160 M=M+Y^2
170 R=R+X*Y
180 NEXT I
189 REM - SI CALCOLA IL COEFFICIENTE E LO SI STAMPA
190 R2=(N*R-J*K)/SQR((N*L-J^2)*(N*M-K^2))
200 PRINT
210 PRINT"COEFFICIENTE DI CORRELAZIONE =";
215 PRINT R2
220 END

```

REGRESSIONE LINEARE

Questo programma linearizza un certo numero di dati, utilizzando il metodo dei minimi quadrati. L'equazione della linea, il coefficiente di determinazione, il coefficiente di correlazione e l'errore standard di stima verranno stampati.

Una volta che la linea sarà stata fissata, si potranno stimare i valori di y per determinati valori di x.

Esempio:

La tabella successiva mostra le altezze e i pesi di undici studenti maschi di un collegio. Si adatti una curva a questi punti. Quale potrà essere mediamente il peso di due studenti alti 70 e 72 pollici.

altezza (pollici)	71	73	64	65	61	70	65	72	63	67	64
peso (libbre)	160	183	154	168	159	180	145	210	132	168	141

(1 pollice = 2,54 cm, 1 libbra = 453,6 g)

REGRESSIONE LINEARE

NUMERO DI PUNTI CONOSCIUTI? 11

X,Y DEL PUNTO 1 ? 71 , 160

X,Y DEL PUNTO 2 ? 73 , 183

X,Y DEL PUNTO 3 ? 64 , 154

X,Y DEL PUNTO 4 ? 65 , 168

X,Y DEL PUNTO 5 ? 61 , 159

X,Y DEL PUNTO 6 ? 70 , 180

X,Y DEL PUNTO 7 ? 65 , 145

X,Y DEL PUNTO 8 ? 72 , 210

X,Y DEL PUNTO 9 ? 63 , 132

X,Y DEL PUNTO 10 ? 67 , 168

X,Y DEL PUNTO 11 ? 64 , 141

$F(X) = -106.791727 + (4.04722312 * X)$

COEFFICIENTE DI DETERMINAZIONE(R²)= .556260313

COEFFICIENTE DI CORRELAZIONE= .745828608

ERRORE STANDARD DI STIMA= 15.4134854

INTERPOLAZIONE: (INSERIRE X=0 PER TERMINARE)

X=? 70

Y = 176.513892

X=? 72

Y = 184.608338

X=? 0

Listing del programma

```
10 PRINT "REGRESSIONE LINEARE"
20 PRINT
30 PRINT "NUMERO DI PUNTI CONOSCIUTI";
40 INPUT N
99 REM - CICLO PER INSERIRE LE COORDINATE DEI PUNTI
100 FOR I=1 TO N
110 PRINT "X,Y DEL PUNTO";I;
120 INPUT X,Y
129 REM - ACCUMULA LE SOMME INTERMEDIE
130 J=J+X
140 K=K+Y
150 L=L+X^2
160 M=M+Y^2
170 R2=R2+X*Y
180 NEXT I
189 REM - CALCOLA IL COEFFICIENTE DELLA CURVA
190 B=(N*R2-K*J)/(N*L-J^2)
200 A=(K-B*J)/N
210 PRINT
220 PRINT "F(X)=";A;" + (";B;" * X)"
229 REM - CALCOLA LA REGRESSIONE
230 J=B*(R2-J*K/N)
240 M=M-K^2/N
250 K=M-J
260 PRINT
270 R2=J/M
280 PRINT "COEFFICIENTE DI ";
281 PRINT "DETERMINAZIONE";
282 PRINT "(R^2)= ";R2
283 PRINT
290 PRINT "COEFFICIENTE DI ";
291 PRINT "CORRELAZIONE=";SQR(R2)
292 PRINT
300 PRINT "ERRORE STANDARD DI STIMA=";
301 PRINT SQR(K/(N-2))
310 PRINT
318 REM - SI CALCOLANO LE ORDINATE Y DEI PUNTI DI CUI SI SONO INSERITE LE
319 REM - ASCISSE X
320 PRINT "INTERPOLAZIONE: ";
321 PRINT "(INSERIRE X=0 PER TERMINARE)"
330 PRINT "X =";
340 INPUT X
349 REM - IL PROGRAMMA RIPARTE O TERMINA? E' RICHIESTA LA RISPOSTA DELL'UTENTE
350 IF X=0 THEN 390
360 PRINT "Y =";A+B*X
370 PRINT
380 GOTO 330
390 END
```

REGRESSIONE LINEARE MULTIPLA

Questo programma calcola i coefficienti di una equazione lineare a variabili multiple utilizzando il metodo dei minimi quadrati. L'equazione ha la seguente forma:

$$y = c + a_1x_1 + a_2x_2 + \dots + a_nx_n$$

dove: y = variabile dipendente

c = costante

a_1, a_2, \dots, a_n = coefficienti delle variabili indipendenti x_1, x_2, \dots, x^n .

Verranno stampati la costante e i coefficienti.

È necessario dare le coordinate x e y dei punti conosciuti. Una volta che è stata trovata l'equazione utilizzando i dati inseriti, si potranno prevedere i valori delle variabili dipendenti per assegnare valori delle variabili indipendenti.

La dimensione dello statement alla linea 30 limita il numero dei dati relativi ai punti conosciuti che l'equazione deve contenere. È possibile variare questo limite secondo il seguente schema:

30 DIM X(N+1), S(N+1), T(N+1), A(N+1, N+2)

dove: N = numero dei dati relativi ai punti conosciuti.

Esempio:

La tabella sottostante mostra l'età, l'altezza e il peso di otto ragazzi. Usando il peso come variabile dipendente si adatta una curva ai sottostanti valori. Si stimi poi il peso di un bambino di sette anni alto 51 pollici.

età (anni)	8	9	6	10	8	9	9	7
altezza (pollici)	48	49	44	59	55	51	55	50
peso (libbre)	59	55	50	80	61	75	67	58

1 pollice = 2,54 cm

1 libbra = 453,6 g

REGRESSIONE LINEARE MULTIPLA

NUMERO DI PUNTI CONOSCIUTI? 8

DI VARIABILI INDIPENDENTI? 2

PUNTO 1

VARIABILE 1 ? 8

VARIABILE 2 ? 48

VARIABILE DIPENDENTE? 59

```

PUNTO 2
  VARIABILE 1 ? 9
  VARIABILE 2 ? 49
  VARIABILE DIPENDENTE? 55
PUNTO 3
  VARIABILE 1 ? 6
  VARIABILE 2 ? 44
  VARIABILE DIPENDENTE? 50
PUNTO 4
  VARIABILE 1 ? 10
  VARIABILE 2 ? 59
  VARIABILE DIPENDENTE? 80
PUNTO 5
  VARIABILE 1 ? 8
  VARIABILE 2 ? 55
  VARIABILE DIPENDENTE? 61
PUNTO 6
  VARIABILE 1 ? 9
  VARIABILE 2 ? 51
  VARIABILE DIPENDENTE? 75
PUNTO 7
  VARIABILE 1 ? 9
  VARIABILE 2 ? 55
  VARIABILE DIPENDENTE? 67
PUNTO 8
  VARIABILE 1 ? 7
  VARIABILE 2 ? 50
  VARIABILE DIPENDENTE? 58

COEFFICIENTI DELLA EQUAZIONI:
  COSTANTE:-15.7021277
  VARIABILE ( 1 ) : 3.68085106
  VARIABILE ( 2 ) : .943262412

COEFFICIENTE DI DETERMINAZIONE
  (R12)= .715697404
COEFFICIENTE DI CORRELAZIONE
MULTIPLA = .845989009
ERRORE STANDARD DI STIMA = 6.42887917

INTERPOLAZIONE(INSERIRE 0 PER TERMINARE IL PROGRAMMA)
  VARIABILE 1 ? 7
  VARIABILE 2 ? 51
  VARIABILE DIPENDENTE = 58.1702128

  VARIABILE 1 ? 0

```

Listing del programma

```

10 PRINT"REGRESSIONE LINEARE MULTIPLA"
20 PRINT
29 REM - FISSA I LIMITI DELLE MATRICI A X(N+1), S(N+1), T(N+1), A(N+1,N+2)
30 DIM X(9),S(9),T(9),A(9,10)
40 PRINT "NUMERO DI PUNTI CONOSCIUTI";
50 INPUT N
60 PRINT "# DI VARIABILI INDIPENDENTI";
70 INPUT V
80 X(1)=1
90 FOR I=1 TO N
100 PRINT "PUNTO";I
110 FOR J=1 TO V
119 REM - INSERIRE LE VARIABILI INDIPENDENTI PER OGNI PUNTO
120 PRINT " VARIABILE";J;
130 INPUT X(J+1)
140 NEXTJ

```

```

149 REM - INSERIRE LA VARIABILE DIPENDENTE PER OGNI PUNTO
150 PRINT "VARIABILE DIPENDENTE";
160 INPUT X(V+2)
169 REM - RIEMPIRE LA MATRICE DA UTILIZZARE NELL'ADATTAMENTO DELLA CURVA
170 FOR K=1 TO V+1
180 FOR L=1 TO V+2
190 A(K,L)=A(K,L)+X(K)*X(L)
200 S(K)=A(K,V+2)
210 NEXT L
220 NEXT K
230 S(V+2)=S(V+2)+X(V+2)^2
240 NEXT I
248 REM - LE ISTRUZIONI DALLA RIGA 250 ALLA RIGA 500 ADATTANO LA CURVA
249 REM - RISOLVENDO IL SISTEMA DI EQUAZIONI LINEARI NELLA MATRICE A<>
250 FOR I=2 TO V+1
260 T(I)=A(1,I)
270 NEXT I
280 FOR I=1 TO V+1
290 J=I
300 IF A(J,I)<>0 THEN 340
305 J=J+1
310 IF J=V+1 THEN 300
320 PRINT "NON ESISTE UNA SOLA SOLUZIONE"
330 GOTO 810
340 FOR K=1 TO V+2
350 B=A(I,K)
360 A(I,K)=A(J,K)
370 A(J,K)=B
380 NEXT K
390 Z=1/A(I,I)
400 FOR K=1 TO V+2
410 A(I,K)=Z*A(I,K)
420 NEXT K
430 FOR J=1 TO V+1
440 IF J=I THEN 490
450 Z=-A(J,I)
460 FOR K=1 TO V+2
470 A(J,K)=A(J,K)+Z*A(I,K)
480 NEXT K
490 NEXT J
500 NEXT I
510 PRINT
520 PRINT "COEFFICIENTI DELLA EQUAZIONI:"
525 PRINT "          COSTANTE:"A(1,V+2)
530 FOR I=2 TO V+1
540 PRINT "VARIABILE (<I-1;>) :";A(I,V+2)
550 NEXT I
560 P=0
570 FOR I=2 TO V+1
580 P=P+A(I,V+2)*(S(I)-T(I)*S(1)/N)
590 NEXT I
600 R=S(V+2)-S(1)^2/N
610 Z=R-P
620 L=N-V-1
630 I=P/V
640 PRINT
650 I=P/R
660 PRINT "COEFFICIENTE DI ";
661 PRINT "DETERMINAZIONE"
665 PRINT "          (R^2)=";I
670 PRINT "COEFFICIENTE DI CORRELAZIONE"
675 PRINT "MULTIPLA =" ;SQR(I)
680 PRINT "ERRORE STANDARD DI STIMA =" ;
681 PRINT SQR(ABS(Z/L))
690 PRINT
698 REM - STIMA DELLE VARIABILI DIPENDENTI TRAMITE VARIABILI INDIPENDENTI
699 REM - INSERITE
700 PRINT "INTERPOLAZIONE";
701 PRINT "<(INSERIRE 0 PER TERMINARE IL PROGRAMMA)>"
710 P=A(1,V+2)

```

```

720 FOR J=1 TO V
730 PRINT "VARIABILE";J;
740 INPUT X:PRINT"? "X
749 REM - TEST DI FINE PROGRAMMA
750 IF X=0 THEN 810
760 P=P+A(J+1,V+2)*X
770 NEXT J
780 PRINT "VARIABILE DIPENDENTE =" ;P
790 PRINT
799 REM - PREMERE RETURN PER ALTRI DATI
800 GOTO 710
810 END

```

REGRESSIONE DI ORDINE ENNESIMO

Il programma ricava i coefficienti di una equazione di ordine ennesimo, utilizzando il metodo dei minimi quadrati. L'equazione ha la seguente formula:

$$y = c + a_1x + a_2x^2 + \dots a_nx^n$$

dove: y = variabile dipendente

c = costante

$a_1, a_2, \dots a_n$ = coefficienti delle rispettive variabili indipendenti $x, x^2, \dots x^n$.

Verranno stampati i coefficienti dell'equazione, il coefficiente di determinazione, il coefficiente di correlazione e l'errore standard di stima.

È necessario dare le coordinate x e y dei punti conosciuti. Dopo aver ricavato l'equazione, sarà possibile determinare i valori di y relativi a valori dati di x .

La dimensione dello statement alla linea 30 limita il grado dell'equazione. È possibile variare questo limite secondo il seguente schema:

30 DIM A(2·D+1), R(D+1, D+2), T(D+2)

dove D = massimo grado dell'equazione.

Esempio:

La tabella riportata qui sotto fornisce lo spazio di arresto (considerando il tempo di reazione, oltre a quello di frenata) di un'auto a varie velocità. Ricavare dai dati una curva esponenziale e stimare lo spazio di arresto alla velocità di 55 miglia orarie.

miglia orarie
spazio di arresto

20	30	40	50	60	70
54	90	138	206	292	396

```
30 DIM A(5),R(3,4),T(4)  
READY.
```

REGRESSIONE DI ORDINE ENNESIMO

```
GRADO DELL'EQUAZIONE? 2  
NUMERO DI PUNTI NOTI? 6  
X,Y DEL PUNTO 1 ? 20 , 54  
X,Y DEL PUNTO 2 ? 30 , 90  
X,Y DEL PUNTO 3 ? 40 , 138  
X,Y DEL PUNTO 4 ? 50 , 206  
X,Y DEL PUNTO 5 ? 60 , 292  
X,Y DEL PUNTO 6 ? 70 , 396
```

```
                  CONSTANTE = 41.7714472  
1 GRADO DEI COEFFICIENTI =-1.09571524  
2 GRADO DEI COEFFICIENTI = .0878571531
```



```

COEFFICIENTE DI DETERMINAZIONE
(R12) = .999927959
COEFFICIENTE DI CORRELAZIONE = .999963979
ERRORE STANDARD DI STIMA = 1.42094106

INTERPOLAZIONE : (INSERIRE 0 PER TERMINARE)
X= ? 55
Y= 247.274998

X= ? 247.275
Y= 5142.84756

X= ? 0

```

Listing del programma

```

10 PRINT "REGRESSIONE DI ORDINE ENNESIMO"
20 PRINT
28 REM - FISSARE I LIMITI DEL GRADO DELL'EQUAZIONE AD A(2*D+1), R(D+1,D+2),
29 REM - T(D+2) (DOVE D= GRADO MASSIMO DELL'EQUAZIONE)
30 DIM A(13),R(7,8),T(8)
40 PRINT "GRADO DELL'EQUAZIONE";
50 INPUT D
60 PRINT "NUMERO DI PUNTI NOTI";
70 INPUT N
80 A(1)=N
89 REM - INTRODUCI LE COORDINATE DEI PUNTI DATI
90 FOR I=1 TO N
100 PRINT "X,Y DEL PUNTO";I;
110 INPUT X,Y
118 REM - LE LINEE DALLA 120 ALLA 200 RIEMPIONO LE MATRICI CON UN SISTEMA DI
119 REM - EQUAZIONI
120 FOR J=2 TO 2*D+1
130 A(J)=A(J)+X*(J-1)
140 NEXT J
150 FOR K=1 TO D+1
160 R(K,D+2)=T(K)+Y*X*(K-1)
170 T(K)=T(K)+Y*X*(K-1)
180 NEXT K
190 T(D+2)=T(D+2)+Y*(D+1)
200 NEXT I
208 REM - LE LINEE DAL 210 AL 490 RISOLVONO IL SISTEMA DI EQUAZIONI
209 REM - NELLE MATRICI
210 FOR J=1 TO D+1
220 FOR K=1 TO D+1
230 R(K,J)=A(J+K-1)
240 NEXT K
250 NEXT J
260 FOR J=1 TO D+1
270 FOR K=J TO D+1
280 IF R(K,J)<>0 THEN 320
290 NEXTK
300 PRINT "NON ESISTE UNA SOLUZIONE UNICA"
310 GOTO 790
320 FOR I=1 TO D+2
330 S=R(J,I)
340 R(J,I)=R(K,I)
350 R(K,I)=S
360 NEXTI
370 Z=1/R(J,J)
380 FOR I=1 TO D+2
390 R(J,I)=Z*R(J,I)
400 NEXTI

```

```

410 FOR K=1 TO D+1
420 IF K=J THEN 470
430 Z=-R(K,J)
440 FOR I=1 TO D+2
450 R(K,I)=R(K,I)+Z*R(J,I)
460 NEXT I
470 NEXT K
480 NEXT J
490 PRINT
495 PRINT "          COSTANTE =";
496 PRINT R(1,D+2)
499 REM - STAMPA I COEFFICIENTI DELL'EQUAZIONE
500 FOR J=1 TO D
510 PRINT J;"GRADO DEI COEFFICIENTI =";
511 PRINT R(J+1,D+2)
520 NEXT J
530 PRINT
539 REM - CALCOLO DELLA REGRESSIONE
540 P=0
550 FOR J=2 TO D+1
560 P=P+R(J,D+2)*T(J)-A(J)*T(1)/N)
570 NEXT J
580 Q=T(D+2)-T(1)^2/N
590 Z=Q-P
600 I=N-D-1
620 PRINT
630 J=P/Q
640 PRINT "COEFFICIENTE DI ";
641 PRINT "DETERMINAZIONE"
645 PRINT "(R^2) =";J
650 PRINT "COEFFICIENTE DI CORRELAZIONE =";
651 PRINT SQR(J)
660 PRINT "ERRORE STANDARD DI STIMA =";
661 PRINT SQR(Z/I)
670 PRINT
679 REM - CALCOLO DELLE ORDINATE Y IN BASE ALLE ASCISSE X INSERITE
680 PRINT "INTERPOLAZIONE :";
681 PRINT "(INSERIRE 0 PER TERMINARE)"
690 P=R(1,D+2)
700 PRINT "X= ";
710 INPUT X
720 IF X=0 THEN 790
730 FOR J=1 TO D
740 P=P+R(J+1,D+2)*X^J
750 NEXT J
760 PRINT "Y= ";P
770 PRINT
780 GOTO 690
790 END

```

REGRESSIONE GEOMETRICA

Il programma adatta una curva geometrica ad una serie di coordinate, utilizzando il metodo dei minimi quadrati.

Verranno stampati l'equazione, il coefficiente di determinazione, il coefficiente di correlazione e l'errore standard di stima.

È necessario dare le coordinate x e y dei punti conosciuti. Una volta adattata la curva, è possibile determinare il valore di y per un dato valore di x.

Esempio:

La tabella riportata qui sotto fornisce le pressioni di un gas misurate a vari volumi in un esperimento. La relazione tra pressione e volume in un gas è espressa dalla seguente formula:

$$PV^k = C$$

dove: P = pressione

V = volume

C e K = costanti

Questa formula può essere riscritta secondo la forma geometrica standard:

$$P = CV^{-k}$$

Da notare l'esponente negativo, cosa di cui il programma deve tener conto nei calcoli.

Si adatta la curva geometrica ai valori seguenti e si stima la pressione quando il gas ha un volume di 90.

volume	56,1	60,7	73,2	88,3	120,1	187,5
pressione	57,0	51,0	39,2	30,2	19,6	10,5

REGRESSIONE GEOMETRICA

NUMERO DI PUNTI CONOSCIUTI? 6 *

X,Y DEL PUNTO 1 ? 56.1 , 57

X,Y DEL PUNTO 2 ? 60.7 , 51

X,Y DEL PUNTO 3 ? 73.2 , 39.2

X,Y DEL PUNTO 4 ? 88.3 , 30.2

X,Y DEL PUNTO 5 ? 120.1 , 19.6

X,Y DEL PUNTO 6 ? 187.5 , 10.5

F(X) = 16103.7139 * X^{-1.40155091}

COEFFICIENTE DI DETERMINAZIONE

(R²) = .999999206

COEFFICIENTE DI CORRELAZIONE = .999999603

ERRORE STANDARD DI STIMA = 6.37591016E-04

INTERPOLAZIONE: (X=0 PER TERMINARE)

X =? 90

Y = 29.3734983

X =? 0

Listing del programma

```
10 PRINT"REGRESSIONE GEOMETRICA"
20 PRINT
30 PRINT"NUMERO DI PUNTI CONOSCIUTI";
40 INPUT N
50 REM - INSERIRE LE COORDINATE DEI PUNTI DATI
100 FOR I=1 TO N
110 PRINT"X,Y DEL PUNTO";I;
120 INPUT X,Y
129 REM - SI ACCUMULANO I VALORI INTERMEDI
130 Y=LOG(Y)
140 X=LOG(X)
150 J=J+X
160 K=K+Y
170 L=L+X^2
180 M=M+Y^2
190 R2=R2+X*Y
200 NEXT I
209 REM - CALCOLA E STAMPA I COEFFICIENTI DELLE EQUAZIONI
210 B=(N*R2-K*J)/(N*L-J^2)
220 A=(K-B*J)/N
230 PRINT
240 PRINT"F(X) =";EXP(A);"* X^";B
249 REM - CALCOLA LA REGRESSIONE
250 J=B*(R2-J*K/N)
260 M=M-K^2/N
270 K=M-J
280 PRINT
290 R2=J/M
300 PRINT"COEFFICIENTE DI ";
301 PRINT"DETERMINAZIONE"
305 PRINT"(R^2) =";R2
306 PRINT
310 PRINT"COEFFICIENTE DI ";
311 PRINT"CORRELAZIONE =";SQR(R2)
315 PRINT
320 PRINT"ERRORE STANDARD DI ";
325 PRINT"STIMA = ";SQR(K/(N-2))
326 PRINT
330 PRINT"
339 REM - STIMA LE ORDINATE Y IN BASE ALLE ASCISSE X
340 PRINT"INTERPOLAZIONE: ";
341 PRINT"(X=0 PER TERMINARE)"
350 PRINT"X =";
360 INPUT X
369 REM - IL PROGRAMMA RIPARTE O TERMINA? SI RICHIEDE LA RISPOSTA DELL'UTENTE
370 IF X=0 THEN 410
380 PRINT"Y =";EXP(A)*X^B
390 PRINT
399 REM - RIPARTIRE PER ALTRI DATI
400 GOTO 350
410 END
```

REGRESSIONE ESPONENZIALE

Il programma ricava i coefficienti di una equazione per una curva esponenziale. L'equazione ha la seguente forma:

$$f(x) = ae^{bx}$$

dove: a e b sono i coefficienti calcolati.

Verranno stampati i coefficienti dell'equazione, il coefficiente di determinazione, il coefficiente di correlazione e l'errore standard di stima.

È necessario dare le coordinate x e y dei punti conosciuti. Una volta adattata la curva, è possibile determinare il valore di y per un dato valore di x.

Esempio:

La tabella riportata qui sotto fornisce il numero di batteri presenti in una coltura in determinati istanti.

Si adatta a questi valori una curva esponenziale e si stima il numero di batteri presenti dopo sette ore.

numero di ore	0	1	2	3	4	5	6
numero di batteri	25	38	58	89	135	206	315

REGRESSIONE ESPONENZIALE

NUMERO DI PUNTI CONOSCIUTI? 7

X,Y DEL PUNTO 1 ? 0 , 25
X,Y DEL PUNTO 2 ? 1 , 38
X,Y DEL PUNTO 3 ? 2 , 58
X,Y DEL PUNTO 4 ? 3 , 89
X,Y DEL PUNTO 5 ? 4 , 135
X,Y DEL PUNTO 6 ? 5 , 206
X,Y DEL PUNTO 7 ? 6 , 315

A = 24.9616634
B = .422375081

COEFFICIENTE DI DETERMINAZIONE
(R12) = .999993572

COEFFICIENTE DI CORRELAZIONE = .999996786

ERRORE STANDARD DI STIMA = 2.53424554E-03

INTERPOLAZIONE: (X=0 PER TERMINARE)

X=? 7
Y = 480.086716

X=? 0

Listing del programma

```
10 PRINT"REGRESSIONE ESPONENZIALE"
20 PRINT
30 PRINT"NUMERO DI PUNTI CONOSCIUTI";
40 INPUT N
50 J=0
60 K=0
70 L=0
80 M=0
90 R2=0
99 REM - INSERIRE LE COORDINATE DEI PUNTI DATI
100 FOR I=1 TO N
110 PRINT"X,Y DEL PUNTO";I;
120 INPUT X,Y
129 REM - SI ACCUMULANO I VALORI INTERMEDI
130 Y=LOG(Y)
140 J=J+X
150 K=K+Y
160 L=L+X^2
170 M=M+Y^2
180 R2=R2+X*Y
190 NEXT I
199 REM - CALCOLA E STAMPA I COEFFICIENTI DELLE EQUAZIONI
200 B=(N*R2-K*J)/(N*L-J^2)
210 A=(K-B*J)/N
220 PRINT
230 PRINT"A =";EXP(A)
240 PRINT"B =";B
249 REM - CALCOLA LA REGRESSIONE
250 J=B*(R2-J*K/N)
260 M=M-K^2/N
270 K=M-J
280 PRINT
290 R2=J/M
300 PRINT"COEFFICIENTE DI ";
301 PRINT"DETERMINAZIONE"
305 PRINT"(R^2) =";R2
306 PRINT
310 PRINT"COEFFICIENTE DI ";
311 PRINT"CORRELAZIONE =";SQR(R2)
315 PRINT
320 PRINT"ERRORE STANDARD DI ";
325 PRINT"STIMA = ";SQR(K/(N-2))
326 PRINT
330 PRINT"
339 REM - STIMA LE ORDINATE Y IN BASE ALLE ASCISSE X
340 PRINT"INTERPOLAZIONE: ";
341 PRINT"(X=0 PER TERMINARE)"
350 PRINT"X =";
360 INPUT X
369 REM - IL PROGRAMMA RIPARTE O TERMINA? SI RICHIEDE LA RISPOSTA DELL'UTENTE
370 IF X=0 THEN 410
380 PRINT"Y =";EXP(A)*EXP(B*X)
390 PRINT
399 REM - RIPARTIRE PER ALTRI DATI
400 GOTO 350
410 END
```

AFFIDABILITA' DI UN SISTEMA

Questo programma calcola l'affidabilità di un sistema operativo soggetto a logoramento e la percentuale di insuccesso di ogni componente.

Esempio:

Determinare l'affidabilità di un sistema operativo computerizzato con componenti riportati nell'elenco sottostante.

	tempo di azione (h)	guasto
CPU	15.000	0,00020
terminale	3.000	0,00010
disco	3.000	0,00015
stampante	1.500	0,00015

AFFIDABILITA' DI UN SISTEMA

(PER TERMINARE IL PROGRAMMA INSERIRE 0)
TEMPO OPERATIVO IN ORE? 1000
NUMERO DI COMPONENTI? 4

COMPONENTE 1
TEMPO MEDIO DI LOGORAMENTO? 15000
PERCENTUALE MEDIA DI MALFUNZIONAMENTI
? 2E-04

COMPONENTE 2
TEMPO MEDIO DI LOGORAMENTO? 3000
PERCENTUALE MEDIA DI MALFUNZIONAMENTI
? 1E-04

COMPONENTE 3
TEMPO MEDIO DI LOGORAMENTO? 3000
PERCENTUALE MEDIA DI MALFUNZIONAMENTI
? 1.5E-04

COMPONENTE 4
TEMPO MEDIO DI LOGORAMENTO? 1500
PERCENTUALE MEDIA DI MALFUNZIONAMENTI
? 1.5E-04

AFFIDABILITA' DEL SISTEMA = .135335283

TEMPO OPERATIVO IN ORE? 0

Listing del programma

```
10 PRINT "AFFIDABILITA' DI UN SISTEMA"
20 PRINT
30 PRINT "(PER TERMINARE IL PROGRAMMA INSERIRE 0)"
40 PRINT "TEMPO OPERATIVO IN ORE";
50 INPUT T
59 REM - TEST PER FINE PROGRAMMA
60 IF T=0 THEN 230
70 PRINT "NUMERO DI COMPONENTI";
80 INPUT N
90 Z=0
99 REM -INSERIRE I DATI PER OGNI COMPONENTI
100 FOR I=1 TO N
105 PRINT
110 PRINT "COMPONENTE";I
120 PRINT "TEMPO MEDIO DI LOGORAMENTO";
130 INPUT W
140 PRINT "PERCENTUALE MEDIA DI MALFUNZIONAMENTI"
150 INPUT F
159 REM - SI INCLUDE OGNI COMPONENTE NELL'AFFIDABILITA
160 Z=Z+1/W+F
170 NEXT I
180 PRINT
189 REM - CALCOLO DELL'AFFIDABILITA', STAMPA
190 Z=EXP(-Z*T)
200 PRINT "AFFIDABILITA' DEL SISTEMA =" ;Z
210 PRINT
219 REM - IL PROGRAMMA RIPARTE
220 GOTO 40
230 END
```


TASSO MEDIO DI CRESCITA, PROIEZIONI

Questo programma calcola il tasso medio di crescita di una ditta, successivamente determina proiezioni per il futuro. Il tasso medio e le proiezioni possono essere calcolati per ogni aspetto della ditta, come le vendite, i profili, il numero di impiegati o il numero di clienti. È necessario avere a disposizione precisi diagrammi relativi ad un certo numero di anni precedenti.


La dimensione dello statement 30 limita il numero dei diagrammi precedenti che si vogliono inserire. Questo numero può essere variato secondo il seguente schema:

30 DIM S(N)

dove N = numero degli anni i cui diagrammi sono noti.

Esempio:

I dati relativi ai prestiti di libri della Claremount County Library sono tabulati nel grafico riportato qui sotto. Qual'è il suo tasso medio di crescita? E quanti libri ci si può aspettare che darà in prestito nel suo decimo e nel suo ventesimo anno di servizio?



TASSO MEDIO DI CRESCITA, PROIEZIONI

NUMERO DI ANNI DIAGRAMMATI STABILITI? 9

DIAGRAMMA: ANNO 1 ? 26

ANNO 2 ? 35

ANNO 3 ? 42

```

        ANNO 4 ? 45
        ANNO 5 ? 41
        ANNO 6 ? 51
        ANNO 7 ? 60
        ANNO 8 ? 62
        ANNO 9 ? 74
TASSO MEDIO DI CRESCITA = 11.88 %
<INSERIRE 0 PER TERMINARE>
VENDITE PREVISTE PER L'ANNO? 10
                                = 81.29
VENDITE PREVISTE PER L'ANNO? 20
                                = 249.88
VENDITE PREVISTE PER L'ANNO? 0

```

Listing del programma

```

10 PRINT" TASSO MEDIO DI CRESCITA,";
15 PRINT" PROIEZIONI"
20 PRINT
28 REM - SI ASSEGNA LA DIMENSIONE DELLA MATRICE S IN BASE AL NUMERO DI ANNI
29 REM - DI CUI SI CONOSCONO I DATI
30 DIM S(20)
40 PRINT"NUMERO DI ANNI DIAGRAMMATI";
45 PRINT" STABILITI";
50 INPUT N
60 FOR I=1 TO N
70 IF I>1 THEN 100
80 PRINT"DIAGRAMMA: ANNO";I;
90 GOTO 110
100 PRINT"          ANNO";I;
110 INPUT S(I)
120 NEXT I
129 REM - INIZIALIZZA LE VARIABILI PER IL PRIMO ANNO
130 T=LOG(S(1))
140 V=0
149 REM - CICLO PER GLI ANNI RIMANENTI DEL PERIODO
150 FOR I=2 TO N
160 L=LOG(S(I))
170 T=T+L
180 V=V+(I-1)*L
190 NEXT I
199 REM - SI CALCOLO IL TASSO MEDIO DI CRESCITA
200 A=6*(2*V/(N-1)-T)/(N)/(N+1)
210 G=EXP(A)-1
219 REM - ARROTONDA, STAMPA
220 PRINT "TASSO MEDIO DI CRESCITA =";
225 PRINT INT(G*10000+.5)/100;"%"
230 PRINT
239 REM - SI CALCOLA IL FATTORE ANNUO DEL TASSO DI CRESCITA
240 S=EXP(T/N-A*(N-1)/2)
250 PRINT"<INSERIRE 0 PER TERMINARE>"
259 REM - INSERIRE IL NUMERO DI ANNI
260 PRINT "VENDITE PREVISTE PER L'ANNO";
270 INPUT Y1
279 REM - TEST PER FINE PROGRAMMA
280 IF Y1=0 THEN 320
289 REM - CALCOLA IL DIAGRAMMA DELLE VENDITE PREVISTE
290 S1=S*(1+G)^(Y1-1)
299 REM - ARROTONDAMENTO, STAMPA
300 PRINTTAB(28);"=";
305 PRINTINT(S1*100+.5)/100
309 REM - RIPARTE PER NUOVI DATI
310 GOTO 260
320 END

```

TABELLA DI AMMORTAMENTO DI UN ONERE

Questo programma mette in tabulato l'ammontare annuo dell'ammortamento. È possibile usare il metodo della somma delle cifre o qualunque metodo con il saggio percentualmente decrescente. È necessario conoscere il prezzo dell'acquisto (valore iniziale), il valore dell'indennità di recupero al termine del periodo di ammortamento e la vita dell'articolo che si desidera ammortizzare. Per poter sviluppare il problema con il metodo dell'ammortamento con il saggio percentualmente calante, è necessario conoscere il metodo percentuale.

Esempi:

La Miracle Corporation ha messo un nuovo tetto in un capannone di sua proprietà per una spesa di \$ 27.000. La ditta ritiene di doverlo sostituire tra nove anni. Quale sarà l'ammortamento annuo utilizzando la somma delle cifre?

Una banca edifica una nuova sede di uffici spendendo 1,2 milioni di dollari. Determinare la tabella di ammortamento dell'onere, utilizzando una percentuale di saggio decrescente del 150% con 30 anni di vita. Si presupponga un valore di indennità di recupero di 250.000 dollari.

TABELLA DI AMMORTAMENTO DI UN ONERE

PREZZO DI ACQUISTO? 27000
VALORE RISPARMIATO? 0
DURATA IN ANNI? 9
INSERIRE 1 PER SOMMA DI CIFRE,
2 PER BILANCIO IN DIMINUZIONE? 1
TASSA DI AMMORTAMENTO 'PER SOMMA
DI CIFRE'

PREZZO \$ 27000
VALORE RISPARMIATO \$ 0
AMMORTAMENTO NETTO \$ 27000
DURATA 9 ANNI .

ANNO	AMMORTAMENTO	BILANCIO
1	5400	21600
2	4800	16800
3	4200	12600
4	3600	9000
5	3000	6000
6	2400	3600
7	1800	1800
8	1200	600
9	600	0

```

ALTRI DATI (1=SI, 0=NO)? 1
PREZZO DI ACQUISTO? 1200000
VALORE RISPARMIATO? 250000
DURATA IN ANNI? 30
INSERIRE 1 PER SOMMA DI CIFRE,
2 PER BILANCIO IN DIMINUZIONE? 2
METODO IN %? 150
    TASSA DI AMMORTAMENTO 'PER
    BILANCIO IN DIMINUZIONE'
    PREZZO L. 1200000
    VALORE RISPARMIATO L. 250000
    AMMORTAMENTO NETTO L. 950000
    DURATA 30 ANNI
    METODO 150 %

```

ANNO	AMMORTAMENTO	BILANCIO
1	47500	902500
2	45125	857375
3	42868.75	814506.25
4	40725.31	773780.94
5	38689.05	735091.89
6	36754.59	698337.3
7	34916.86	663420.44
8	33171.02	630249.42
9	31512.47	598736.95
10	29936.85	568800.1
11	28440	540360.1
12	27018	513342.1
13	25667.1	487675
14	24383.75	463291.25
15	23164.56	440126.69
16	22006.33	418120.36
17	20906.02	397214.34
18	19860.72	377353.62
19	18867.68	358485.94
20	17924.3	340561.64
21	17028.08	323533.56
22	16176.68	307356.88
23	15367.84	291989.04
24	14599.45	277389.59
25	13869.48	263520.11
26	13176.01	250344.1
27	12517.2	237826.9
28	11891.34	225935.56
29	11296.78	214638.78
30	10731.94	203906.84

```

ALTRI DATI (1=SI, 0=NO)? 0

```

Listing del programma

```

10 PRINT "TABELLA DI AMMORTAMENTO DI UN ONERE "
20 PRINT
29 REM - INSERIRE IL VALORE INIZIALE E ARROTONDARE AL CENTESIMO
30 PRINT "PREZZO DI ACQUISTO";
40 INPUT V
50 V=INT(V*100+.5)/100
59 REM - INSERIRE IL VALORE FINALE E ARROTONDARE AL CENTESIMO
60 PRINT "VALORE RISPARMIATO";
70 INPUT S

```

```

80 S=INT(S*100+.5)/100
89 REM - CALCOLA LA SOMMA DA AMMORTARE
90 D=V-S
99 REM - INSERIRE DURATA AMMORTAMENTO
100 PRINT "DURATA IN ANNI";
110 INPUT Y
119 REM - SCEGLIERE IL METODO DI AMMORTAMENTO
120 PRINT "INSERIRE 1 PER SOMMA DI CIFRE,"
125 PRINT "2 PER BILANCIO IN DIMINUZIONE";
130 INPUT X
140 IF X=2 THEN 450
150 IF X<>1 THEN 120
158 REM - METODO 'PER SOMMA DI CIFRE'
159 REM - R1 E' IL TOTALE AMMORTATO CUMULATIVO
160 R1=0
170 N=1
260 PRINT "TASSA DI AMMORTAMENTO 'PER SOMMA"
265 PRINT "DI CIFRE'"
270 PRINT TAB(17);"PREZZO $";V
280 PRINT TAB(5);"VALORE RISPARMIATO $";S
290 PRINT TAB(5);"AMMORTAMENTO NETTO $";
291 PRINT V-S
300 PRINT TAB(13);"DURATA";Y;"ANNI"
310 PRINT
320 PRINT "ANNO AMMORTAMENTO";
321 PRINT " BILANCIO"
322 N=11
325 FOR I=1 TO Y
326 REM - CONTROLLO DI RIEMPIMENTO SCHERMO(23 RIGHE)
327 IF N/20=INT(N/20) THEN GOSUB 800
329 REM - CALCOLA L'AMMORTAMENTO ED ARROTONDA AL CENTESIMO
330 R=2*D*(Y-I+1)/((Y+1)*Y)
340 R=INT(R*100+.5)/100
349 REM - ACCUMULA L'AMMORTAMENTO
350 R1=R1+R
359 REM - CALCOLA IL BILANCIO DA AMMORTARE
360 B=D-R1
369 REM - TEST PER AMMORTAMENTO COMPLETATO
370 IF B>=0 THEN 410
380 R1=R1+R
400 B=0
410 PRINT I,RTAB(22)B
420 N=N+1
430 NEXT I
440 GOTO 700
448 REM - METODO 'PER BILANCIO IN DIMINUZIONE'
449 REM - INSERIRE IL TASSO DI DIMINUZIONE PERCENTUALE DEL BILANCIO
450 PRINT "METODO IN %";
460 INPUT M:PRINT"?M"
469 REM - CONVERTE LE PERCENTUALI IN DECIMALI
470 M=M/100
479 REM - N CONTA LE LINEE STAMPATE IN OGNI PAGINA
489 REM - R E' IL TOTALE DA AMMORTARE
490 R=D
570 PRINT TAB(8);"TASSA DI AMMORTAMENTO 'PER"
575 PRINTTAB(12);"BILANCIO IN DIMINUZIONE'"
580 PRINT TAB(17);"PREZZO L.";V
590 PRINT TAB(5);"VALORE RISPARMIATO L.";S
600 PRINT TAB(5);"AMMORTAMENTO NETTO L.";
601 PRINT V-S
610 PRINT TAB(13);"DURATA";Y;"ANNI"
620 PRINTTAB(13);"METODO ";M*100;"%"
630 PRINT
640 PRINT "ANNO AMMORTAMENTO";
641 PRINT " BILANCIO"
642 PRINT
643 N=11
645 FOR I=1 TO Y
649 REM - CALCOLA L'AMMORTAMENTO E ARROTONDA AL CENTESIMO

```


STAMPA DI ASSEGNI

Questo programma stampa un assegno. È necessario inserire data, ammontare e beneficiario dell'assegno. Il programma traduce la data e l'ammontare in parole e li stampa, posizionandoli nella maniera voluta, sull'assegno.

Si consideri il programma listato nel seguito come un esempio di programma che scrive assegni. Gli assegni generalmente hanno infatti una spaziatura diversa da quella prevista dal programma seguente. Il metodo per tradurre le cifre in parole è generalmente applicabile. La spaziatura può essere variata in funzione del tipo di assegno da stampare.

I comandi per aprire o chiudere il flusso dei dati alla stampante sono situati rispettivamente agli statements numero 250 e 625. Lo statement PRINT tra 250 e 625 non ha bisogno di essere sostituito con PRINT #, quando al 250 si usa il comando CMD.

Quando il programma chiede: PRONTO A STAMPARE L'ASSEGNO?, ciò significa che esso sollecita l'utente ad inserire un assegno in bianco nel dispositivo stampante. L'assegno sarà posizionato una linea più in su rispetto alla linea in cui la data deve essere stampata.

Quando l'assegno è stato inserito, è sufficiente premere il tasto RETURN (non è necessario nessun altro dato in entrata), e l'assegno verrà stampato.

Esempio:

Tra gli assegni che la Miracle Corporation deve scrivere; uno, per Jackson, è di L. 4.975.890, un altro, per Freida Alexander, è di L. 103.750. Stampare gli assegni usando il computer.

STAMPA DI ASSEGNI

```
DATA (GGMMAA)? 30877
PER FINIRE IL PROGRAMMA INTRODUCI 'FINE'
NOME DEL BENEFICIARIO? GRUPPO EDITORIALE
COGNOME DEL BENEFICIARIO? JACKSON
AMMONTARE DELL'ASSEGNO? 4957500
SIETE PRONTI A STAMPARE L'ASSEGNO
```

HEAVENLY BANK
EMERYVILLE OFFICE
4120 ASHBY AVENUE
EMERYVILLE, CA 94601

NO. 328

3 LUGLIO, 1980

L. ~~4975890~~
4975890

PAY TO THE ORDER OF GRUPPO EDITORIALE JACKSON

QUATTRO MILIONI NOVECENTOSETTANTACINQUEMILA OTTOCENTONOVANTA LIRE

MIRACLE CORPORATION
1111 COUNTRY ROAD
COUNTRYVILLE, CA 94132

1328252158

NOME DEL BENEFICIARIO? FREIDA
COGNOME DEL BENEFICIARIO? ALEXANDER
AMMONTARE DELL'ASSEGNO? 103750
SIETE PRONTI A STAMPARE L'ASSEGNO

HEAVENLY BANK
EMERYVILLE OFFICE
4120 ASHBY AVENUE
EMERYVILLE, CA 94601

NO. 382

3 LUGLIO, 1980

L. ~~103750~~
103750

PAY TO THE ORDER OF FREIDA ALEXANDER

CENTOTREMILA SETTECENTOCINQUANTA LIRE

MIRACLE CORPORATION
1111 COUNTRY ROAD
COUNTRYVILLE, CA 94132

1328252158

NOME DEL BENEFICIARIO? FINE

Listing del programma

```
10 PRINT "D STAMPA DI ASSEGNI"  
20 PRINT  
25 DIMA$(27) DIMM$(12)  
30 DATA "UNO", "DUE", "TRE", "QUATTRO"  
31 DATA "CINQUE", "SEI", "SETTE", "OTTO"  
40 DATA "NOVE", "DIECI", "UNDICI", "DODICI"  
41 DATA "TREDICI", "QUATTORDICI"  
42 DATA "QUINDICI", "SEDICI"  
50 DATA "DICIASSETTE", "DICIOTTO"
```



```

51 DATA"DICIANNOVE", "VENTI", "TRENTA"
52 DATA"QUARANTA", "CINQUANTA", "SESSANTA"
53 DATA"SETTANTA", "OTTANTA", "NOVANTA"
54 DATA"GENNAIO", "FEBBRAIO", "MARZO"
55 DATA"APRILE", "MAGGIO", "GIUGNO", "LUGLIO"
56 DATA"AGOSTO", "SETTEMBRE", "OTTOBRE"
57 DATA"NOVEMBRE", "DICEMBRE"
58 FOR I=1 TO 27: READ A$(I): NEXT I: FOR I=1 TO 12: READ M$(I): NEXT I
59 REM - INSERIRE LA DATA SENZA LE VIRGOLE. GIORNI E ANNI DEVONO CONTENERE
60 REM - DUE CIFRE
61 PRINT "DATA (GGMMAA)";
62 INPUT D
63 PRINT "PER FINIRE IL PROGRAMMA";
64 PRINT "INTRODUCI FINE"
65 PRINT "COGNOME DEL BENEFICIARIO";
66 INPUT F$
67 REM - FINE PROGRAMMA?
68 IF F$="FINE" THEN 790
69 PRINT "COGNOME DEL BENEFICIARIO";
70 INPUT L$
71 PRINT "AMMONTARE DELL'ASSEGNO";
72 INPUT A:PRINT
73 REM - INSERIRE UNA ASSEGNO IN BIANCO NELLA STAMPANTE, E PREMERE UN TASTO
74 REM - QUALSIASI, QUANDO SIETE PRONTI
75 PRINT "SIETE PRONTI A STAMPARE L'ASSEGNO";
76 GET Z$: IF Z$="" THEN GOTO 200
77 PRINT
78 PRINT
79 PRINT
80 PRINT
81 REM - SPEZZA IL NUMERO DELLA DATA INTRODOTTA IN GIORNO, MESE, ANNO
82 D2=INT(D/10000)
83 D1=INT((D-D2*10000)/100)
84 D3=INT(D-(D2*100+D1)*100)
85 D3=D3+1900
86 REM - RAGGIUNGE IL MESE OPPORTUNO NELLA TABELLA DATI
87 X$=M$(D1)
88 OPEN 4,4:CMD4
89 REM - STAMPA LA DATA
90 PRINT TAB(20);D2;X$;";";D3
91 PRINT
92 REM - STAMPA L'AMMONTARE DUE VOLTE, LA PRIMA VOLTA NEL RETTANGOLO SCURO
93 PRINT TAB(24);"L.3";A;"■"
94 PRINT TAB(26);A
95 PRINT
96 PRINT F$;" ";L$
97 PRINT
98 REM - L'AMMONTARE DELL'ASSEGNO E' CORRETTO ?
99 IF A<=0 THEN 770
100 A1=A
101 REM - L'AMMONTARE E' IN MILIONI ?
102 N1=INT(A1/1E6)
103 N3=N1
104 REM - NON SI POSSONO STAMPARE ASSEGNI SUPERIORI A L. 999.999.999
105 IF N1>99 THEN A2=N1:N3=N1:GOSUB 1449:N1=A2:IF N1=0 THEN 380
106 IF N1=1 THEN PRINT "UN MILIONE ";GOTO 390
107 IF N1=0 THEN 390
108 GOSUB 640
109 PRINT " MILIONI "
110 A1=A1-N3*1E6
111 REM - L'AMMONTARE E' IN MIGLIAIA ?
112 N1=INT(A1/1E3)
113 N3=N1
114 IF N1=0 THEN 440
115 IF N1>99 THEN A2=N1:N3=N1:GOSUB 1449:N1=A2
116 IF N1=1 THEN PRINT "MILLE ";GOTO 440
117 GOSUB 640
118 PRINT "MILA "
119 A1=A1-N3*1E3
120 REM - L'AMMONTARE E' IN CENTINAIA?

```

```

450 A2=A1
460 GOSUB 1449
470 A1=A2
519 REM - E LIRE?
520 IF A1<1 THEN 570
550 A1=INT(A1*100+.5)/100
555 N1=A1 : GOSUB 640
559 REM - LE LIRE SONO STAMPATE IN FORMA NUMERICA
560 PRINT " LIRE"
569 REM - SPAZIO PER L'USCITA DALL'ASSEGNO
570 PRINT
580 PRINT
590 PRINT
600 PRINT
610 PRINT
620 PRINT
625 REM ?#4:CLOSE4
630 RESTORE:GOTO 120
639 REM - SOTTOPROGRAMMA PER OTTENERE LA TRASCRIZIONE DELLA SOMMA IN LETTERE
640 IF N1<21 THEN 730
650 RESTORE
651 X#=A#((N1-20)/10+20)
670 PRINT X#;
680 A3=N1-INT(N1/10)*10
690 IF A3=0 THEN 760
691 IF A3=1 OR A3=8 THEN PRINT"III";
711 X#=A#(A3)
720 GOTO 750
730 REM
731 X#=A#(N1)
750 PRINT X#;
759 REM - FINE DEL SOTTOPROGRAMMA
760 RETURN
770 PRINT, "*****VUOTO*****"
780 GOTO 570
790 END
1449 REM
1450 N2=INT(A2/100)
1455 IF N2=1 THEN 1500
1460 IF N2=0 THEN 1519
1490 N1=N2:GOSUB 640
1500 PRINT "CENTO";
1510 A2=A2-N2*100
1519 RETURN

```

COSTO DI UNA RICETTA

Questo programma calcola il costo ed il costo per servizio di una ricetta. Per ogni ingrediente è necessario dare il prezzo dell'acquisto, l'ammontare dell'acquisto, la quantità usata nella ricetta e il numero di unità di ricette per unità di acquisto.

Esempio:

Qui sotto è riportata una ricetta per un dolce alla fragola. Si calcolino il costo della ricetta e il costo per servizio. Quale sarà il costo per servizio, se il dolce è di dodici porzioni? Vengono forniti i fattori di conversione e i prezzi degli ingredienti.

Dolce alla fragola ————— 8 porzioni

3 tazze di farina	2,5 tazze/libbra	\$ 1.59	5 libbre
3+ 1/4 cucchiaini da tè di lievito in polvere	15 cucchiaini/oncia	0,43	4 once
1/4 tazza di zucchero	2 tazze/libbra	1,24	5 libbre
1+ 1/4 cucchiaini da tè di sale	6 cucchiaini/oncia	0,29	1 libbra
1/2 tazza di burro	2 tazze/libbra	1,49	1 libbra
1 uovo	12 dozzina	0,75	1 dozzina
2/3 di tazza di latte	4 tazze quart	0,40	1 quart
3 porzioni di fragole	—————	0,49	1 porzione
1/2 porzione di panna montata	—————	0,59	1/2 porzione.

COSTO DI UNA RICETTA

NUMERO DI INGREDIENTI? 9

INGREDIENTE 1 :

COSTO UNITARIO PER QUANTITA' IN VENDITA? 1.59

NUMERO DI UNITA' IN VENDITA? 5

UNITA' DI MISURA PER LA RICETTA? 2.5

UNITA' DI MISURA RICHIESTE

PER LA RICETTA

? 3

INGREDIENTE 2 :

COSTO UNITARIO PER QUANTITA' IN VENDITA? .43

NUMERO DI UNITA' IN VENDITA? 4

UNITA' DI MISURA PER LA RICETTA? 15

UNITA' DI MISURA RICHIESTE

PER LA RICETTA

? 3.25

INGREDIENTE 3 :

COSTO UNITARIO PER QUANTITA' IN VENDITA? 1.24

NUMERO DI UNITA' IN VENDITA? 5

UNITA' DI MISURA PER LA RICETTA? 2

UNITA' DI MISURA RICHIESTE

PER LA RICETTA

? .25

INGREDIENTE 4 :
COSTO UNITARIO PER QUANTITA' IN VENDITA? .29
NUMERO DI UNITA' IN VENDITA? 1
UNITA' DI MISURA PER LA RICETTA? 96
UNITA' DI MISURA RICHIESTE
PER LA RICETTA
? 1.25

INGREDIENTE 5 :
COSTO UNITARIO PER QUANTITA' IN VENDITA? 1.49
NUMERO DI UNITA' IN VENDITA? 1
UNITA' DI MISURA PER LA RICETTA? 2
UNITA' DI MISURA RICHIESTE
PER LA RICETTA
? .5

INGREDIENTE 6 :
COSTO UNITARIO PER QUANTITA' IN VENDITA? .75
NUMERO DI UNITA' IN VENDITA? 1
UNITA' DI MISURA PER LA RICETTA? 12
UNITA' DI MISURA RICHIESTE
PER LA RICETTA
? 1

INGREDIENTE 7 :
COSTO UNITARIO PER QUANTITA' IN VENDITA? .4
NUMERO DI UNITA' IN VENDITA? 1
UNITA' DI MISURA PER LA RICETTA? 4
UNITA' DI MISURA RICHIESTE
PER LA RICETTA
? .6666667

INGREDIENTE 8 :
COSTO UNITARIO PER QUANTITA' IN VENDITA? .49
NUMERO DI UNITA' IN VENDITA? 1
UNITA' DI MISURA PER LA RICETTA? 1
UNITA' DI MISURA RICHIESTE
PER LA RICETTA
? 3

INGREDIENTE 9 :
COSTO UNITARIO PER QUANTITA' IN VENDITA? .59
NUMERO DI UNITA' IN VENDITA? 1
UNITA' DI MISURA PER LA RICETTA? 1
UNITA' DI MISURA RICHIESTE
PER LA RICETTA
? 1

NUMERO DI PORZIONI? 8

COSTO TOTALE PER UNA RICETTA =\$ 3
COSTO PER PORZIONE =\$.38

PIU' O MENO PORZIONI(1=SI, 0=NO)? 1
NUMERO DI PORZIONI? 12

COSTO TOTALE PER UNA RICETTA =\$ 3
COSTO PER PORZIONE =\$.25

PIU' O MENO PORZIONI(1=SI, 0=NO)? 0

Listing del programma

```
10 PRINT"$$$COSTO DI UNA RICETTA"  
15 PRINT
```

```

18 REM - LE ISTRUZIONI DALLA RIGA 30 ALLA RIGA 180 RICHIEDONO L'INTRODUZIONE
19 REM - DI DATI DA PARTE DELL'UTENTE
20 PRINT"NUMERO DI INGREDIENTI";
30 INPUT N:PRINT"?";N
39 REM - CICLO PER LA RICHIESTA DI DATI PER OGNI INGREDIENTE
40 INPUT N
50 PRINT
60 PRINT "INGREDIENTE";I;":"
70 PRINT "COSTO UNITARIO PER QUANTITA' IN VENDITA";
80 INPUT C
90 PRINT "NUMERO DI UNITA' IN VENDITA";
100 INPUT U
110 PRINT "UNITA' DI MISURA ";
111 PRINT "PER LA RICETTA";
120 INPUT F
130 PRINT "UNITA' DI MISURA RICHIESTE"
131 PRINT "PER LA RICETTA"
140 INPUT R
149 REM - COSTO TOTALE DI OGNI INGREDIENTE PER LA QUANTITA' USATA
150 P=P+C/U/F*R
160 NEXT I
165 PRINT
170 PRINT "NUMERO DI PORZIONI";
180 INPUT S
190 PRINT
199 REM - ARROTONDAMENTO DEL COSTO AL CENTESIMO E STAMPA DEL RISULTATO
200 PRINT "COSTO TOTALE PER ";
201 PRINT "UNA RICETTA =# ";
205 PRINT INT(P*100+.5)/100
210 PRINT "COSTO PER PORZIONE =# ";
215 PRINT INT(P/S*100+.5)/100
220 PRINT
229 REM - CALCOLO DI ALTRI PREZZI PER PORZIONE
230 PRINT "PIU' O MENO PORZIONI";
231 PRINT "<1=SI, 0=NO>";
240 INPUT N
250 IF N=1 THEN 170
260 END

```

OPZIONE

Dopo essersi familiarizzati con le operazioni di questo programma, è possibile abbreviarlo, inserendo le informazioni richieste per ogni ingrediente in un'unica linea. I necessari cambiamenti di programma sono elencati nell'esempio seguente.

Esempio:

Calcolare il costo per porzione del dolce alla fragola descritto nell'esercizio precedente, nel caso questo sia servito senza crema.

```

COSTO DI UNA RICETTA
NUMERO DI INGREDIENTI? 3
INGREDIENTE 1 :
? 1.59 / 5 / 2.5 / 3

```

```

INGREDIENTE 2 :
? .43 , 4 , 15 , 3.25

INGREDIENTE 3 :
? 1.24 , 5 , 2 , .25

INGREDIENTE 4 :
? .29 , 1 , 96 , 1.25

INGREDIENTE 5 :
? 1.49 , 1 , 2 , .5

INGREDIENTE 6 :
? .75 , 1 , 12 , 1

INGREDIENTE 7
? .4 , 1 , 4 , .6666667

INGREDIENTE 8
? .49 , 1 , 1 , 3

NUMERO-DI PORZIONI? 8

COSTO TOTALE PER UNA RICETTA =$ 2.41
COSTO PER PORZIONE =$ .3

PIU' O MENO PORZIONI(1=SI, 0=NO)? 12

```

Opzione

```

1 REM OPZIONE 55-70
10 PRINT"COSTO DI UNA RICETTA"
:
:
50 PRINT
55 REM - INSERIRE C,U,F,R
56 REM - DOVE C=COSTO UNITARIO PER QUANTITA' IN VENDITA
57 REM -      U=NUMERO DI UNITA' IN VENDITA
58 REM -      F=UNITA' DI MISURA PER LA RICETTA
59 REM -      R=UNITA' DI MISURA RICHIESTE PER LA RICETTA
60 PRINT "INGREDIENTE";I;":"
70 INPUT C,U,F,R
149 REM - COSTO TOTALE DI OGNI INGREDIENTE PER LA QUANTITA' USATA
:
:
260 END


```

CONTROLLO O RILEVAMENTO DI MAPPA

Dovuto alla cortesia di: Robert Irving; Northridge California

Questo programma calcola l'errore di chiusura e l'area di un appezzamento di terreno del quale è accessibile la poligonale del perimetro. Il programma calcola anche quanto a nord e a est è lontana la fine di una poligonale aperta rispetto alle sue origini (distanza percorsa verso il nord e distanza percorsa verso l'est). Le coordinate dell'origine possono essere inserite per una poligonale aperta. I valori negativi delle distanze verso nord e verso est indicano il sud e l'ovest, rispetto alle coordinate 0,0 dell'origine del rilevamento.

I singoli lati della poligonale possono essere sia segmenti che archi di circonferenze. Per calcolare la poligonale, è necessario conoscere la direzione e la lunghezza di ogni lato. Per ciascun lato curvo, si debbono conoscere il raggio, la direzione della corda e la sua lunghezza (o il raggio, la misura dell'arco e la tangente).




Schema per la numerazione dei quadranti. Gli angoli sono misurati partendo dalla linea NORD SUD.

Per un rilevamento chiuso, si prende ogni intersezione dei lati come punto di partenza, e il numero delle linee e degli archi, iniziando con uno, in senso orario intorno al perimetro. Se qualche arco è uguale o superiore a 180 gradi, lo si deve dividere in archi più piccoli, ciascuno inferiore ai 180 gradi.

Per convenzione i topografi misurano le direzioni dall'asse nord-sud. Questa convenzione fu stabilita prima dell'uso dei computers, così questa funzione trigonometrica può essere rilevata facilmente su tavole che non superano i 90 gradi. Per ciascun lato è necessario inserire il numero di quadrante e i gradi, i minuti e i secondi est o ovest rispetto all'asse nord-sud. Il programma indica la direzione del lato (ad esempio SO = Sud Ovest), e converte il quadrante, i gradi, ecc. in un angolo azimutale. L'angolo viene misurato dal nord per 360 gradi.

Un lato curvo, o arco, è definito da due lati ausiliari, ciascuno dei quali è il raggio dell'arco stesso. La direzione del primo lato ausiliario è la direzione del raggio dal primo estremo incontrato dell'arco al centro stesso. È possibile calcolare questa direzione mediante la direzione della tangente dell'arco in quel punto, in quanto il raggio è perpendicolare alla tangente. La mappa può mostrare la direzione della tangente. In caso contrario, è possibile calcolarla aggiungendo alla direzione della corda un angolo, il cui valore è metà dell'angolo formato dall'arco, così come mostrato in figura.



Direzione della tangente = direzione della corda + $\delta/2$

La direzione del secondo raggio va dal centro dell'arco all'altro estremo. Il valore di questa lunghezza viene definito con un numero negativo, per segnalare al computer che questo lato e il precedente non sono lati del perimetro, ma lati ausiliari di un arco.

Il programma chiede che la distanza e la direzione vengano espressi come numeri. I lati vanno inseriti in gruppi di dieci (o meno). Dopo l'ultimo inserimento nel gruppo, è possibile correggere ciascun lato. È necessario inserire entrambi i lati ausiliari dello stesso arco nello stesso gruppo. Per terminare un gruppo, è possibile inserire la direzione O, e inserire gli altri lati in gruppi successivi.

Quando un gruppo è stato corretto, per quel gruppo viene stampata una tabella poligonale. In questa si potranno leggere ciascun numero di lato, la sua direzione, l'angolo azimutale e la distanza, oltre alla distanza incrementale e cumulativa percorsa verso nord e verso est. La distanza cumulativa percorsa verso nord e verso est risultante dopo l'ultimo lato di una rilevazione chiusa permette di stabilire l'errore di chiusura. L'angolo dell'arco, il raggio, l'area del settore, la lunghezza della corda e quella della tangente sono stampati tra i due lati ausiliari di ciascun lato curvo.

Dopo la stampa dell'ultimo lato di una rilevazione chiusa, verrà stampata l'area dell'appezzamento. Il calcolo dell'area risulta estremamente accurato, se si verificano le seguenti condizioni:

- 1° l'errore di chiusura è piccolo (0,5 cm per un lotto in cui si prevede una abitazione)
- 2° l'area è sufficientemente piccola in modo tale che la curvatura terrestre sia non significativa. Nel caso di aree superiori a parecchie decine di chilometri quadri, è necessario tener conto anche di questo ultimo fattore.

Esempio:

La seguente figura mostra un lotto con un lato curvo. I numeri dei lati sono segnati con un cerchietto. Le direzioni e le distanze sono assegnate per ogni lato. Calcolare l'errore di chiusura e l'area del lotto.


Nel programma i punti cardinali vengono indicati come segue:

N = nord

S = sud

E = est

W = ovest.



- ① S39°0''E
149.83
- ② S39°0''E
50.00
- ③ N85°23'53''W
50.00
- ④ N85°23'53''W
114.32
- ⑤ N1°5'0''E
132.78
- ⑥ N46°0'0''E
14.00
- ⑦ S89°0'0''E
25.46

CONTROLLO DI MAPPA

PERIMETRO APERTO (1) O CHIUSO (0)? 0

NUOVA SERIE DI LATI

LATO N. 1

QUADRANTE, GRADI, MINUTI, SECONDI

? 2, 39, 0, 0

DISTANZA

(NEGATIVA SE RAGGIO DIRETTO VERSO L'ESTERNO)? 149.83

LATO N. 2
 QUADRANTE, GRADI, MINUTI, SECONDI
 ? 2 , 39 , 0 , 0
 DISTANZA
 <NEGATIVA SE RAGGIO DIRETTO VERSO L'ESTERNO>? 50

LATO N. 3
 QUADRANTE, GRADI, MINUTI, SECONDI
 ? 4 , 85 , 23 , 53
 DISTANZA
 <NEGATIVA SE RAGGIO DIRETTO VERSO L'ESTERNO>? -50

LATO N. 4
 QUADRANTE, GRADI, MINUTI, SECONDI
 ? 4 , 85 , 23 , 53
 DISTANZA
 <NEGATIVA SE RAGGIO DIRETTO VERSO L'ESTERNO>? 114.32

LATO N. 5
 QUADRANTE, GRADI, MINUTI, SECONDI
 ? 1 , 1 , 5 , 0
 DISTANZA
 <NEGATIVA SE RAGGIO DIRETTO VERSO L'ESTERNO>? 132.78

LATO N. 6
 QUADRANTE, GRADI, MINUTI, SECONDI
 ? 1 , 46 , 0 , 0
 DISTANZA
 <NEGATIVA SE RAGGIO DIRETTO VERSO L'ESTERNO>? 14

LATO N. 7
 QUADRANTE, GRADI, MINUTI, SECONDI
 ? 2 , 39 , 0 , 0
 DISTANZA
 <NEGATIVA SE RAGGIO DIRETTO VERSO L'ESTERNO>? 25.46

LATO N. 8
 QUADRANTE, GRADI, MINUTI, SECONDI
 ? 0 , 0 , 0 , 0
 QUALE LATO VA CORRETTO IN QUESTA SERIE
 (0=NESSUN ALTRO CAMBIAMENTO)? 0
 ORIGINE 0 / 0

LATO/DIREZ. 1 /SE
 DIST./AZIMUT 141 0 0 / 149.83
 DEL N/DEL E -116.44 / 94.291
 DIST. N/DIST. E -116.44 / 94.291

LATO/DIREZ. 2 /SE
 DIST./AZIMUT 141 0 0 / 50
 DEL N/DEL E -38.857 / 31.466
 DIST. N/DIST. E -155.297 / 125.757

ARCO: 124 23 53
 R= 50 A= 5427.889 C= 88.457 T= 94.83

LATO/DIREZ. 3 /NE
 DIST./AZIMUT 85 23 53 / 50
 DEL N/DEL E 4.012 / 49.839
 DIST. N/DIST. E -151.285 / 175.596

** PREMERE 'C' PER NUOVA SERIE LATI **

LATO/DIREZ. 4 /NE
 DIST./AZIMUT 85 23 53 / 114.32
 DEL N/DEL E 9.172 / 113.951
 DIST. N/DIST. E -142.113 / 289.547

LATO/DIREZ. 5 /NE
 DIST./AZIMUT 1 5 0 / 132.78

DEL N/DEL E 132.756 / 2.51
DIST. N/DIST. E -9.357 / 292.058

LATO/DIREZ. 6 /NE
DIST./AZIMUT 46 0 0 / 14
DEL N/DEL E 9.725 / 10.071
DIST. N/DIST. E .368 / 302.129

** PREMERE 'C' PER NUOVA SERIE LATI **

LATO/DIREZ. 7 /SE
DIST./AZIMUT 91 0 0 / 25.46
DEL N/DEL E -.444 / 25.456
DIST. N/DIST. E -.076 / 327.585

ALTRI DATI (1=SI, 0=NO)? 0
L'AREA E' 31626.573 MQ.

Listing del programma

```
10 REM - CONTROLLO DI MAPPA
12 REM - PER MAPPE CHIUSE SEGUIRE LA DIREZIONE ORARIA
13 REM - INIZIARE COL PUNTO A DESTRA DI OGNI LATO DEL PERIMETRO
14 REM - CALCOLARE I LATI AUSILIARI COME RAGGI AD OGNI
15 REM - ESTREMITA' DI CIASCUN ARCO. ARCHI < 180 GRADI
16 REM
17 REM - VALORE DI PI GRECO, PI=3.14159265
18 REM - K0= NUMERO DI LATI PER VOLTA
20 K0=10
30 DIM B(10),L(10)
39 REM - FNR(X) ARROTONDA X A TRE NUMERI DECIMALI
40 DEF FNR(X)=INT(X*1000+.5)/1000
49 REM - R E' IL FATTORE DI CONVERSIONE PER GRADI IN RADIANTI
50 R=1.745329251E-2
60 PRINT "CONTROLLO DI MAPPA":PRINT
70 PRINT "PERIMETRO APERTO (1) O CHIUSO (0)":
80 INPUT F
90 IF F=0 THEN 120
100 PRINT "ORIGINE DISTANZA NORD, DISTANZA EST":
110 INPUT N,E
120 PRINT:PRINT "NUOVA SERIE DI LATI " :PRINT
125 G=H
130 FOR K=1 TO K0
139 REM - INSERIRE LA DIREZIONE E LA DISTANZA PER IL PROSSIMO LATO
140 GOSUB 2000
149 REM - SE LA DIREZIONE E' 0, SI INSERISCE UN'ALTRA SERIE DI LATI
150 IF Q=0 THEN 170
155 G=G+1
160 GOTO 240
169 REM - ZERO LATI INUTILIZZATI IN QUESTA SERIE
170 IF K=K0 THEN 230
180 FOR J=K+1 TO K0
190 B(J)=0
200 L(J)=0
210 NEXT J
230 K=K0
240 NEXT K
260 PRINT "QUALE LATO VA CORRETTO IN QUESTA SERIE"
261 PRINT " (0=NESSUN ALTRO CAMBIAMENTO)":
270 INPUT K
279 REM - NESSUN ALTRO CAMBIAMENTO SE SI INSERISCE 0
280 IF K=0 THEN 310
285 K=K-H
290 GOSUB 2000
300 GOTO 260
309 REM - CALCOLO DEI VALORI E STAMPA DELLA TABELLA
```

```

310 PRINT "ORIGINE ";FNR(N);"/";FNR(E)
340 PRINT
350 FOR K=1 TO K0
360 L1=L(K)
361 Z=Z+1
369 REM - RILEVAMENTO PER L'ARCO
370 IF L1<0 THEN 1100
380 IF L1=0 THEN 900
388 REM - CALCOLO DELL'INCREMENTO DI DISTANZA A NORD E A SUD
389 REM -(CONVERSIONE DELLA DIREZIONE DA GRADI A RADIANTI)
390 L=L(K)*COS(B(K)*R)
400 D=L(K)*SIN(B(K)*R)
410 N=N+L
420 E=E+D
429 REM - INCREMENTO DELL'AREA
430 A=A-E*L+N*D
440 PRINT "LATO/DIREZ.";TAB(17);H+K;"/";
449 REM - DETERMINARE LA DIREZIONE IN BASE AL RILEVAMENTO
450 IF B(K)=0 THEN 470
460 GOTO 490
470 PRINT "N"
480 GOTO 830
490 IF B(K)<90 THEN 510
500 GOTO 530
510 PRINT "NE"
520 GOTO 830
530 IF B(K)=90 THEN 550
540 GOTO 570
550 PRINT "E"
560 GOTO 830
570 IF B(K)<180 THEN 590
580 GOTO 610
590 PRINT "SE"
600 GOTO 830
610 IF B(K)=180 THEN 630
620 GOTO 650
630 PRINT "S"
640 GOTO 830
650 IF B(K)<270 THEN 670
660 GOTO 690
670 PRINT "SO"
680 GOTO 830
690 IF B(K)=270 THEN 710
700 GOTO 730
710 PRINT "O"
720 GOTO 830
730 IF B(K)<360 THEN 750
740 GOTO 770
750 PRINT "NO"
760 GOTO 830
770 IF B(K)=360 THEN 790
780 GOTO 810
790 PRINT "N"
800 GOTO 830
810 B(K)=B(K)-360
820 GOTO 450
829 REM - DEFINIRE LA DIREZIONE IN GRADI, MINUTI, SECONDI
830 D1=INT(B(K))
840 M1=(B(K)-D1)*60
850 M=INT(M1)
860 S=INT((M1-M)*60+.5)
870 PRINT "DIST./AZIMUT ";D1;"/N/S";"/";FNR(L(K))
880 PRINT "DEL N/DEL E ";FNR(L);"/";FNR(D)
881 PRINT "DIST. N/DIST. E ";FNR(N);"/";FNR(E)
885 PRINT
890 L(K)=L1
891 IF Z<3 THEN 900
895 PRINT "*** PREMERE 'C' PER NUOVA SERIE LATI ***"
896 REM GET W$:IF W$="" THEN 896

```

```

897 PRINT
898 Z=0
900 NEXT K
910 H=0
920 PRINT "ALTRI DATI (1=SI, 0=NO)",
930 INPUT U
940 IF U<>0 THEN 120
949 REM - NON SI CALCOLA L'AREA PER SPEZZATE APERTE
950 IF F<>0 THEN 1000
960 A=ABS(A/2)
970 PRINT "L AREA E'",FNR(A);"MQ."
980 PRINT
1000 END
1099 REM - SI CALCOLANO I LATI CURVI E LI SIS INSERISCE NELLA TABELLA
1100 C=ABS(B(K)-B(K-1))
1110 C=ABS(180-C)
1120 D=-L1
1130 L(K)=D
1140 A1=C/180*PI*D
1150 C1=2*D*SIN(C/2/R)
1160 T=D*TAN(C/2/R)
1170 B9=B(K)-B(K-1)
1180 IF B9<-180 THEN 1210
1190 IF B9>180 THEN 1210
1200 IF B9<0 THEN 1230
1210 A=A+A1
1220 GOTO 1240
1230 A=A-A1
1240 D1=INT(C)
1250 M1=(C-D1)*60
1260 M=INT(M1)
1270 S=INT((M1-M)*60+.5)
1280 PRINT "ARCO:"D1,M;S:FNRINT " R=";FNR(D);"A=";FNR(A1);"C=";
1290 PRINT FNR(C1);"T=";FNR(T)
1300 PRINT
1320 GOTO 390
1999 REM -INSERIRE I DATI DI UN LATO
2000 B(K)=0
2010 L(K)=0
2020 PRINT "LATO N. ";H+K
2021 PRINT "QUADRANTE, GRADI,";
2022 PRINT "MINUTI, SECONDI"
2030 INPUT Q,D,M,S
2040 IF Q=0 THEN 2270
2050 IF Q>4 THEN 2020
2060 IF Q<0 THEN 2020
2070 IF D<0 THEN 2020
2080 IF M<0 THEN 2020
2090 IF S<0 THEN 2020
2100 B(K)=D+(M+S/60)/60
2110 IF B(K)>90 THEN 2020
2120 IF Q=1 THEN 2220
2130 IF Q=2 THEN 2150
2140 GOTO 2170
2150 B(K)=180-B(K)
2160 GOTO 2220
2170 IF Q=3 THEN 2190
2180 GOTO 2210
2190 B(K)=180+B(K)
2200 GOTO 2220
2210 IF Q>4 THEN 2220
2220 PRINT "DISTANZA "
2221 PRINT "(NEGATIVA SE RAGGIO DIRETTO VERSO L'ESTERNO)";
2230 INPUT L(K)
2240 IF L(K)>0 THEN 2270
2250 IF ABS(L(K))>ABS(L(K-1)) THEN 2220
2270 RETURN
9999 END

```

GIORNO DELLA SETTIMANA

Questo programma calcola quale giorno della settimana corrisponde ad una certa data. Per esempio, il 25 Dicembre 1980 è caduto di giovedì.

È necessario inserire la data in forma numerica, nell'ordine giorno (D), mese (M), anno (Y). Ad esempio, il 12 Settembre 1975 va inserito come 12,9,1975, facendo attenzione che i vari elementi siano separati da virgole e non da trattini o altro.

Esempi:

Francesca è nata il 4 Marzo 1953. Che giorno era?

Lo zio Giovanni ebbe un appuntamento il 30 settembre 1977. Che giorno era?

```
GIORNO DELLA SETTIMANA
<INSERIRE 0,0,0 PER TERMINARE IL PROGRAMMA>
GIORNO,MESE,ANNO? 4 , 3 , 1953
MERCOLEDI'

GIORNO,MESE,ANNO? 30 , 9 , 1977
VENERDI'

GIORNO,MESE,ANNO? 0 , 0 , 0
```

Listing del programma

```
10 PRINT "GIORNO DELLA SETTIMANA"
20 PRINT
29 REM - SI RICHIEDE L'INSERIMENTO DI DATI DA PARTE DELL'UTENTE
30 PRINT "<INSERIRE 0,0,0 PER TERMINARE IL PROGRAMMA>"
40 PRINT "GIORNO,MESE,ANNO:"
50 INPUT D,M,Y
60 IF M<0 THEN 100
70 IF D<0 THEN 100
80 IF Y<0 THEN 100
90 GOTO 360
99 REM - E' NECESSARIO RISISTEMARE I DATI INTRODOTTI PER I CALCOLI?
100 IF M<2 THEN 130
109 REM - RISISTEMARE I DATI INTRODOTTI
110 M=M+12
120 Y=Y-1
129 REM - CALCOLA IL NUMERO DEL GIORNO
130 N=D+2*M+INT(.6*(M+1))+Y+INT(Y/4)-INT(Y/100)+INT(Y/400)+2
140 N=INT((N/7-INT(N/7))*7+.5)
149 REM - TROVA IL NUMERO CORRETTO DEL GIORNO, LO TRASFORMA NEL GIORNO, STAMPA
150 IF N<0 THEN 180
160 PRINT "SABATO"
170 GOTO 340
180 IF N<1 THEN 210
```

```

120 PRINT "DOMENICA"
130 GOTO 340
140 IF ND=2 THEN 240
150 PRINT "LUNEDI'"
160 GOTO 340
170 IF ND=3 THEN 270
180 PRINT "MARTEDI'"
190 GOTO 340
200 IF ND=4 THEN 300
210 PRINT "MERCOLEDI'"
220 GOTO 340
230 IF ND=5 THEN 330
240 PRINT "GIOVEDI'"
250 GOTO 340
260 PRINT "VENERDI'"
270 PRINT
280 REM - IL PROGRAMMA RIPARTE
290 GOTO 40
300 END

```

NUMERO DI GIORNI TRA DUE DATE

Questo programma calcola il numero di giorni intercorrenti tra due date. Si tiene conto degli anni bisestili. Il programma prende come ipotesi che tra oggi e domani intercorra un giorno. Così, per esempio, vi sono due giorni tra il 1° Marzo e il 3 Marzo dello stesso anno.

Vi sono alcune precauzioni necessarie per assicurare un uso appropriato del programma. Come prima cosa, bisogna inserire innanzitutto la prima tra le due date. In secondo luogo, le date vanno inserite in forma numerica (3 invece di Marzo) e nell'ordine corretto (giorno, mese, anno; ad es. 17,3,1976). I vari elementi devono essere separati da virgole e non da trattini o altro. In terzo luogo, l'anno non deve essere abbreviato (1976 e non 76) anche se entrambe le date sono dello stesso secolo. Per ultimo, il numero inserito per il mese non deve essere più grande di dodici e il numero dei giorni non può essere più grande del numero di giorni relativi al mese inserito. In questi casi, verrà stampato il messaggio DATA IMPOSSIBILE, il quale segnalerà che è stata inserita una data assurda, del tipo 32, 14, 1975. È possibile comunque che si abbia una risposta inesatta.

Nota: i dati vanno inseriti in quest'ordine: mese (M), giorno (D), anno (Y)

Esempio:

Giovanni è nato l'8 Agosto 1951. Quanti giorni avrà per il suo trentesimo compleanno?

NUMERO DI GIORNI TRA DUE DATE

```
PRIMA DATA? 8 , 8 , 1951
SECONDA DATA? 8 , 8 , 1981
DIFFERENZA = 10958 GIORNI

ALTRI DATI (1=SI, 0=NO)? 0
```

Listing del programma

```
10 PRINT"NUMERO DI GIORNI TRA DUE DATE"
20 PRINT
30 REM - LE ISTRUZIONI DALLA RIGA 30 ALLA 50 RICHIEDONO L'INTRODUZIONE DI DATI
40 REM - DA PARTE DELL'UTENTE
50 PRINT "PRIMA DATA";
60 INPUT D1,M1,Y1
70 PRINT "SECONDA DATA";
80 INPUT D2,M2,Y2
90 REM - ASSEGNA LE VARIABILI DA UTILIZZARE NEL SOTTOPROGRAMMA
100 M=M1
110 D=D1
120 Y=Y1
130 GOSUB 230
140 REM - SALVA IL NUMERO DI GIORNI CALCOLATO IN N
150 N=A
```



```

119 REM - ASSEGNA LE VARIABILI DA UTILIZZARE NEL SOTTOPROGRAMMA
120 M=M2
130 D=D2
140 Y=Y2
150 GOSUB 230
159 REM - CALCOLA LA DIFFERENZA E LA STAMPA
160 N=A-N
170 PRINT "DIFFERENZA =" ; N ; "GIORNI"
180 PRINT
189 REM - IL PROGRAMMA RIPARTE O TERMINA? SI RICHIEDE LA RISPOSTA DELL'UTENTE
190 PRINT "ALTRI DATI (1=SI, 0=NO)";
200 INPUT X
210 IF X=1 THEN 20
219 REM - FINE PROGRAMMA
220 GOTO 460
226 REM - SOTTOPROGRAMMA PER CALCOLARE IL NUMERO DI GIORNI INTERCORRENTI FRA
227 REM - 0/0/0 E G/M/A. INIZIA CON UN TEST DI CONTROLLO SULLA CONGRUENZA DELLA
228 REM - DATA.
229 REM - RAGGIUNGE IL TEST OPPORTUNO IN BASE AL NUMERO DI GIORNI DEL MESE.
230 ON M GOTO 260,280,260,340,260,340,260,260,340,260,340,260
240 PRINT "DATA IRREALE"
249 REM - INTERROMPERE I CALCOLI, RITORNARE AL PROGRAMMA PRINCIPALE
250 RETURN
259 REM - IL MESE HA 31 GIORNI
260 IF D>31 THEN 240
270 GOTO 350
279 REM - IL MESE E' FEBBRAIO; L'ANNO E' BISESTILE?
280 IF Y/4<>INT(Y/4) THEN 310
290 IF Y/400=INT(Y/400) THEN 320
300 IF Y/100<>INT(Y/100) THEN 320
309 REM - NON E' UN ANNO BISESTILE, IL MESE HA 28 GIORNI
310 IF D>28 THEN 240
319 REM - E' UN ANNO BISESTILE; IL MESE HA 29 GIORNI
320 IF D>29 THEN 240
330 GOTO 350
339 REM - IL MESE HA 30 GIORNI
340 IF D>30 THEN 230
349 REM-TABELLA DEL NUMERO DI GIORNI DAL PRIMO DELL'ANNO AL PRIMO DI OGNI MESE
350 DATA 0,31,59,90,120,151,181,212
351 DATA 243,273,304,334
360 RESTORE
361 FOR Q=1 TO M
362 READ A
363 NEXT Q
368 REM - OTTENERE IL NUMERO DI GIORNI DAL 1 DI GENNAIO AL PRIMO DEL MESE
369 REM - DALLA TAVOLA DEI DATI
379 REM - CALCOLA IL NUMERO DI GIORNI DA 0/0/0 A G/M/A
380 A=A+Y*365+INT(Y/4)+D+1-INT(Y/100)+INT(Y/400)
389 REM - PUO' ESSERE UN ANNO BISESTILE
390 IF INT(Y/4)<>Y/4 THEN 450
409 REM - PROSEGUI IL TEST PER ANNO BISESTILE
410 IF Y/400=INT(Y/400) THEN 430
420 IF Y/100=INT(Y/100) THEN 440
428 REM - SE IL MESE E' GENNAIO O FEBBRAIO, RIASSESTARE IL NUMERO DI GIORNI
429 REM - CALCOLATO
430 IF M>2 THEN 450
440 A=A-1
449 REM - FINE DEL SOTTOPROGRAMMA, RIENTRARE NEL PROGRAMMA PRINCIPALE
450 RETURN
460 END

```

OPZIONE

Si potrebbe, per accorciare il programma, omettere il test relativo alle date impossibili. Si deve rilevare che, se viene inserito un mese con un numero maggio-

re di 12, quando questo test viene omissso, ne deriverà un errore di entrata. Le linee di programma che potrebbero essere tolte sono elencate nell'esempio successivo.

Esempio:

Quanti giorni intercorrono tra il 4 Luglio e Natale?

NUMERO DI GIORNI TRA DUE DATE

PRIMA DATA? 4 , 7 , 1977

SECONDA DATA? 25 , 12 , 1977

DIFFERENZA = 174 GIORNI

ALTRI DATI (1=SI, 0=NO)? 0

Opzione

1 REM - OPZIONE 110,150

10 PRINT"NUMERO DI GIORNI TRA DUE DATE"

⋮

30 Y=Y1

100 GOSUB 350

109 REM - SALVA IL NUMERO DI GIORNI CALCOLATO IN N

⋮

140 Y=Y2

150 GOSUB 350

159 REM - CALCOLA LA DIFFERENZA E LA STAMPA

⋮

227 REM - 0/0/0 E G/M/A.

⋮

(Cancellare le rishe da 228 a 340)

⋮

348 REM - TABELLA DEL NUMERO DI GIORNI DAL PRIMO DELL'ANNO AL PRIMO DI OGNI

349 REM - MESE

⋮

460 END

CONVERSIONI DI UNITA' DI MISURA

Questo programma converte una misura data in unità inglesi in unità metriche. Le conversioni disponibili in questo programma sono le seguenti:

1	Inches	(pollici)	in centimetri
2	Feet	(piedi)	in centimetri
3	Feet	(piedi)	in metri
4	Yards	(yarde)	in metri
5	Miles	(miglia)	in chilometri
6	Teaspoons		in centimetri cubi
7	Teblespoons		in centimetri cubi
8	Cups		in litri
9	Pints	(pinte)	in litri
10	Quarts		in litri
11	Gallons	(galloni)	in litri
12	Bushels		in litri
13	Pecks		in litri
14	Ounces	(once)	in grammi
15	Pounds		in chilogrammi
16	Tons		in chilogrammi
17	Gradi Fahrenheit		in gradi Celsius

È necessario dare il valore della misura inglese e il numero della conversione (dall'1 al 17 come elencato più sopra) con la quale si desidera operare.

Esempio:

Operare le seguenti conversioni:

8,5 miles = chilometri

75° Fahrenheit = °Celsius

10 Gallons = litri.

CONVERSIONI DI UNITA' DI MISURA

INTRODURRE 0 PER TERMINARE IL PROGRAMMA
QUALE CONVERSIONE VI OCCORRE? 5

VALORE DA CONVERTIRE ? 8.5

8.5 MIGLIA = 13.6765 CHILOMETRI

QUALE CONVERSIONE VI OCCORRE? 17

VALORE DA CONVERTIRE ? 75

75 GRADI FARENHEIT = 23.8888889 CELSIUS

QUALE CONVERSIONE VI OCCORRE? 11

VALORE DA CONVERTIRE ? 10

10 GALLONI = 37.85 LITRI

QUALE CONVERSIONE VI OCCORRE? 0

Listing del programma

```
10 PRINT "CONVERSIONI DI UNITA' DI MISURA"
20 PRINT
29 REM - STABILIRE LE VARIABILI PER 17 FATTORI DI CONVERSIONE
30 DIM C(17)
39 REM - CICLO PER ASSEGNARE I FATTORI DI CONVERSIONE IN C()
40 FOR N=1 TO 17
50 READ C(N)
60 NEXT N
69 REM - TABELLA DEI DATI PER 17 FATTORI DI CONVERSIONE
70 DATA 2.540,30.480,.3048,.9144
71 DATA 1.609,4.929,14.788,.2366
72 DATA .4732,.9463,3.785,35.24
73 DATA 8.809,28.3495,.4536,907.2
80 DATA .6214
89 REM - PRENDERE IL NUMERO DI CONVERSIONE DALLA DESCRIZIONE DEL PROGRAMMA
90 PRINT "INTRODURRE 0 PER TERMINARE IL PROGRAMMA"
100 PRINT "QUALE CONVERSIONE ";
101 PRINT "VI OCCORRE";
110 INPUT N
119 REM - FINE DEL PROGRAMMA ?
120 IF N=0 THEN 540
129 REM - CONVERSIONE DISPONIBILE ?
130 IF N>17 THEN 100
140 PRINT "VALORE DA CONVERTIRE ";
150 INPUT I
159 REM - SI EFFETUA LA CONVERSIONE USANDO IL GIUSTO FATTORE DI CONVERSIONE
160 R=I*C(N)
169 REM - DIRIGE IL PROGRAMMA ALLE UNITA' DI CONVERSIONE OPPORTUNE, STAMPA I
169 REM - RISULTATI
170 IF N<10 THEN 175
173 ON N-9 GOTO 360,380,400,420,440,460,480,500
175 ON N GOTO 180,200,220,240,260,280,300,320,340
180 PRINT I;"POLLCICI =";R;"CENTIMETRI"
190 GOTO 520
200 PRINT I;"PIEDI =";R;"CENTIMETRI"
210 GOTO 520
220 PRINT I;"PIEDI =";R;"METRI"
230 GOTO 520
240 PRINT I;"YARDE =";R;"METRI"
250 GOTO 520
260 PRINT I;"MIGLIA =";R;"CHILOMETRI"
270 GOTO 520
280 PRINT I;"TSP. =";R;
281 PRINT "CENTIMETRI CUBI"
290 GOTO 520
300 PRINT I;"TBSP. =";R;
301 PRINT "CENTIMETRI CUBI"
310 GOTO 520
320 PRINT I;"CUPS =";R;"LITRI"
330 GOTO 520
```

```

340 PRINT I;"PINTE =";R;"LITRI"
350 GOTO 520
360 PRINT I;"QUARTS =";R;"LITRI"
370 GOTO 520
380 PRINT I;"GALLONI =";R;"LITRI"
390 GOTO 520
400 PRINT I;"BUSHELS =";R;"LITRI"
410 GOTO 520
420 PRINT I;"PECKS =";R;"LITRI"
430 GOTO 520
440 PRINT I;"ONCE =";R;"GRAMMI"
450 GOTO 520
460 PRINT I;"LIBBRE =";R;"CHIOLOGRAMMI"
470 GOTO 520
480 PRINT I;"TONS =";R;"CHIOLOGRAMMI"
490 GOTO 520
499 REM - CONVERSIONE DI GRADI FARENHEIT IN GRADI CELSIUS
500 R=(I-32)*5/9
510 PRINT I;"GRADI FARENHEIT =";R;
511 PRINT "CELSIUS"
520 PRINT PRINT
529 REM - IL PROGRAMMA RIFARTE
530 GOTO 100
540 END

```

ORDINAMENTO ALFABETICO

Questo programma mette in ordine alfabetico un elenco di parole o di frasi.

I numeri possono essere parti di una frase alfanumerica. Tuttavia, non possono essere messi in ordine numerico, a meno che contengano lo stesso numero di digits. I numeri con meno digits possono essere presi in considerazione, aggiungendo il necessario numero di zeri alla loro sinistra. Così, se i numeri da classificare si estendono alle centinaia, il numero 13 dovrà essere inserito come 013.

Per risparmiare spazi di memoria, lo statement 70 viene scritto in modo da limitarsi al massimo numero di termini che si vogliono ordinare alfabeticamente. La dimensione dello statement può essere variata nella seguente forma:

70 DIM A \$(N)

dove N = numero di voci da ordinare.

Esempio:

Mettere in ordine alfabetico i seguenti nomi:

Robert Wilson
Susan W. James
Kent Smith
Michael Mitchell
Ann T. McGowan
Alexander Lee II
Mary Mitchell
David Bowers
Steven Evans
Carol Jameson
Linda North

```
70 DIM A$(11)
READY.
```

ORDINAMENTO ALFABETICO

```
PER TERMINARE IL PROGRAMMA INTRODURRE 0
NUMERO DI VOCI? 11
VOCE 1 ? WILSON ROBERT
VOCE 2 ? JAMES SUSAN W.
VOCE 3 ? SMITH KENT
VOCE 4 ? MITCHELL MICHAEL
VOCE 5 ? MCGOWAN ANN T.
```

VOCE 6 ? LEE ALEXANDER II
 VOCE 7 ? MITCHELL MARY
 VOCE 8 ? BOWERS DAVID
 VOCE 9 ? EVANS STEVEN
 VOCE 10 ? JAMESON CAROL
 VOCE 11 ? NORTH LINDA
 BOWERS DAVID
 EVANS STEVEN
 JAMES SUSAN W.
 JAMESON CAROL
 LEE ALEXANDER II
 MCGOWAN ANN T.
 MITCHELL MARY
 MITCHELL MICHAEL
 NORTH LINDA
 SMITH KENT
 WILSON ROBERT

NUMERO DI VOCI? 0

Listing del programma

```

10 PRINT "ORDINAMENTO ALFABETICO"
20 PRINT
30 PRINT "PER TERMINARE IL PROGRAMMA INTRODURRE 0"
40 PRINT "NUMERO DI VOCI";
50 INPUT N
60 IF N=0 THEN 330
69 REM - VALORE MASSIMO DEL NUMERO DELLE VOCI DA INSERIRE DI VOLTA IN VOLTA
70 DIM A$(N)
80 FOR I=1 TO N
90 PRINT "VOCE";I;
100 INPUT A$(I)
110 NEXT I
120 M=N
126 REM - LA TECNICA DI ORDINAMENTO USATA CONFRONTA LE VOCI DATE SECONDO
127 REM - INCREMENTI DECRESCENTI. IL PRIMO PASSO CONFRONTA N/2 ELEMENTI
128 REM - SEPARATAMENTE, IL SECONDO (N/2)/2, E COSI' VIA FINCHE' L'INCREMENTO
129 REM - E' ESAURITO.
130 T=M/2
131 M=INT(T)
140 IF M=0 THEN 280
150 K=N-M
160 J=1
170 I=J
180 L=I+M
190 IF A$(I)<=A$(L) THEN 250
200 T=A$(I)
210 A$(I)=A$(L)
220 A$(L)=T
230 I=I-M
240 IF I>=1 THEN 180
250 J=J+1
260 IF J>K THEN 130
270 GOTO 170
280 FOR I=1 TO N
281 REM - VERIFICA DI SCHERMO COMPLETO (20 RIGHE)
282 IF I/20<>INT(I/20) THEN 290
283 REM - ASPETTARE L'ORDINE DELL'OPERATORE PER PASSARE AL NUOVO SCHERMO
284 PRINT "INTRODUCI 'C' PER CONTINUARE";
285 INPUT W$:PRINT"? "W$
290 PRINT A$(I)
300 NEXT I
310 PRINT
319 REM - IL PROGRAMMA RIPARTE
320 GOTO 40
330 END
  
```

OPZIONE

È possibile ordinare un elenco secondo l'ordine alfabetico inverso, o partendo dai numeri più alti o da quelli più bassi. I necessari cambiamenti di programma sono elencati nell'esempio successivo.

Esempio:

I punteggi di una gara tra studenti spaziano dall'82 al 117. Si ordinino gli studenti in ordine decrescente rispetto al loro punteggio.

89 Bowers
102 Evans
111 James
100 Jameson
99 Lee
117 McGowan
102 Mitchell
82 Mitchell
97 North
91 Smith
108 Wilson

```
70 DIM A$(11)
READY.
```

ORDINAMENTO ALFABETICO

```
PER TERMINARE IL PROGRAMMA INTRODURRE 0
NUMERO DI VOCI? 11
VOCE 1 ? 089 BOWERS
VOCE 2 ? 102 EVANS
VOCE 3 ? 111 JAMES
VOCE 4 ? 100 JAMESON
VOCE 5 ? 099 LEE
VOCE 6 ? 117 MCGOWAN
VOCE 7 ? 102 MITCHELL
VOCE 8 ? 082 MITCHELL
VOCE 9 ? 097 NORTH
VOCE 10 ? 091 SMITH
VOCE 11 ? 108 WILSON
117 MCGOWAN
111 JAMES
108 WILSON
102 MITCHELL
102 EVANS
100 JAMESON
099 LEE
097 NORTH
091 SMITH
089 BOWERS
082 MITCHELL

NUMERO DI VOCI? 0
```

```

1 REM - ORDINAMENTO ALFABETICO -OPZIONE 190
:
:
10 PRINT "ORDINAMENTO ALFABETICO"
:
:
130 L=I+M
190 IF A$(I)>=A$(L) THEN 250
200 T#=A$(I)

```


L. 12.000

Cod. 551 D ISBN 88-7056-111-9

Il volume raccoglie ben 75 programmi originali scritti in un Basic generico, al fine di renderli direttamente ed immediatamente utilizzabili, salvo poche e semplici modifiche, sulla maggior parte dei personal computer in commercio, a cassetta come a floppy disk.

Per ciascuno, dopo una descrizione introduttiva, viene fornito il listing e un campione di esecuzione. Così come sono, i programmi proposti (tutti verificati) costituiscono un valido ausilio per chiunque, non avendo tempo né voglia di programmare, deve risolvere problemi di matematica, statistica, finanza o, genericamente, di pratica utilità. Oppure, mediante rielaborazione, con ampliamenti e modifiche, tagli e giunzioni di parti di essi, sono un'ottima base di partenza per tutti coloro che intendono prender dimestichezza con la programmazione e nello stesso tempo personalizzare i programmi illustrati.



57 PROGRAMMI IN BASIC PER IL VOSTRO COMPUTER

Lon Poole
Mary Borchers
Carrol Donahue

GRUPPO
EDITORIALE
JACKSON

